Revised^7 Report on the Algorithmic Language Scheme

Alex Shinn and John Cowan (*Editors*)
Aaron W. Hsu
Arthur A. Gleckler
Bradley Lucier
Emmanuel Medernach
Alaric Snell-pym
Benjamin L. Russel
David Rush
Jeffrey T. Read

9 March 2011

SUMMARY

The report gives a defining description of the programming language Scheme. Scheme is a statically scoped and properly tail-recursive dialect of the Lisp programming language invented by Guy Lewis Steele Jr. and Gerald Jay Sussman. It was designed to have an exceptionally clear and simple semantics and few different ways to form expressions. A wide variety of programming paradigms, including imperative, functional, and message passing styles, find convenient expression in Scheme.

The introduction offers a brief history of the language and of the report.

The first three chapters present the fundamental ideas of the language and describe the notational conventions used for describing the language and for writing programs in the language.

Chapters 4 and 5 describe the syntax and semantics of expressions, definitions, programs, and modules.

Chapter 6 describes Scheme's built-in procedures, which include all of the language's data manipulation and input/output primitives.

Chapter 7 provides a formal syntax for Scheme written in extended BNF, along with a formal denotational semantics. An example of the use of the language follows the formal syntax and semantics.

Chapter A provides a list of the standard modules and the identifiers that they export.

Chapter B provides a list of optional but standardized implementation features.

The report concludes with a list of references and an alphabetic index.

***** DRAFT*****
July 23, 2011 July 25, 2011

CONTENTS

INTRODUCTION

Programming languages should be designed not by piling feature on top of feature, but by removing the weaknesses and restrictions that make additional features appear necessary. Scheme demonstrates that a very small number of rules for forming expressions, with no restrictions on how they are composed, suffice to form a practical and efficient programming language that is flexible enough to support most of the major programming paradigms in use today.

Scheme was one of the first programming languages to incorporate first class procedures as in the lambda calculus, thereby proving the usefulness of static scope rules and block structure in a dynamically typed language. Scheme was the first major dialect of Lisp to distinguish procedures from lambda expressions and symbols, to use a single lexical environment for all variables, and to evaluate the operator position of a procedure call in the same way as an operand position. By relying entirely on procedure calls to express iteration, Scheme emphasized the fact that tail-recursive procedure calls are essentially goto's that pass arguments. Scheme was the first widely used programming language to embrace first class escape procedures, from which all previously known sequential control structures can be synthesized. A subsequent version of Scheme introduced the concept of exact and inexact numbers, an extension of Common Lisp's generic arithmetic. More recently, Scheme became the first programming language to support hygienic macros, which permit the syntax of a block-structured language to be extended in a consistent and reliable manner.

Background

The first description of Scheme was written in 1975 [30]. A revised report [27] appeared in 1978, which described the evolution of the language as its MIT implementation was upgraded to support an innovative compiler [28]. Three distinct projects began in 1981 and 1982 to use variants of Scheme for courses at MIT, Yale, and Indiana University [23, 19, 12]. An introductory computer science textbook using Scheme was published in 1984 [3].

As Scheme became more widespread, local dialects began to diverge until students and researchers occasionally found it difficult to understand code written at other sites. Fifteen representatives of the major implementations of Scheme therefore met in October 1984 to work toward a better and more widely accepted standard for Scheme. Their report, the RRRS [6], was published at MIT and Indiana University in the summer of 1985. Further revision took place in the spring of 1986, resulting in the R^{3}RS [25]. Work in the spring of 1988 the resulted in R^{4}RS [8], which became the basis for the IEEE Standard for the Scheme Programming Language in 1991 [15]. In 1998, several additions to the IEEE standard, including high-level hygienic macros, multiple return values and `eval`, were finalized as the R^{5}RS [2].

In the Fall of 2006, work began on a more ambitious standard, including many new improvements and a general change in style from descriptive, reporting on the state of existing implementations, to prescriptive, specifying how a conformant implementation should behave. The resulting standard, the R^{6}RS, was completed in August 2007 [1], and was organized as a core language and set of standard libraries. The size and goals of the R^{6}RS, however, were controversial, and in a poll taken many Scheme implementors reported no intention of updating to the new standard.

In August 2009, the Scheme Steering Committee decided to divide the standard into two separate but compatible languages — a "small" language, suitable for educators, researchers and embedded languages, focused on R^{5}RS compatibility, and a "large" language focused on the practical needs of mainstream software development. The present report describes the "small" language of that effort.

We intend this report to belong to the entire Scheme community, and so we grant permission to copy it in whole or in part without fee. In particular, we encourage implementors of Scheme to use this report as a starting point for manuals and other documentation, modifying it as necessary.

Acknowledgements

We would like to thank the members of the Steering Committee, William Clinger, Marc Feeley, Chris Hanson, Jonathan Rees and Olin Shivers for their support and guidance. We'd like to thank the following people for their help: Per Bothner, Taylor Campbell, Ray Dillinger, Brian Harvey, Shiro Kawai, Jonathan Kraut, Thomas Lord, Vincent Manis, Jeronimo Pellegrini, Jussi Piitulainen, Alex Queiroz, Jim Rees, Jay Reynolds Freeman, Malcolm Tredinnick, Denis Washington, Andy Wingo, Andre van Tonder and hopefully many before before this report is finalized!

In addition we would like to thank all the past editors, and the people who helped them in turn: Alan Bawden, Michael Blair, George Carrette, Andy Cromarty, Pavel Curtis, Jeff Dalton, Olivier Danvy, Ken Dickey, Bruce Duba, Andy Freeman, Richard Gabriel, Yekta Gürsel, Ken Haase, Robert Hieb, Paul Hudak, Morry Katz, Chris Lindblad, Mark Meyer, Jim Miller, Jim Philbin, John Ramsdell, Mike Shaff, Jonathan Shapiro, Julie Sussman, Perry Wagle, Daniel Weise, Henry Wu, and Ozan Yigit. We thank Carol Fessenden, Daniel Friedman, and Christopher Haynes for permission to use text from the Scheme 311 version 4 reference manual. We thank Texas Instruments, Inc. for permission to use text from the *TI Scheme Language Reference Manual*[32]. We gladly acknowledge the

influence of manuals for MIT Scheme[19], T[24], Scheme 84[13],Common Lisp[29], and Algol 60[20].

DESCRIPTION OF THE LANGUAGE

1. Overview of Scheme

1.1. Semantics

This section gives an overview of Scheme's semantics. A detailed informal semantics is the subject of chapters 3 through 6. For reference purposes, section 7.2 provides a formal semantics of Scheme.

Following Algol, Scheme is a statically scoped programming language. Each use of a variable is associated with a lexically apparent binding of that variable.

Scheme is a dynamically typed language. Types are associated with values (also called objects) rather than with variables. Other dynamically typed languages include APL, Common Lisp, JavaScript, Perl, Python, Ruby, and Smalltalk. Statically typed languages, by contrast, associate types with variables and expressions as well as with values, and include Algol 60, C, C++, C#, Haskell, Java, ML, and Pascal.

All objects created in the course of a Scheme computation, including procedures and continuations, have unlimited extent. No Scheme object is ever destroyed. The reason that implementations of Scheme do not (usually!) run out of storage is that they are permitted to reclaim the storage occupied by an object if they can prove that the object cannot possibly matter to any future computation. Other languages in which most objects have unlimited extent include APL, C#, Common Lisp, Haskell, Java, JavaScript, ML, Perl, Python, Ruby, and Smalltalk.

Implementations of Scheme are required to be properly tail-recursive. This allows the execution of an iterative computation in constant space, even if the iterative computation is described by a syntactically recursive procedure. Thus with a properly tail-recursive implementation, iteration can be expressed using the ordinary procedure-call mechanics, so that special iteration constructs are useful only as syntactic sugar. See section 3.5. Other languages that are required to be properly tail-recursive include Haskell and ML.

Scheme procedures are objects in their own right. Procedures can be created dynamically, stored in data structures, returned as results of procedures, and so on. Other languages with these properties include Common Lisp, Haskell, JavaScript, ML, and Perl.

One distinguishing feature of Scheme is that continuations, which in most other languages only operate behind the scenes, also have "first-class" status. Continuations are useful for implementing a wide variety of advanced control constructs, including non-local exits, backtracking, and coroutines. See section 6.4. Some versions of ML also provide first-class continuations.

Arguments to Scheme procedures are always passed by value, which means that the actual argument expressions are evaluated before the procedure gains control, regardless of whether the procedure needs the result of the evaluation. Other languages that always pass arguments by value include APL, C#, Common Lisp, Java, JavaScript, ML, Pascal, Python, Ruby, and Smalltalk. This is distinct from the lazy-evaluation semantics of Haskell, or the call-by-name semantics of Algol 60, where an argument expression is not evaluated unless its value is needed by the procedure.

Scheme's model of arithmetic is designed to remain as independent as possible of the particular ways in which numbers are represented within a computer. In Scheme, every integer is a rational number, every rational is a real, and every real is a complex number. Thus the distinction between integer and real arithmetic, so important to many programming languages, does not appear in Scheme. In its place is a distinction between exact arithmetic, which corresponds to the mathematical ideal, and inexact arithmetic on approximations. As in Common Lisp, exact arithmetic is not limited to integers.

1.2. Syntax

Scheme, like most dialects of Lisp, employs a fully parenthesized prefix notation for programs and (other) data; the grammar of Scheme generates a sublanguage of the language used for data. An important consequence of this simple, uniform representation is the susceptibility of Scheme programs and data to uniform treatment by other Scheme programs. For example, the `eval` procedure evaluates a Scheme program expressed as data.

The `read` procedure performs syntactic as well as lexical decomposition of the data it reads. The `read` procedure parses its input as data (section 7.1.2), not as program.

The formal syntax of Scheme is described in section 7.1.

1.3. Notation and terminology

1.3.1. Base and optional features

Every identifier defined in this report appears in one of several *modules*. Identifiers defined in the `base` module are not marked specially in the body of the report. A summary of all the standard modules and the features they provide is given in Appendix A.

Implementations must provide the `base` module and all the identifiers exported from it. Implementations are free to provide or omit the other modules given in this report,

but each module must either be provided in its entirety, exporting no additional identifiers, or else omitted altogether.

Implementations may provide other modules not described in this report. Implementations may also extend the function of any identifier in this report, provided the extensions are not in conflict with the language reported here. In particular, implementations must support portable code by providing a mode of operation in which the lexical syntax does not conflict with the lexical syntax described in this report.

1.3.2. Error situations and unspecified behavior

When speaking of an error situation, this report uses the phrase "an error is signalled" to indicate that implementations must detect and report the error. If such wording does not appear in the discussion of an error, then implementations are not required to detect or report the error, though they are encouraged to do so. An error situation that implementations are not required to detect is usually referred to simply as "an error."

An error is signalled by raising a non-continuable exception, as if by the procedure `raise`. The object raised is implementation-dependent and need not be a fresh object every time.

For example, it is an error for a procedure to be passed an argument that the procedure is not explicitly specified to handle, even though such domain errors are seldom mentioned in this report. Implementations may extend a procedure's domain of definition to include such arguments.

This report uses the phrase "may report a violation of an implementation restriction" to indicate circumstances under which an implementation is permitted to report that it is unable to continue execution of a correct program because of some restriction imposed by the implementation. Implementation restrictions are of course discouraged, but implementations are encouraged to report violations of implementation restrictions.

For example, an implementation may report a violation of an implementation restriction if it does not have enough storage to run a program.

If the value of an expression is said to be "unspecified," then the expression must evaluate to some object without signalling an error, but the value depends on the implementation; this report explicitly does not say what value should be returned.

In addition to errors signalled by situations described in this report, programmers may signal their own errors and handle signalled errors as described in section 6.5.

1.3.3. Entry format

Chapters 4 and 6 are organized into entries. Each entry describes one language feature or a group of related features, where a feature is either a syntactic construct or a built-in procedure. An entry begins with one or more header lines of the form

template *category*

for identifiers in the `base` module, or

template *module category*

where *module* is the short name of a module as defined in Appendix A.

If *category* is "syntax," the entry describes an expression type, and the template gives the syntax of the expression type. Components of expressions are designated by syntactic variables, which are written using angle brackets, for example, ⟨expression⟩, ⟨variable⟩. Syntactic variables should be understood to denote segments of program text; for example, ⟨expression⟩ stands for any string of characters which is a syntactically valid expression. The notation

⟨thing$_1$⟩ ...

indicates zero or more occurrences of a ⟨thing⟩, and

⟨thing$_1$⟩ ⟨thing$_2$⟩ ...

indicates one or more occurrences of a ⟨thing⟩.

If *category* is "auxiliary syntax," then the entry describes a syntax binding that may occur only as part of specific surrounding expressions. Any use as an independent syntactic construct or identifier is an error.

If *category* is "procedure," then the entry describes a procedure, and the header line gives a template for a call to the procedure. Argument names in the template are *italicized*. Thus the header line

`(vector-ref` *vector k*`)` procedure

indicates that the built-in procedure `vector-ref` takes two arguments, a vector *vector* and an exact non-negative integer k (see below). The header lines

`(make-vector` k`)` procedure
`(make-vector` k *fill*`)` procedure

indicate that the `make-vector` procedure must be defined to take either one or two arguments.

It is an error for an operation to be presented with an argument that it is not specified to handle. For succinctness, we follow the convention that if an argument name is also the name of a type listed in section 3.2, then that argument must be of the named type. For example, the header line for `vector-ref` given above dictates that the first argument to `vector-ref` must be a vector. The following

naming conventions also imply type restrictions:

obj	any object
list, $list_1$, ... $list_j$, ...	list (see section 6.3.2)
z, z_1, ... z_j, ...	complex number
x, x_1, ... x_j, ...	real number
y, y_1, ... y_j, ...	real number
q, q_1, ... q_j, ...	rational number
n, n_1, ... n_j, ...	integer
k, k_1, ... k_j, ...	exact non-negative integer

1.3.4. Evaluation examples

The symbol "$\Longrightarrow$" used in program examples should be read "evaluates to." For example,

```
(* 5 8)                    ⟹  40
```

means that the expression `(* 5 8)` evaluates to the object `40`. Or, more precisely: the expression given by the sequence of characters "`(* 5 8)`" evaluates, in the initial environment, to an object that may be represented externally by the sequence of characters "`40`." See section 3.3 for a discussion of external representations of objects.

1.3.5. Naming conventions

By convention, the names of procedures that always return a boolean value usually end in "`?`." Such procedures are called predicates.

By convention, the names of procedures that store values into previously allocated locations (see section 3.4) usually end in "`!`." Such procedures are called mutation procedures. By convention, the value returned by a mutation procedure is unspecified.

By convention, "`->`" appears within the names of procedures that take an object of one type and return an analogous object of another type. For example, `list->vector` takes a list and returns a vector whose elements are the same as those of the list.

2. Lexical conventions

This section gives an informal account of some of the lexical conventions used in writing Scheme programs. For a formal syntax of Scheme, see section 7.1.

2.1. Identifiers

Most identifiers allowed by other programming languages are also acceptable to Scheme. The precise rules for forming identifiers vary among implementations of Scheme, but in all implementations, a sequence of letters, digits, and "extended alphabetic characters" that does not have a prefix which is a valid number is an identifier. However, the `.` token used in the list syntax is not an identifier. Here are some examples of identifiers:

```
lambda                  q
list->vector            +soup+
+                       V17a
<=?                     a34kTMNs
->string                ...
the-word-recursion-has-many-meanings
```

Extended alphabetic characters may be used within identifiers as if they were letters. The following are always permitted as alphabetic characters (but implementations may allow a variety of other characters in identifiers):

```
! $ % & * + - . / : < = > ? @ ^ _ ~
```

In addition, any character can be used within an identifier when specified via an ⟨inline hex escape⟩. For example, the identifier `H\x65;llo` is the same as the identifier `Hello`, and in an implementation which has the appropriate Unicode character the identifier `\x3BB;` is the same as the identifier λ.

As a convenience, identifiers may also be written as a sequence of zero or more characters enclosed within vertical bars (|), analogous to string literals. Vertical bars and other characters can be included in the identifier with an ⟨inline hex escape⟩. Thus the identifier `|foo bar|` is the same as the identifier `foo\x20;bar`. Note that `||` is a valid identifier that is not equal to any other identifier.

See section 7.1.1 for a formal syntax of identifiers.

Identifiers have two uses within Scheme programs:

- Any identifier may be used as a variable or as a syntactic keyword (see sections 3.1 and 4.3).
- When an identifier appears as a literal or within a literal (see section 4.1.2), it is being used to denote a *symbol* (see section 6.3.3).

In contrast with earlier revisions of the report [2], the syntax of data distinguishes between upper and lower case in identifiers and in characters specified via their names.

Implementations may support case-insensitive syntax for backward compatibility or for other reasons. If they do so, they must support the following optional directives to control case folding.

```
#!fold-case
#!no-fold-case
```

These directives may appear anywhere comments may appear and are treated as comments, except that they affect the reading of subsequent tokens. The `#!fold-case`

directive causes the reader to case-fold (as if by `string-foldcase`) each ⟨identifier⟩ and ⟨character name⟩. The `#!no-fold-case` directive causes the reader to return to the default, non-folding behavior. The scope of these directives is all subsequent `read` operations for the port from which they are read, until another such directive is encountered. No other ports are affected.

2.2. Whitespace and comments

Whitespace characters are spaces and newlines. (Implementations typically provide additional whitespace characters such as tab or page break.) Whitespace is used for improved readability and as necessary to separate tokens from each other, a token being an indivisible lexical unit such as an identifier or number, but is otherwise insignificant. Whitespace may occur between any two tokens, but not within a token. Whitespace may also occur inside a string, where it is significant.

The lexical syntax includes several comment forms. Comments are treated exactly like whitespace.

A semicolon (`;`) indicates the start of a line comment. The comment continues to the end of the line on which the semicolon appears. Comments are invisible to Scheme, but the end of the line is visible as whitespace. This prevents a comment from appearing in the middle of an identifier or number.

Another way to indicate a comment is to prefix a ⟨datum⟩ (cf. section 7.1.2) with `#;`, possibly with ⟨whitespace⟩ before the ⟨datum⟩. The comment consists of the comment prefix `#;`, the space, and the ⟨datum⟩ together. This notation is useful for "commenting out" sections of code.

Block comments may be indicated with properly nested `#|` and `|#` pairs.

```
#|
   The FACT procedure computes the factorial
   of a non-negative integer.
|#
(define fact
  (lambda (n)
    (if (= n 0)
        #;(= n 1)
        1        ;Base case: return 1
        (* n (fact (- n 1))))))
```

2.3. Other notations

For a description of the notations used for numbers, see section 6.2.

`. + -` These are used in numbers, and may also occur anywhere in an identifier. A delimited plus or minus sign by itself is also an identifier. A delimited period (not occurring within a number or identifier) is used in the notation for pairs (section 6.3.2), and to indicate a rest-parameter in a formal parameter list (section 4.1.4). Note that a sequence of two or more periods *is* an identifier.

`( )` Parentheses are used for grouping and to notate lists (section 6.3.2).

`'` The single quote character is used to indicate literal data (section 4.1.2).

`` ` `` The backquote character is used to indicate partly-constant data (section 4.2.8).

`, ,@` The character comma and the sequence comma at-sign are used in conjunction with backquote (section 4.2.8).

`"` The double quote character is used to delimit strings (section 6.3.5).

`\` Backslash is used in the syntax for character constants (section 6.3.4) and as an escape character within string constants (section 6.3.5) and identifiers (section 7.1.1).

`[ ] { }` Left and right square brackets and curly braces are reserved for possible future extensions to the language.

`#` Sharp sign is used for a variety of purposes depending on the character that immediately follows it:

`#t #f` These are the boolean constants (section 6.3.1).

`#\` This introduces a character constant (section 6.3.4).

`#(` This introduces a vector constant (section 6.3.6). Vector constants are terminated by `)` .

`#u8(` This introduces a bytevector constant (section 6.3.7). Bytevector constants are terminated by `)` .

`#e #i #b #o #d #x` These are used in the notation for numbers (section 6.2.4).

`#`⟨n⟩`=` `#`⟨n⟩`#` These are used for labeling and referencing other literal data (section 4.2.10).

3. Basic concepts

3.1. Variables, syntactic keywords, and regions

An identifier may name a type of syntax, or it may name a location where a value can be stored. An identifier that names a type of syntax is called a *syntactic keyword* and is said to be *bound* to that syntax. An identifier that names a location is called a *variable* and is said to be *bound* to

that location. The set of all visible bindings in effect at some point in a program is known as the *environment* in effect at that point. The value stored in the location to which a variable is bound is called the variable's value. By abuse of terminology, the variable is sometimes said to name the value or to be bound to the value. This is not quite accurate, but confusion rarely results from this practice.

Certain expression types are used to create new kinds of syntax and to bind syntactic keywords to those new syntaxes, while other expression types create new locations and bind variables to those locations. These expression types are called *binding constructs*. Those that bind syntactic keywords are listed in section 4.3. The most fundamental of the variable binding constructs is the `lambda` expression, because all other variable binding constructs can be explained in terms of `lambda` expressions. The other variable binding constructs are `let`, `let*`, `letrec`, `letrec*`, `let-values`, `let*-values`, and `do` expressions (see sections 4.1.4, 4.2.2, and 4.2.4).

Like Algol 60, Common Lisp, Pascal, Python, Ruby, and Smalltalk, Scheme is a statically scoped language with block structure. To each place where an identifier is bound in a program there corresponds a *region* of the program text within which the binding is visible. The region is determined by the particular binding construct that establishes the binding; if the binding is established by a `lambda` expression, for example, then its region is the entire `lambda` expression. Every mention of an identifier refers to the binding of the identifier that established the innermost of the regions containing the use. If there is no binding of the identifier whose region contains the use, then the use refers to the binding for the variable in the top level environment, if any (chapters 4 and 6); if there is no binding for the identifier, it is said to be *unbound*.

3.2. Disjointness of types

No object satisfies more than one of the following predicates:

```
boolean?        pair?
symbol?         number?
char?           string?
vector?         bytevector?
port?           procedure?
```

These predicates define the types *boolean*, *pair*, *symbol*, *number*, *char* (or *character*), *string*, *vector*, *bytevector*, *port*, and *procedure*. The empty list is a special object of its own type; it satisfies none of the above predicates.

Although there is a separate boolean type, any Scheme value can be used as a boolean value for the purpose of a conditional test. As explained in section 6.3.1, all values count as true in such a test except for `#f`. This report uses the word "true" to refer to any Scheme value except `#f`, and the word "false" to refer to `#f`.

3.3. External representations

An important concept in Scheme (and Lisp) is that of the *external representation* of an object as a sequence of characters. For example, an external representation of the integer 28 is the sequence of characters "`28`," and an external representation of a list consisting of the integers 8 and 13 is the sequence of characters "`(8 13)`."

The external representation of an object is not necessarily unique. The integer 28 also has representations "`#e28.000`" and "`#x1c`," and the list in the previous paragraph also has the representations "`( 08 13 )`" and "`(8 . (13 . ()))`" (see section 6.3.2).

Many objects have standard external representations, but some, such as procedures, do not have standard representations (although particular implementations may define representations for them).

An external representation may be written in a program to obtain the corresponding object (see `quote`, section 4.1.2).

External representations can also be used for input and output. The procedure `read` (section 6.7.2) parses external representations, and the procedure `write` (section 6.7.3) generates them. Together, they provide an elegant and powerful input/output facility.

Note that the sequence of characters "`(+ 2 6)`" is *not* an external representation of the integer 8, even though it *is* an expression evaluating to the integer 8; rather, it is an external representation of a three-element list, the elements of which are the symbol `+` and the integers 2 and 6. Scheme's syntax has the property that any sequence of characters that is an expression is also the external representation of some object. This can lead to confusion, since it may not be obvious out of context whether a given sequence of characters is intended to denote data or program, but it is also a source of power, since it facilitates writing programs such as interpreters and compilers that treat programs as data (or vice versa).

The syntax of external representations of various kinds of objects accompanies the description of the primitives for manipulating the objects in the appropriate sections of chapter 6.

3.4. Storage model

Variables and objects such as pairs, vectors, and strings implicitly denote locations or sequences of locations. A string, for example, denotes as many locations as there

are characters in the string. (These locations need not correspond to a full machine word.) A new value may be stored into one of these locations using the `string-set!` procedure, but the string continues to denote the same locations as before.

An object fetched from a location, by a variable reference or by a procedure such as `car`, `vector-ref`, or `string-ref`, is equivalent in the sense of `eqv?` (section 6.1) to the object last stored in the location before the fetch.

Every location is marked to show whether it is in use. No variable or object ever refers to a location that is not in use. Whenever this report speaks of storage being allocated for a variable or object, what is meant is that an appropriate number of locations are chosen from the set of locations that are not in use, and the chosen locations are marked to indicate that they are now in use before the variable or object is made to denote them.

In many systems it is desirable for constants (i.e. the values of literal expressions) to reside in read-only-memory. To express this, it is convenient to imagine that every object that denotes locations is associated with a flag telling whether that object is mutable or immutable. In such systems literal constants and the strings returned by `symbol->string` are immutable objects, while all objects created by the other procedures listed in this report are mutable. It is an error to attempt to store a new value into a location that is denoted by an immutable object.

3.5. Proper tail recursion

Implementations of Scheme are required to be *properly tail-recursive*. Procedure calls that occur in certain syntactic contexts defined below are *tail calls*. A Scheme implementation is properly tail-recursive if it supports an unbounded number of active tail calls. A call is *active* if the called procedure may still return. Note that this includes calls that may be returned from either by the current continuation or by continuations captured earlier by `call-with-current-continuation` that are later invoked. In the absence of captured continuations, calls could return at most once and the active calls would be those that had not yet returned. A formal definition of proper tail recursion can be found in [10].

Rationale:

Intuitively, no space is needed for an active tail call because the continuation that is used in the tail call has the same semantics as the continuation passed to the procedure containing the call. Although an improper implementation might use a new continuation in the call, a return to this new continuation would be followed immediately by a return to the continuation passed to the procedure. A properly tail-recursive implementation returns to that continuation directly.

Proper tail recursion was one of the central ideas in Steele and Sussman's original version of Scheme. Their first Scheme interpreter implemented both functions and actors. Control flow was expressed using actors, which differed from functions in that they passed their results on to another actor instead of returning to a caller. In the terminology of this section, each actor finished with a tail call to another actor.

Steele and Sussman later observed that in their interpreter the code for dealing with actors was identical to that for functions and thus there was no need to include both in the language.

A *tail call* is a procedure call that occurs in a *tail context*. Tail contexts are defined inductively. Note that a tail context is always determined with respect to a particular lambda expression.

- The last expression within the body of a lambda expression, shown as ⟨tail expression⟩ below, occurs in a tail context. The same is true of the bodies of a `case-lambda` expression.

  `(lambda` ⟨formals⟩
  ⟨definition⟩* ⟨expression⟩* ⟨tail expression⟩`)`

- If one of the following expressions is in a tail context, then the subexpressions shown as ⟨tail expression⟩ are in a tail context. These were derived from rules in the grammar given in chapter 7 by replacing some occurrences of ⟨expression⟩ with ⟨tail expression⟩. Only those rules that contain tail contexts are shown here.

  `(if` ⟨expression⟩ ⟨tail expression⟩ ⟨tail expression⟩`)`
  `(if` ⟨expression⟩ ⟨tail expression⟩`)`

  `(cond` ⟨cond clause⟩$^+$`)`
  `(cond` ⟨cond clause⟩* `(else` ⟨tail sequence⟩`))`

  `(case` ⟨expression⟩
  ⟨case clause⟩$^+$`)`
  `(case` ⟨expression⟩
  ⟨case clause⟩*
  `(else` ⟨tail sequence⟩`))`

  `(and` ⟨expression⟩* ⟨tail expression⟩`)`
  `(or` ⟨expression⟩* ⟨tail expression⟩`)`

  `(when` ⟨test⟩ ⟨tail sequence⟩`)`
  `(unless` ⟨test⟩ ⟨tail sequence⟩`)`

  `(let (`⟨binding spec⟩*`)` ⟨tail body⟩`)`
  `(let` ⟨variable⟩ `(`⟨binding spec⟩*`)` ⟨tail body⟩`)`
  `(let* (`⟨binding spec⟩*`)` ⟨tail body⟩`)`
  `(letrec (`⟨binding spec⟩*`)` ⟨tail body⟩`)`
  `(letrec* (`⟨binding spec⟩*`)` ⟨tail body⟩`)`
  `(let-values (`⟨formals⟩*`)` ⟨tail body⟩`)`

`(let*-values (`⟨formals⟩`*) `⟨tail body⟩`)`

`(let-syntax (`⟨syntax spec⟩`*) `⟨tail body⟩`)`
`(letrec-syntax (`⟨syntax spec⟩`*) `⟨tail body⟩`)`

`(begin `⟨tail sequence⟩`)`

```
(do (⟨iteration spec⟩*)
    (⟨test⟩ ⟨tail sequence⟩))
  ⟨expression⟩*)
```

where

⟨cond clause⟩ ⟶ `(`⟨test⟩ ⟨tail sequence⟩`)`
⟨case clause⟩ ⟶ `((`⟨datum⟩`*)` ⟨tail sequence⟩`)`

⟨tail body⟩ ⟶ ⟨definition⟩* ⟨tail sequence⟩
⟨tail sequence⟩ ⟶ ⟨expression⟩* ⟨tail expression⟩

- If a `cond` expression is in a tail context, and has a clause of the form `(`⟨expression$_1$⟩ `=>` ⟨expression$_2$⟩`)` then the (implied) call to the procedure that results from the evaluation of ⟨expression$_2$⟩ is in a tail context. ⟨expression$_2$⟩ itself is not in a tail context.

Certain built-in procedures are also required to perform tail calls. The first argument passed to `apply` and to `call-with-current-continuation`, and the second argument passed to `call-with-values`, must be called via a tail call. Similarly, `eval` must evaluate its first argument as if it were in tail position within the `eval` procedure.

In the following example the only tail call is the call to `f`. None of the calls to `g` or `h` are tail calls. The reference to `x` is in a tail context, but it is not a call and thus is not a tail call.

```
(lambda ()
  (if (g)
      (let ((x (h)))
        x)
      (and (g) (f))))
```

Note: Implementations are allowed, but not required, to recognize that some non-tail calls, such as the call to `h` above, can be evaluated as though they were tail calls. In the example above, the `let` expression could be compiled as a tail call to `h`. (The possibility of `h` returning an unexpected number of values can be ignored, because in that case the effect of the `let` is explicitly unspecified and implementation-dependent.)

4. Expressions

Expression types are categorized as *primitive* or *derived*. Primitive expression types include variables and procedure calls. Derived expression types are not semantically primitive, but can instead be defined as macros. Suitable definitions of some of the derived expressions are given in section 7.3.

4.1. Primitive expression types

4.1.1. Variable references

⟨variable⟩ syntax

An expression consisting of a variable (section 3.1) is a variable reference. The value of the variable reference is the value stored in the location to which the variable is bound. It is an error to reference an unbound variable.

```
(define x 28)
x                       ⟹  28
```

4.1.2. Literal expressions

`(quote `⟨datum⟩`)` syntax
`'`⟨datum⟩ syntax
⟨constant⟩ syntax

`(quote `⟨datum⟩`)` evaluates to ⟨datum⟩. ⟨Datum⟩ may be any external representation of a Scheme object (see section 3.3). This notation is used to include literal constants in Scheme code.

```
(quote a)               ⟹  a
(quote #(a b c))        ⟹  #(a b c)
(quote (+ 1 2))         ⟹  (+ 1 2)
```

`(quote `⟨datum⟩`)` may be abbreviated as `'`⟨datum⟩. The two notations are equivalent in all respects.

```
'a                      ⟹  a
'#(a b c)               ⟹  #(a b c)
'()                     ⟹  ()
'(+ 1 2)                ⟹  (+ 1 2)
'(quote a)              ⟹  (quote a)
''a                     ⟹  (quote a)
```

Numerical constants, string constants, character constants, and boolean constants evaluate "to themselves"; they need not be quoted.

```
'"abc"                  ⟹  "abc"
"abc"                   ⟹  "abc"
'145932                 ⟹  145932
145932                  ⟹  145932
'#t                     ⟹  #t
#t                      ⟹  #t
```

As noted in section 3.4, it is an error to alter a constant (i.e. the value of a literal expression) using a mutation procedure like `set-car!` or `string-set!`.

4.1.3. Procedure calls

(⟨operator⟩ ⟨operand$_1$⟩ ...) syntax

A procedure call is written by simply enclosing in parentheses expressions for the procedure to be called and the arguments to be passed to it. The operator and operand expressions are evaluated (in an unspecified order) and the resulting procedure is passed the resulting arguments.

```
(+ 3 4)                    ⟹  7
((if #f + *) 3 4)          ⟹  12
```

A number of procedures are available as the values of variables in the initial environment; for example, the addition and multiplication procedures in the above examples are the values of the variables `+` and `*`. New procedures are created by evaluating `lambda` expressions (see section 4.1.4).

Procedure calls may return any number of values (see `values` in section 6.4). Most of the procedures defined in this report return one value or, for procedures such as `apply`, pass on the values returned by a call to one of their arguments. Exceptions are noted in the individual descriptions.

Procedure calls are also called *combinations*.

Note: In contrast to other dialects of Lisp, the order of evaluation is unspecified, and the operator expression and the operand expressions are always evaluated with the same evaluation rules.

Note: Although the order of evaluation is otherwise unspecified, the effect of any concurrent evaluation of the operator and operand expressions is constrained to be consistent with some sequential order of evaluation. The order of evaluation may be chosen differently for each procedure call.

Note: In many dialects of Lisp, the empty combination, `()`, is a legitimate expression. In Scheme, combinations must have at least one subexpression, so `()` is not a syntactically valid expression.

4.1.4. Procedures

(`lambda` ⟨formals⟩ ⟨body⟩) syntax

Syntax: ⟨Formals⟩ should be a formal arguments list as described below, and ⟨body⟩ should be a sequence of one or more expressions.

Semantics: A `lambda` expression evaluates to a procedure. The environment in effect when the `lambda` expression was evaluated is remembered as part of the procedure. When the procedure is later called with some actual arguments, the environment in which the `lambda` expression was evaluated will be extended by binding the variables in the formal argument list to fresh locations, the corresponding actual argument values will be stored in those locations, and the expressions in the body of the `lambda` expression will be evaluated sequentially in the extended environment. The result(s) of the last expression in the body will be returned as the result(s) of the procedure call.

```
(lambda (x) (+ x x))         ⟹  a procedure
((lambda (x) (+ x x)) 4)     ⟹  8

(define reverse-subtract
  (lambda (x y) (- y x)))
(reverse-subtract 7 10)      ⟹  3

(define add4
  (let ((x 4))
    (lambda (y) (+ x y))))
(add4 6)                     ⟹  10
```

⟨Formals⟩ should have one of the following forms:

- (⟨variable$_1$⟩ ...): The procedure takes a fixed number of arguments; when the procedure is called, the arguments will be stored in the bindings of the corresponding variables.
- ⟨variable⟩: The procedure takes any number of arguments; when the procedure is called, the sequence of actual arguments is converted into a newly allocated list, and the list is stored in the binding of the ⟨variable⟩.
- (⟨variable$_1$⟩ ... ⟨variable$_n$⟩ `.` ⟨variable$_{n+1}$⟩): If a space-delimited period precedes the last variable, then the procedure takes n or more arguments, where n is the number of formal arguments before the period (there must be at least one). The value stored in the binding of the last variable will be a newly allocated list of the actual arguments left over after all the other actual arguments have been matched up against the other formal arguments.

It is an error for a ⟨variable⟩ to appear more than once in ⟨formals⟩.

```
((lambda x x) 3 4 5 6)       ⟹  (3 4 5 6)
((lambda (x y . z) z)
 3 4 5 6)                    ⟹  (5 6)
```

4.1.5. Conditionals

(`if` ⟨test⟩ ⟨consequent⟩ ⟨alternate⟩) syntax
(`if` ⟨test⟩ ⟨consequent⟩) syntax

Syntax: ⟨Test⟩, ⟨consequent⟩, and ⟨alternate⟩ may be arbitrary expressions.

Semantics: An `if` expression is evaluated as follows: first, ⟨test⟩ is evaluated. If it yields a true value (see section 6.3.1), then ⟨consequent⟩ is evaluated and its value(s) is(are) returned. Otherwise ⟨alternate⟩ is evaluated and its value(s) is(are) returned. If ⟨test⟩ yields `#f` and no ⟨alternate⟩ is specified, then the result of the expression is unspecified.

```
(if (> 3 2) 'yes 'no)       ⟹  yes
(if (> 2 3) 'yes 'no)       ⟹  no
(if (> 3 2)
    (- 3 2)
    (+ 3 2))                ⟹  1
```

4.1.6. Assignments

`(set!` ⟨variable⟩ ⟨expression⟩`)` syntax

⟨Expression⟩ is evaluated, and the resulting value is stored in the location to which ⟨variable⟩ is bound. ⟨Variable⟩ must be bound either in some region enclosing the `set!` expression or at top level. The result of the `set!` expression is unspecified.

```
(define x 2)
(+ x 1)                     ⟹  3
(set! x 4)                  ⟹  unspecified
(+ x 1)                     ⟹  5
```

4.2. Derived expression types

The constructs in this section are hygienic, as discussed in section 4.3. For reference purposes, section 7.3 gives macro definitions that will convert most of the constructs described in this section into the primitive constructs described in the previous section.

4.2.1. Conditionals

`(cond` ⟨clause$_1$⟩ ⟨clause$_2$⟩ ...`)` syntax
`else` auxiliary syntax
`=>` auxiliary syntax

Syntax: Each ⟨clause⟩ should be of the form

`(`⟨test⟩ ⟨expression$_1$⟩ ...`)`

where ⟨test⟩ is any expression. Alternatively, a ⟨clause⟩ may be of the form

`(`⟨test⟩ `=>` ⟨expression⟩`)`

The last ⟨clause⟩ may be an "else clause," which has the form

`(else` ⟨expression$_1$⟩ ⟨expression$_2$⟩ ...`)`.

Semantics: A `cond` expression is evaluated by evaluating the ⟨test⟩ expressions of successive ⟨clause⟩s in order until one of them evaluates to a true value (see section 6.3.1). When a ⟨test⟩ evaluates to a true value, then the remaining ⟨expression⟩s in its ⟨clause⟩ are evaluated in order, and the result(s) of the last ⟨expression⟩ in the ⟨clause⟩ is(are) returned as the result(s) of the entire `cond` expression.

If the selected ⟨clause⟩ contains only the ⟨test⟩ and no ⟨expression⟩s, then the value of the ⟨test⟩ is returned as the result. If the selected ⟨clause⟩ uses the `=>` alternate form, then the ⟨expression⟩ is evaluated. Its value must be a procedure that accepts one argument; this procedure is then called on the value of the ⟨test⟩ and the value(s) returned by this procedure is(are) returned by the `cond` expression.

If all ⟨test⟩s evaluate to `#f`, and there is no else clause, then the result of the conditional expression is unspecified; if there is an else clause, then its ⟨expression⟩s are evaluated, and the value(s) of the last one is(are) returned.

```
(cond ((> 3 2) 'greater)
      ((< 3 2) 'less))      ⟹  greater
(cond ((> 3 3) 'greater)
      ((< 3 3) 'less)
      (else 'equal))        ⟹  equal
(cond ((assv 'b '((a 1) (b 2))) => cadr)
      (else #f))            ⟹  2
```

`(case` ⟨key⟩ ⟨clause$_1$⟩ ⟨clause$_2$⟩ ...`)` syntax

Syntax: ⟨Key⟩ may be any expression. Each ⟨clause⟩ should have the form

`((`⟨datum$_1$⟩ ...`)` ⟨expression$_1$⟩ ⟨expression$_2$⟩ ...`)`,

where each ⟨datum⟩ is an external representation of some object. All the ⟨datum⟩s must be distinct. Alternatively, a ⟨clause⟩ may be of the form

`((`⟨datum$_1$⟩ ...`) =>` ⟨expression⟩`)`

The last ⟨clause⟩ may be an "else clause," which has one of the forms

`(else` ⟨expression$_1$⟩ ⟨expression$_2$⟩ ...`)`

or

`(else =>` ⟨expression⟩`)`.

Semantics: A `case` expression is evaluated as follows. ⟨Key⟩ is evaluated and its result is compared against each ⟨datum⟩. If the result of evaluating ⟨key⟩ is equivalent (in the sense of `eqv?`; see section 6.1) to a ⟨datum⟩, then the expressions in the corresponding ⟨clause⟩ are evaluated from left to right and the result(s) of the last expression in the ⟨clause⟩ is(are) returned as the result(s) of the `case` expression.

If the result of evaluating ⟨key⟩ is different from every ⟨datum⟩, then if there is an else clause its expressions are evaluated and the result(s) of the last is(are) the result(s) of the `case` expression; otherwise the result of the `case` expression is unspecified.

If the selected ⟨clause⟩ or else clause uses the `=>` alternate form, then the ⟨expression⟩ is evaluated. Its value must be a procedure that accepts one argument; this procedure is then called on the value of the ⟨key⟩ and the value(s) returned by this procedure is(are) returned by the `case` expression.

```
(case (* 2 3)
  ((2 3 5 7) 'prime)
  ((1 4 6 8 9) 'composite)) ⟹  composite
(case (car '(c d))
  ((a) 'a)
  ((b) 'b))                 ⟹  unspecified
(case (car '(c d))
  ((a e i o u) 'vowel)
  ((w y) 'semivowel)
  (else => (lambda (x) x))) ⟹  c
```

`(and` ⟨test$_1$⟩ ...`)` syntax

The ⟨test⟩ expressions are evaluated from left to right, and if any expression evaluates to `#f` (see section 6.3.1), `#f` is returned. Any remaining expressions are not evaluated. If all the expressions evaluate to true values, the value of the last expression is returned. If there are no expressions then `#t` is returned.

```
(and (= 2 2) (> 2 1))       ⟹  #t
(and (= 2 2) (< 2 1))       ⟹  #f
(and 1 2 'c '(f g))         ⟹  (f g)
(and)                       ⟹  #t
```

`(or` ⟨test$_1$⟩ ...`)` syntax

The ⟨test⟩ expressions are evaluated from left to right, and the value of the first expression that evaluates to a true value (see section 6.3.1) is returned. Any remaining expressions are not evaluated. If all expressions evaluate to `#f` or if there are no expressions, `#f` is returned.

```
(or (= 2 2) (> 2 1))        ⟹  #t
(or (= 2 2) (< 2 1))        ⟹  #t
(or #f #f #f)               ⟹  #f
(or (memq 'b '(a b c))
    (/ 3 0))                ⟹  (b c)
```

`(when` ⟨test⟩ ⟨expression$_1$⟩ ⟨expression$_2$⟩ ...`)` syntax

The ⟨test⟩ expression is evaluated, and if it is a true value the expressions are evaluated in order. The result of the `when` expression is unspecified.

`(unless` ⟨test⟩ ⟨expression$_1$⟩ ⟨expression$_2$⟩ ...`)` syntax

The ⟨test⟩ expression is evaluated, and if it is `#f` the expressions are evaluated in order. The result of the `unless` expression is unspecified.

4.2.2. Binding constructs

The binding constructs `let`, `let*`, `letrec`, `letrec*`, `let-values`, and `let*-values` give Scheme a block structure, like Algol 60. The syntax of the first four constructs is identical, but they differ in the regions they establish for their variable bindings. In a `let` expression, the initial values are computed before any of the variables become bound; in a `let*` expression, the bindings and evaluations are performed sequentially; while in `letrec` and `letrec*` expressions, all the bindings are in effect while their initial values are being computed, thus allowing mutually recursive definitions. `Let-values` and `let*-values` are analogous to `let` and `let*` respectively, but are designed to handle multiple-valued expressions, binding different identifiers to each returned value.

`(let` ⟨bindings⟩ ⟨body⟩`)` syntax

Syntax: ⟨Bindings⟩ should have the form

`((`⟨variable$_1$⟩ ⟨init$_1$⟩`) ...)`,

where each ⟨init⟩ is an expression, and ⟨body⟩ should be a sequence of zero or more definitions followed by a sequence of one or more expressions. It is an error for a ⟨variable⟩ to appear more than once in the list of variables being bound.

Semantics: The ⟨init⟩s are evaluated in the current environment (in some unspecified order), the ⟨variable⟩s are bound to fresh locations holding the results, the ⟨body⟩ is evaluated in the extended environment, and the value(s) of the last expression of ⟨body⟩ is(are) returned. Each binding of a ⟨variable⟩ has ⟨body⟩ as its region.

```
(let ((x 2) (y 3))
  (* x y))                  ⟹  6

(let ((x 2) (y 3))
  (let ((x 7)
        (z (+ x y)))
    (* z x)))               ⟹  35
```

See also named `let`, section 4.2.4.

`(let*` ⟨bindings⟩ ⟨body⟩`)` syntax

Syntax: ⟨Bindings⟩ should have the form

`((`⟨variable$_1$⟩ ⟨init$_1$⟩`) ...)`,

and ⟨body⟩ should be a sequence of sequence of zero or more definitions followed by a one or more expressions.

Semantics: `Let*` is similar to `let`, but the bindings are performed sequentially from left to right, and the region of a binding indicated by `(`⟨variable⟩ ⟨init⟩`)` is that part of the `let*` expression to the right of the binding. Thus the second binding is done in an environment in which the first binding is visible, and so on. The ⟨variable⟩s need not be distinct.

```
(let ((x 2) (y 3))
  (let* ((x 7)
         (z (+ x y)))
    (* z x)))                    ⟹  70
```

`(letrec` ⟨bindings⟩ ⟨body⟩`)` syntax

Syntax: ⟨Bindings⟩ should have the form

`(`(⟨variable$_1$⟩ ⟨init$_1$⟩) ...`)`,

and ⟨body⟩ should be a sequence of sequence of zero or more definitions followed by a one or more expressions. It is an error for a ⟨variable⟩ to appear more than once in the list of variables being bound.

Semantics: The ⟨variable⟩s are bound to fresh locations holding undefined values, the ⟨init⟩s are evaluated in the resulting environment (in some unspecified order), each ⟨variable⟩ is assigned to the result of the corresponding ⟨init⟩, the ⟨body⟩ is evaluated in the resulting environment, and the value(s) of the last expression in ⟨body⟩ is(are) returned. Each binding of a ⟨variable⟩ has the entire `letrec` expression as its region, making it possible to define mutually recursive procedures.

```
(letrec ((even?
          (lambda (n)
            (if (zero? n)
                #t
                (odd? (- n 1)))))
         (odd?
          (lambda (n)
            (if (zero? n)
                #f
                (even? (- n 1))))))
  (even? 88))
                                 ⟹  #t
```

One restriction on `letrec` is very important: it must be possible to evaluate each ⟨init⟩ without assigning or referring to the value of any ⟨variable⟩. If this restriction is violated, then it is an error. The restriction is necessary because Scheme passes arguments by value rather than by name. In the most common uses of `letrec`, all the ⟨init⟩s are `lambda` expressions and the restriction is satisfied automatically. Another restriction is that the continuation of each ⟨init⟩ should not be invoked more than once.

It must be possible to evaluate each ⟨init⟩ without assigning or referring to the value of the corresponding ⟨variable⟩ or the ⟨variable⟩ of any of the bindings that follow it in ⟨bindings⟩.

`(letrec*` ⟨bindings⟩ ⟨body⟩`)` syntax

Syntax: ⟨Bindings⟩ should have the form

`(`(⟨variable$_1$⟩ ⟨init$_1$⟩) ...`)`,

and ⟨body⟩ should be a sequence of sequence of zero or more definitions followed by a one or more expressions. It is an error for a ⟨variable⟩ to appear more than once in the list of variables being bound.

Semantics: The ⟨variable⟩s are bound to fresh locations, each ⟨variable⟩ is assigned in left-to-right order to the result of evaluating the corresponding ⟨init⟩, the ⟨body⟩ is evaluated in the resulting environment, and the values of the last expression in ⟨body⟩ are returned. Despite the left-to-right evaluation and assignment order, each binding of a ⟨variable⟩ has the entire `letrec*` expression as its region, making it possible to define mutually recursive procedures.

`(let-values` ⟨mvbindings⟩ ⟨body⟩`)` syntax

Syntax: ⟨Mvbindings⟩ should have the form

`(((`⟨variable$_1$⟩ ⟨variable$_2$⟩ ...`)` ⟨init⟩`)` ...`)`,

where each ⟨init⟩ is an expression, and ⟨body⟩ should be a sequence of zero or more definitions followed by a sequence of one or more expressions. It is an error for a ⟨variable⟩ to appear more than once in the list of variables being bound.

Semantics: The ⟨init⟩s are evaluated in the current environment (in some unspecified order) as if by invoking `call-with-values`, the ⟨variable⟩s are bound to fresh locations holding the values returned by the ⟨init⟩s, the ⟨body⟩ is evaluated in the extended environment, and the value(s) of the last expression of ⟨body⟩ is(are) returned. Each binding of a ⟨variable⟩ has ⟨body⟩ as its region.

It is an error if an ⟨init⟩ returns more or fewer values than the number of ⟨variable⟩s associated with it.

```
(let-values (((root rem) (exact-integer-sqrt 32)))
  (* root rem))                  ⟹  35
```

`(let*-values` ⟨mvbindings⟩ ⟨body⟩`)` syntax

Syntax: ⟨Mvbindings⟩ should have the form

`(((`⟨variable$_1$⟩ ⟨variable$_2$⟩ ...`)` ⟨init⟩`)` ...`)`,

and ⟨body⟩ should be a sequence of sequence of zero or more definitions followed by a one or more expressions.

Semantics: `Let-values*` is similar to `let-values`, but the bindings are performed sequentially from left to right, and the region of a binding indicated by `((`⟨variable⟩ ...`)` ⟨init⟩`))` is that part of the `let*-values` expression to the right of the binding. Thus the second set of bindings are done in an environment in which the first set of bindings is visible, and so on.

⟨Variable⟩s in one ⟨binding⟩ may be the same as those in other ⟨binding⟩s, though it is an error for two ⟨variable⟩s in the same ⟨binding⟩ to be the same.
In all six constructs, the continuation of each expression used to compute initial values must not be invoked more than once.

```
(letrec* ((p
           (lambda (x)
             (+ 1 (q (- x 1)))))
          (q
           (lambda (y)
             (if (zero? y)
                 0
                 (+ 1 (p (- y 1))))))
          (x (p 5))
          (y x))
  y)
                              ⟹  5
```

4.2.3. Sequencing

(**begin** ⟨form⟩ ...) syntax
(**begin** ⟨expression$_1$⟩ ⟨expression$_2$⟩ ...) syntax

The **begin** keyword has two different roles, depending on its context:

- It may appear as a form in a ⟨body⟩, or at the ⟨top-level⟩, or directly nested in a **begin** form that is one of these three types. In this case, the **begin** form must have the shape specified in the first header line. This use of begin acts as a splicing form - the forms inside the ⟨body⟩ are spliced into the surrounding body, as if the begin wrapper were not actually present.
- It may appear as an ordinary expression and must have the shape specified in the second header line. In this case, the ⟨expression⟩s are evaluated sequentially from left to right, and the value(s) of the last ⟨expression⟩ is(are) returned. This expression type is used to sequence side effects such as assignments or input and output.

```
(define x 0)

(begin (set! x 5)
       (+ x 1))               ⟹  6

(begin (display "4 plus 1 equals ")
       (display (+ 4 1)))     ⟹  unspecified
           and prints  4 plus 1 equals 5
```

4.2.4. Iteration

(**do** ((⟨variable$_1$⟩ ⟨init$_1$⟩ ⟨step$_1$⟩) syntax
  ...)
  (⟨test⟩ ⟨expression⟩ ...)
 ⟨command⟩ ...)

Do is an iteration construct. It specifies a set of variables to be bound, how they are to be initialized at the start, and how they are to be updated on each iteration. When a termination condition is met, the loop exits after evaluating the ⟨expression⟩s.

Do expressions are evaluated as follows: The ⟨init⟩ expressions are evaluated (in some unspecified order), the ⟨variable⟩s are bound to fresh locations, the results of the ⟨init⟩ expressions are stored in the bindings of the ⟨variable⟩s, and then the iteration phase begins.

Each iteration begins by evaluating ⟨test⟩; if the result is false (see section 6.3.1), then the ⟨command⟩ expressions are evaluated in order for effect, the ⟨step⟩ expressions are evaluated in some unspecified order, the ⟨variable⟩s are bound to fresh locations, the results of the ⟨step⟩s are stored in the bindings of the ⟨variable⟩s, and the next iteration begins.

If ⟨test⟩ evaluates to a true value, then the ⟨expression⟩s are evaluated from left to right and the value(s) of the last ⟨expression⟩ is(are) returned. If no ⟨expression⟩s are present, then the value of the **do** expression is unspecified.

The region of the binding of a ⟨variable⟩ consists of the entire **do** expression except for the ⟨init⟩s. It is an error for a ⟨variable⟩ to appear more than once in the list of **do** variables.

A ⟨step⟩ may be omitted, in which case the effect is the same as if (⟨variable⟩ ⟨init⟩ ⟨variable⟩) had been written instead of (⟨variable⟩ ⟨init⟩).

```
(do ((vec (make-vector 5))
     (i 0 (+ i 1)))
    ((= i 5) vec)
  (vector-set! vec i i))    ⟹  #(0 1 2 3 4)

(let ((x '(1 3 5 7 9)))
  (do ((x x (cdr x))
       (sum 0 (+ sum (car x))))
      ((null? x) sum)))     ⟹  25
```

(**let** ⟨variable⟩ ⟨bindings⟩ ⟨body⟩) syntax

"Named **let**" is a variant on the syntax of **let** which provides a more general looping construct than **do** and may also be used to express recursions. It has the same syntax and semantics as ordinary **let** except that ⟨variable⟩ is bound within ⟨body⟩ to a procedure whose formal arguments are the bound variables and whose body is ⟨body⟩. Thus the execution of ⟨body⟩ may be repeated by invoking the procedure named by ⟨variable⟩.

```
(let loop ((numbers '(3 -2 1 6 -5))
           (nonneg '())
           (neg '()))
  (cond ((null? numbers) (list nonneg neg))
        ((>= (car numbers) 0)
         (loop (cdr numbers)
               (cons (car numbers) nonneg)
               neg))
        ((< (car numbers) 0)
```

```
(loop (cdr numbers)
      nonneg
      (cons (car numbers) neg)))))
 ⟹  ((6 1 3) (-5 -2))
```

4.2.5. Delayed evaluation

`(delay` ⟨expression⟩`)` lazy module syntax

The `delay` construct is used together with the procedure `force` to implement *lazy evaluation* or *call by need*. `(delay` ⟨expression⟩`)` returns an object called a *promise* which at some point in the future may be asked (by the `force` procedure) to evaluate ⟨expression⟩, and deliver the resulting value. The effect of ⟨expression⟩ returning multiple values is unspecified.

`(lazy` ⟨expression⟩`)` lazy module syntax

The `lazy` construct is similar to `delay`, but its argument must evaluate to a promise. The returned promise, when forced, will evaluate to whatever the original promise would evaluate to if it had been forced. The effect of ⟨expression⟩ returning multiple values is unspecified.

See the description of `force` (section 6.4) for a more complete description of `lazy` and `delay`.

4.2.6. Dynamic Bindings

`(parameterize ((param value) ...)` syntax
⟨body$_1$⟩

The value of the *param* expressions must be parameter objects. The `parameterize` form is used to change the values returned by parameter objects for the dynamic extent of the body. The expressions *param* and `(`*converter value*`)` are evaluated in an unspecified order. The ⟨expr⟩s are evaluated in order in a dynamic extent during which calls to the *param* parameter objects return the result of the corresponding `(`*converter value*`)`. The result(s) of the last ⟨expr⟩ is(are) returned as the result(s) of the entire `parameterize` form.

If an implementation supports multiple threads of execution, then `parameterize` must not change the associated values of any parameters in any thread created before or after the `parameterize` form.

See the description of `make-parameter` (section 6.4) for a more complete description of `parameterize`.

4.2.7. Exception Handling

`(guard (`⟨variable⟩ syntax
⟨cond clause$_1$⟩ ⟨cond clause$_2$⟩ ...`)`
⟨body⟩`)`

Syntax: Each ⟨cond clause⟩ is as in the specification of `cond`.

Semantics: Evaluating a `guard` form evaluates ⟨body⟩ with an exception handler that binds the raised object to ⟨variable⟩ and within the scope of that binding evaluates the clauses as if they were the clauses of a `cond` expression. That implicit `cond` expression is evaluated with the continuation and dynamic extent of the `guard` expression. If every ⟨cond clause⟩'s ⟨test⟩ evaluates to `#f` and there is no `else` clause, then `raise` is re-invoked on the raised object within the dynamic extent of the original call to `raise` except that the current exception handler is that of the `guard` expression.

The final expression in a ⟨cond⟩ clause is in a tail context if the `guard` expression itself is.

See section 6.5 for a more complete discussion of exceptions.

4.2.8. Quasiquotation

`(quasiquote` ⟨qq template⟩`)` syntax
`` ` ``⟨qq template⟩ syntax
`unquote` auxiliary syntax
`unquote-splicing` auxiliary syntax

"Backquote" or "quasiquote" expressions are useful for constructing a list or vector structure when most but not all of the desired structure is known in advance. If no commas appear within the ⟨qq template⟩, the result of evaluating `` ` ``⟨qq template⟩ is equivalent to the result of evaluating `'`⟨qq template⟩. If a comma appears within the ⟨qq template⟩, however, the expression following the comma is evaluated ("unquoted") and its result is inserted into the structure instead of the comma and the expression. If a comma appears followed immediately by an at-sign (`@`), then the following expression must evaluate to a list; the opening and closing parentheses of the list are then "stripped away" and the elements of the list are inserted in place of the comma at-sign expression sequence. A comma at-sign should only appear within a list or vector ⟨qq template⟩.

```
`(list ,(+ 1 2) 4)          ⟹  (list 3 4)
(let ((name 'a)) `(list ,name ',name))
          ⟹  (list a (quote a))
`(a ,(+ 1 2) ,@(map abs '(4 -5 6)) b)
          ⟹  (a 3 4 5 6 b)
`(( foo ,(- 10 3)) ,@(cdr '(c)) . ,(car '(cons)))
          ⟹  ((foo 7) . cons)
`#(10 5 ,(sqrt 4) ,@(map sqrt '(16 9)) 8)
          ⟹  #(10 5 2 4 3 8)
```

Quasiquote forms may be nested. Substitutions are made only for unquoted components appearing at the same nesting level as the outermost backquote. The nesting level increases by one inside each successive quasiquotation, and decreases by one inside each unquotation.

```
`(a `(b ,(+ 1 2) ,(foo ,(+ 1 3) d) e) f)
          ⟹  (a `(b ,(+ 1 2) ,(foo 4 d) e) f)
(let ((name1 'x)
      (name2 'y))
  `(a `(b ,,name1 ,',name2 d) e))
          ⟹  (a `(b ,x ,'y d) e)
```

A quasiquote expression may return either fresh, mutable objects or literal structure for any structure that is constructed at run time during the evaluation of the expression. Portions that do not need to be rebuilt are always literal. Thus,

```
(let ((a 3)) `((1 2) ,a ,4 ,'five 6))
```

may be equivalent to either of the following expressions:

```
`((1 2) 3 4 five 6)

(let ((a 3))
  (cons '(1 2)
        (cons a (cons 4 (cons 'five '(6))))))
```

However, it is not equivalent to this expression:

```
(let ((a 3)) (list (list 1 2) a 4 'five 6))
```

The two notations `⟨qq template⟩ and (quasiquote ⟨qq template⟩) are identical in all respects. ,⟨expression⟩ is identical to (unquote ⟨expression⟩), and ,@⟨expression⟩ is identical to (unquote-splicing ⟨expression⟩). The external syntax generated by write for two-element lists whose car is one of these symbols may vary between implementations.

```
(quasiquote (list (unquote (+ 1 2)) 4))
          ⟹  (list 3 4)
'(quasiquote (list (unquote (+ 1 2)) 4))
          ⟹  `(list ,(+ 1 2) 4)
     i.e., (quasiquote (list (unquote (+ 1 2)) 4))
```

It is a error if any of the identifiers quasiquote, unquote, or unquote-splicing appear in positions within a ⟨qq template⟩ otherwise than as described above.

4.2.9. Case-lambda

(case-lambda ⟨clause$_1$⟩ ⟨clause$_2$⟩ ...) syntax

Syntax: Each ⟨clause⟩ should be of the form (⟨formals⟩ ⟨body⟩)where ⟨formals⟩ and ⟨body⟩ have the same syntax as in a lambda expression.

Semantics: A case-lambda expression evaluates to a procedure that accepts a variable number of arguments and is lexically scoped in the same manner as procedures resulting from a lambda expression. When the procedure is called, then the first ⟨clause⟩ for which the arguments agree with ⟨formals⟩ is selected, where agreement is specified as for the ⟨formals⟩ of a lambda expression. The variables of ⟨formals⟩ are bound to fresh locations, the values of the arguments are stored in those locations, the ⟨body⟩ is evaluated in the extended environment, and the results of ⟨body⟩ are returned as the results of the procedure call.

It is an error for the arguments not to agree with the ⟨formals⟩ of any ⟨clause⟩.

```
(define plus
  (case-lambda
    (() 0)
    ((x) x)
    ((x y) (+ x y))
    ((x y z) (+ (+ x y) z))
    (args (apply + args))))

(plus)                    ⟹ 0
(plus 1)                  ⟹ 1
(plus 1 2 3)              ⟹ 6
```

4.2.10. Reader Labels

#⟨n⟩=⟨datum⟩ syntax
#⟨n⟩# syntax

⟨N⟩ must be an exact unsigned decimal integer. The syntax #⟨n⟩=⟨datum⟩ reads as ⟨datum⟩, except that within the syntax of ⟨datum⟩ the ⟨datum⟩ is labelled by ⟨n⟩.

The syntax #⟨n⟩# serves as a reference to some object labelled by #⟨n⟩=; the result is the same object as the #⟨n⟩= as compared with eqv? (see section 6.1). This permits notation of structures with shared or circular substructure.

```
(let ((x (list 'a 'b 'c)))
  (set-cdr! (cddr x) x)
  x)                      ⟹ #0=(a b c . #0#)
```

A reference #⟨n⟩# may occur only after a label #⟨n⟩=; forward references are not permitted. In addition, the reference may not appear as the labelled object itself (that is, one may not write #⟨n⟩= #⟨n⟩#), because the object labelled by #⟨n⟩= is not well defined in this case.

It is an error for a ⟨program⟩ or ⟨module⟩ to include literal circular references:

```
#1=(begin (display #\x) . #1#)
                          ⟹ error
```

4.3. Macros

Scheme programs can define and use new derived expression types, called *macros*. Program-defined expression types have the syntax

(⟨keyword⟩ ⟨datum⟩ ...)

where ⟨keyword⟩ is an identifier that uniquely determines the expression type. This identifier is called the *syntactic keyword*, or simply *keyword*, of the macro. The number of the ⟨datum⟩s, and their syntax, depends on the expression type.

Each instance of a macro is called a *use* of the macro. The set of rules that specifies how a use of a macro is transcribed into a more primitive expression is called the *transformer* of the macro.

The macro definition facility consists of two parts:

- A set of expressions used to establish that certain identifiers are macro keywords, associate them with macro transformers, and control the scope within which a macro is defined, and
- a pattern language for specifying macro transformers.

The syntactic keyword of a macro may shadow variable bindings, and local variable bindings may shadow keyword bindings. All macros defined using the pattern language are "hygienic" and "referentially transparent" and thus preserve Scheme's lexical scoping [16, 17, 4, 9, 11]:

- If a macro transformer inserts a binding for an identifier (variable or keyword), the identifier will in effect be renamed throughout its scope to avoid conflicts with other identifiers. Note that a `define` at top level may or may not introduce a binding; see section 5.2.
- If a macro transformer inserts a free reference to an identifier, the reference refers to the binding that was visible where the transformer was specified, regardless of any local bindings that may surround the use of the macro.

4.3.1. Binding constructs for syntactic keywords

`Let-syntax` and `letrec-syntax` are analogous to `let` and `letrec`, but they bind syntactic keywords to macro transformers instead of binding variables to locations that contain values. Syntactic keywords may also be bound at top level; see section 5.3.

`(let-syntax` ⟨bindings⟩ ⟨body⟩`)` syntax

Syntax: ⟨Bindings⟩ should have the form

`((`⟨keyword⟩ ⟨transformer spec⟩`) ...)`

Each ⟨keyword⟩ is an identifier, each ⟨transformer spec⟩ is an instance of `syntax-rules`, and ⟨body⟩ should be a sequence of one or more expressions. It is an error for a ⟨keyword⟩ to appear more than once in the list of keywords being bound.

Semantics: The ⟨body⟩ is expanded in the syntactic environment obtained by extending the syntactic environment of the `let-syntax` expression with macros whose keywords are the ⟨keyword⟩s, bound to the specified transformers. Each binding of a ⟨keyword⟩ has ⟨body⟩ as its region.

```
(let-syntax ((when (syntax-rules ()
                     ((when test stmt1 stmt2 ...)
                      (if test
                          (begin stmt1
                                 stmt2 ...))))))
  (let ((if #t))
    (when if (set! if 'now))
    if))                    ⟹  now

(let ((x 'outer))
  (let-syntax ((m (syntax-rules () ((m) x))))
    (let ((x 'inner))
      (m))))                ⟹  outer
```

`(letrec-syntax` ⟨bindings⟩ ⟨body⟩`)` syntax

Syntax: Same as for `let-syntax`.

Semantics: The ⟨body⟩ is expanded in the syntactic environment obtained by extending the syntactic environment of the `letrec-syntax` expression with macros whose keywords are the ⟨keyword⟩s, bound to the specified transformers. Each binding of a ⟨keyword⟩ has the ⟨bindings⟩ as well as the ⟨body⟩ within its region, so the transformers can transcribe expressions into uses of the macros introduced by the `letrec-syntax` expression.

```
(letrec-syntax
  ((my-or (syntax-rules ()
            ((my-or) #f)
            ((my-or e) e)
            ((my-or e1 e2 ...)
             (let ((temp e1))
               (if temp
                   temp
                   (my-or e2 ...)))))))
  (let ((x #f)
        (y 7)
        (temp 8)
        (let odd?)
        (if even?))
    (my-or x
           (let temp)
           (if y)
           y)))             ⟹  7
```

4.3.2. Pattern language

A ⟨transformer spec⟩ has one of the following form:

`(syntax-rules (`⟨literal⟩ `...)` syntax
⟨syntax rule⟩ `...)`
`(syntax-rules` ⟨ellipsis⟩ `(`⟨literal⟩ `...)` syntax
⟨syntax rule⟩ `...)`
_ auxiliary syntax
... auxiliary syntax

Syntax: Each ⟨literal⟩, as well as the ⟨ellipsis⟩ in the second form must be an identifier, and each ⟨syntax rule⟩ should be of the form

`(`⟨pattern⟩ ⟨template⟩`)`

The ⟨pattern⟩ in a ⟨syntax rule⟩ is a list ⟨pattern⟩ whose first subform is an identifier.

A ⟨pattern⟩ is either an identifier, a constant, or one of the following

`(`⟨pattern⟩ `...)`
`(`⟨pattern⟩ ⟨pattern⟩ `...` `.` ⟨pattern⟩`)`
`(`⟨pattern⟩ `...` ⟨pattern⟩ ⟨ellipsis⟩ ⟨pattern⟩ `...)`
`(`⟨pattern⟩ `...` ⟨pattern⟩ ⟨ellipsis⟩ ⟨pattern⟩ `...`
`.` ⟨pattern⟩`)`
`#(`⟨pattern⟩ `...)`
`#(`⟨pattern⟩ `...` ⟨pattern⟩ ⟨ellipsis⟩ ⟨pattern⟩ `...)`

and a template is either an identifier, a constant, or one of the following

`(`⟨element⟩ `...)`
`(`⟨element⟩ ⟨element⟩ `...` `.` ⟨template⟩`)`
`(`⟨ellipsis⟩ ⟨template⟩`)`
`#(`⟨element⟩ `...)`

where an ⟨element⟩ is a ⟨template⟩ optionally followed by an ⟨ellipsis⟩. An ⟨ellipsis⟩ is the identifier specified in the second form of `syntax-rules`, or the default identifier `...` (three consecutive periods) otherwise.

Semantics: An instance of `syntax-rules` produces a new macro transformer by specifying a sequence of hygienic rewrite rules. A use of a macro whose keyword is associated with a transformer specified by `syntax-rules` is matched against the patterns contained in the ⟨syntax rule⟩s, beginning with the leftmost ⟨syntax rule⟩. When a match is found, the macro use is transcribed hygienically according to the template.

An identifier appearing within a ⟨pattern⟩ may be an underscore ("_"), a literal identifier listed in the list of ⟨literal⟩s, or the ⟨ellipsis⟩. All other identifiers appearing within a ⟨pattern⟩ are *pattern variables*.

The keyword at the beginning of the pattern in a ⟨syntax rule⟩ is not involved in the matching and is not considered a pattern variable or literal identifier.

Pattern variables match arbitrary input elements and are used to refer to elements of the input in the template. It is an error for the same pattern variable to appear more than once in a ⟨pattern⟩.

Underscores also match arbitrary input subforms but are not pattern variables and so cannot be used to refer to those elements. If an underscore appears in the ⟨literal⟩s list, then that takes precedence and underscores in the ⟨pattern⟩ match as literals. Multiple underscores may appear in a ⟨pattern⟩.

Identifiers that appear in `(`⟨literal⟩ `...)` are interpreted as literal identifiers to be matched against corresponding subforms of the input. A subform in the input matches a literal identifier if and only if it is an identifier and either both its occurrence in the macro expression and its occurrence in the macro definition have the same lexical binding, or the two identifiers are equal and both have no lexical binding.

A subpattern followed by ⟨ellipsis⟩ can match zero or more elements of the input, unless ⟨ellipsis⟩ appears in the ⟨literal⟩s in which case it is matched as a literal.

More formally, an input form F matches a pattern P if and only if:

- P is an underscore ("_").
- P is a non-literal identifier; or
- P is a literal identifier and F is an identifier with the same binding; or
- P is a list `(`P_1 `...` P_n`)` and F is a list of n forms that match P_1 through P_n, respectively; or
- P is an improper list `(`P_1 P_2 `...` P_n `.` P_{n+1}`)` and F is a list or improper list of n or more forms that match P_1 through P_n, respectively, and whose nth "cdr" matches P_{n+1}; or
- P is of the form `(`P_1 `...` P_{e-1} P_e ⟨ellipsis⟩ P_{m+1} `...` P_n`)` where F is a proper list of n forms, the first $e-1$ of which match P_1 through P_{e-1}, respectively, whose next $m-k$ forms each match P_e, whose remaining $n-m$ forms match P_{m+1} through P_n; or
- P is of the form `(`P_1 `...` P_{e-1} P_e ⟨ellipsis⟩ P_{m+1} `...` P_n `.` P_x`)` where F is an list or improper list of n forms, the first $e-1$ of which match P_1 through P_{e-1}, whose next $m-k$ forms each match P_e, whose remaining $n-m$ forms match P_{m+1} through P_n, and whose nth and final cdr matches P_x; or
- P is a vector of the form `#(`P_1 `...` P_n`)` and F is a vector of n forms that match P_1 through P_n; or
- P is of the form `#(`P_1 `...` P_{e-1} P_e ⟨ellipsis⟩ P_{m+1} `...`P_n`)` where F is a vector of n forms the first $e-1$ of which match P_1 through P_{e-1}, whose next $m-k$ forms each match P_e, and whose remaining $n-m$ forms matche P_{m+1} through P_n; or

- P is a datum and F is equal to P in the sense of the `equal?` procedure.

It is an error to use a macro keyword, within the scope of its binding, in an expression that does not match any of the patterns.

When a macro use is transcribed according to the template of the matching ⟨syntax rule⟩, pattern variables that occur in the template are replaced by the subforms they match in the input. Pattern variables that occur in subpatterns followed by one or more instances of the identifier ⟨ellipsis⟩ are allowed only in subtemplates that are followed by as many instances of ⟨ellipsis⟩. They are replaced in the output by all of the subforms they match in the input, distributed as indicated. It is an error if the output cannot be built up as specified.

Identifiers that appear in the template but are not pattern variables or the identifier ⟨ellipsis⟩ are inserted into the output as literal identifiers. If a literal identifier is inserted as a free identifier then it refers to the binding of that identifier within whose scope the instance of `syntax-rules` appears. If a literal identifier is inserted as a bound identifier then it is in effect renamed to prevent inadvertent captures of free identifiers.

A template of the form (⟨ellipsis⟩ ⟨template⟩) is identical to ⟨template⟩, except that ellipses within the template have no special meaning. That is, any ellipses contained within ⟨template⟩ are treated as ordinary identifiers. In particular, the template (⟨ellipsis⟩ ⟨ellipsis⟩) produces a single ⟨ellipsis⟩. This allows syntactic abstractions to expand into forms containing ellipses.

```
(define-syntax be-like-begin
  (syntax-rules ()
    ((be-like-begin name)
     (define-syntax name
       (syntax-rules ()
         ((name expr (... ...))
          (begin expr (... ...))))))))

(be-like-begin sequence)
(sequence 1 2 3 4)          ⟹ 4
```

As an example, if `let` and `cond` are defined as in section 7.3 then they are hygienic (as required) and the following is not an error.

```
(let ((=> #f))
  (cond (#t => 'ok)))       ⟹ ok
```

The macro transformer for `cond` recognizes `=>` as a local variable, and hence an expression, and not as the top-level identifier `=>`, which the macro transformer treats as a syntactic keyword. Thus the example expands into

```
(let ((=> #f))
  (if #t (begin => 'ok)))
```

instead of

```
(let ((=> #f))
  (let ((temp #t))
    (if temp ('ok temp))))
```

which would result in an invalid procedure call.

4.3.3. Signalling errors in macros

`(syntax-error` ⟨message⟩ ⟨args⟩ `...)` syntax

`syntax-error` is similar to `error` except that implementations with an expansion pass separate from evaluation should signal an error as soon as the `syntax-error` form is expanded. This can be used as a `syntax-rules` ⟨template⟩ for a ⟨pattern⟩ that is an invalid use of the macro, which can provide more descriptive error messages. ⟨message⟩ should be a string literal, and ⟨args⟩ arbitrary forms providing additional information.

```
(define-syntax simple-let
  (syntax-rules ()
    ((_ (head ... ((x . y) val) . tail)
        body1 body2 ...)
     (syntax-error
      "expected an identifier but got"
      (x . y)))
    ((_ ((name val) ...) body1 body2 ...)
     ((lambda (name ...) body1 body2 ...)
      val ...))))
```

5. Program structure

5.1. Programs

A Scheme program consists of a sequence of *program parts*: expressions, definitions, syntax definitions, record-type definitions, imports, and includes. A collection of program parts may be encapsulated in a module to be reused by multiple programs. Expressions are described in chapter 4; the other program parts, as well as modules, are the subject of the rest of the present chapter.

Programs and modules are typically stored in files, although programs may be entered interactively to a running Scheme system, and other paradigms are possible.

Program parts other than expressions that are present at the top level of a program can be interpreted declaratively. They cause bindings to be created in the top level environment or modify the value of existing top-level bindings. Expressions occurring at the top level of a program are interpreted imperatively; they are executed in order when the program is invoked or loaded, and typically perform some kind of initialization.

At the top level of a program `(begin` ⟨form$_1$⟩ ...`)` is equivalent to the sequence of expressions, definitions, and syntax definitions that form the body of the `begin`.

5.2. Definitions

Definitions are valid in some, but not all, contexts where expressions are allowed. They are valid only at the top level of a ⟨program⟩ and at the beginning of a ⟨body⟩.

A definition should have one of the following forms:

- `(define` ⟨variable⟩ ⟨expression⟩`)`
- `(define (`⟨variable⟩ ⟨formals⟩`)` ⟨body⟩`)`

  ⟨Formals⟩ should be either a sequence of zero or more variables, or a sequence of one or more variables followed by a space-delimited period and another variable (as in a lambda expression). This form is equivalent to

  ```
  (define ⟨variable⟩
    (lambda (⟨formals⟩) ⟨body⟩)).
  ```

- `(define (`⟨variable⟩ `.` ⟨formal⟩`)` ⟨body⟩`)`

  ⟨Formal⟩ should be a single variable. This form is equivalent to

  ```
  (define ⟨variable⟩
    (lambda ⟨formal⟩ ⟨body⟩)).
  ```

5.2.1. Top level definitions

At the top level of a program, a definition

```
(define ⟨variable⟩ ⟨expression⟩)
```

has essentially the same effect as the assignment expression

```
(set! ⟨variable⟩ ⟨expression⟩)
```

if ⟨variable⟩ is bound to non-syntax. However, if ⟨variable⟩ is not bound, or is bound to a *syntax definition* (see below), then the definition will bind ⟨variable⟩ to a new location before performing the assignment, whereas it would be an error to perform a `set!` on an unbound variable.

```
(define add3
  (lambda (x) (+ x 3)))
(add3 3)                    ⟹  6
(define first car)
(first '(1 2))              ⟹  1
```

Some implementations of Scheme use an initial environment in which all possible variables are bound to locations, most of which contain undefined values. Top level definitions in such an implementation are truly equivalent to assignments.

5.2.2. Internal definitions

Definitions may occur at the beginning of a ⟨body⟩ (that is, the body of a `lambda`, `let`, `let*`, `letrec`, `letrec*`, `let-syntax`, or `letrec-syntax` expression or that of a definition of an appropriate form). Such definitions are known as *internal definitions* as opposed to the top level definitions described above. The variable defined by an internal definition is local to the ⟨body⟩. That is, ⟨variable⟩ is bound rather than assigned, and the region of the binding is the entire ⟨body⟩. For example,

```
(let ((x 5))
  (define foo (lambda (y) (bar x y)))
  (define bar (lambda (a b) (+ (* a b) a)))
  (foo (+ x 3)))            ⟹  45
```

An expanded ⟨body⟩ containing internal definitions can always be converted into a completely equivalent `letrec*` expression. For example, the `let` expression in the above example is equivalent to

```
(let ((x 5))
  (letrec* ((foo (lambda (y) (bar x y)))
            (bar (lambda (a b) (+ (* a b) a))))
    (foo (+ x 3))))
```

Just as for the equivalent `letrec*` expression, it must be possible to evaluate each ⟨expression⟩ of every internal definition in a ⟨body⟩ without assigning or referring to the value of the corresponding ⟨variable⟩ or the ⟨variable⟩ of any of the definitions that follow it in ⟨body⟩.

It is an error to define the same identifier more than once in the same ⟨body⟩.

Wherever an internal definition may occur `(begin` ⟨definition$_1$⟩ `...)` is equivalent to the sequence of definitions that form the body of the `begin`.

5.3. Syntax definitions

Syntax definitions are valid wherever definitions are. They have the following form:

`(define-syntax` ⟨keyword⟩ ⟨transformer spec⟩`)`

⟨Keyword⟩ is an identifier, and the ⟨transformer spec⟩ should be an instance of `syntax-rules`. If the `define-syntax` occurs at the top-level, then the top-level syntactic environment is extended by binding the ⟨keyword⟩ to the specified transformer, but existing references to any top-level binding for ⟨keyword⟩ remain unchanged. Otherwise, it is an *internal syntax definition*, and is local to the ⟨body⟩ in which it is defined.

```
(let ((x 1) (y 2))
  (define-syntax swap!
    (syntax-rules ()
      ((swap! a b)
       (let ((tmp a))
         (set! a b)
         (set! b tmp)))))
  (swap! x y)
  (list x y))                ⟹ (2 1)
```

Although macros may expand into definitions and syntax definitions in any context that permits them, it is an error for a definition or syntax definition to define an identifier whose binding is needed to determine the meaning of the definition itself, or any preceeding definition in a group of internal definitions. Similarly in a group of internal definitions, it is an error for a definition to define an identifier whose binding is needed to determine the boundary between the group and the expressions that follow the group. For example, the following are errors:

```
(define define 3)

(begin (define begin list))

(let-syntax
  ((foo (syntax-rules ()
          ((foo (proc args ...) body ...)
           (define proc
             (lambda (args ...)
               body ...))))))
  (let ((x 3))
    (foo (plus x y) (+ x y))
    (define foo x)
    (plus foo x)))
```

5.4. Record-type definitions

This section describes syntax for creating new data types, called record types. A predicate, constructor, and field accessors and modifiers are defined for each record type. Record-type definitions are valid wherever definitions are.

`(define-record-type` *name constructor pred field* `...)`
syntax

Syntax: ⟨name⟩ and ⟨pred⟩ should be identifiers. The ⟨constructor⟩ should be of the form

`(`⟨constructor name⟩ ⟨field name⟩ `...)`

and each ⟨field⟩ should be of the form

`(`⟨field name⟩ ⟨accessor name⟩`)`

| `(`⟨field name⟩ ⟨accessor name⟩ ⟨modifier name⟩`)`

It is an error for the same identifier to occur more than once as a field name.

`define-record-type` is generative: each use creates a new record type that is distinct from all existing types, including other record types and Scheme's predefined types.

An instance of `define-record-type` is equivalent to the following definitions:

- ⟨name⟩ is bound to a representation of the record type itself, possibly as a syntactic form.
- ⟨constructor name⟩ is bound to a procedure that takes as many arguments as there are ⟨field name⟩s in the `(`⟨constructor name⟩ `...)` subform and returns a new record of type ⟨name⟩. Fields whose names are listed with ⟨constructor name⟩ have the corresponding argument as their initial value. The initial values of all other fields are unspecified.
- ⟨pred⟩ is a predicate that returns `#t` when given a value returned by ⟨constructor name⟩ and `#f` for everything else.
- Each ⟨accessor name⟩ is a procedure that takes a record of type ⟨name⟩ and returns the current value of the corresponding field. It is an error to pass an accessor a value which is not a record of the appropriate type.
- Each ⟨modifier name⟩ is a procedure that takes a record of type ⟨name⟩ and a value which becomes the new value of the corresponding field; an unspecified value is returned. It is an error to pass a modifier a first argument which is not a record of the appropriate type.

The following definition

```
(define-record-type <pare>
  (kons x y)
  pare?
  (x kar set-kar!)
  (y kdr))
```

defines `kons` to be a constructor, `kar` and `kdr` to be accessors, `set-kar!` to be a modifier, and `pare?` to be a predicate for instances of `<pare>`.

```
(pare? (kons 1 2))        ⟹ #t
(pare? (cons 1 2))        ⟹ #f
(kar (kons 1 2))          ⟹ 1
(kdr (kons 1 2))          ⟹ 2
(let ((k (kons 1 2)))
  (set-kar! k 3)
  (kar k))                ⟹ 3
```

5.5. Modules

Modules provide a way to encapsulate programs and manage the top-level namespace. This section defines the notation and semantics for modules.

5.5.1. Module Syntax

A module definition takes the following form:

`(module` ⟨module name⟩
⟨module declaration⟩ ...`)`

⟨module name⟩ is a list of identifiers or exact integers used to identify the module uniquely when importing from other programs or modules.

A ⟨module declaration⟩ can be any of:

- `(export` ⟨export spec⟩ ...`)`
- `(import` ⟨import set⟩ ...`)`
- `(begin` ⟨command or definition⟩ ...`)`
- `(include` ⟨$filename_1$⟩ ⟨$filename_2$⟩ ...`)`
- `(include-ci` ⟨$filename_1$⟩ ⟨$filename_2$⟩ ...`)`
- `(cond-expand` ⟨cond-expand clause⟩ ...`)`

An `export` declaration specifies a list of identifiers which may be made visible to other modules or programs. An ⟨export spec⟩ must have one of the following forms:

- ⟨identifier⟩
- `(rename` ⟨$identifier_1$⟩ ⟨$identifier_2$⟩`)`

In an ⟨export spec⟩, an ⟨identifier⟩ names a single binding defined within or imported into the module, where the external name for the export is the same as the name of the binding within the module. A rename spec exports the binding named by ⟨$identifier_1$⟩ in each `(`⟨$identifier_1$⟩ ⟨$identifier_2$⟩`)` pairing, using ⟨$identifier_2$⟩ as the external name.

An `import` declaration provides a way to import the identifiers exported by a module. Each ⟨import set⟩ names a set of bindings from another module and possibly specifies local names for the imported bindings. It must be one of the following:

- ⟨module name⟩
- `(only` ⟨import set⟩ ⟨identifier⟩ ...`)`
- `(except` ⟨import set⟩ ⟨identifier⟩ ...`)`
- `(prefix` ⟨import set⟩ ⟨identifier⟩`)`
- `(rename` ⟨import set⟩ `(`⟨$identifier_1$⟩ ⟨$identifier_2$⟩`)` ...`)`

In the first form, all of the identifiers in the named module's export clauses are imported with the same names (or the exported names if exported with a `rename` form). The additional ⟨import set⟩ forms modify this set as follows:

- An `only` form produces a subset of the given ⟨import set⟩, including only the listed identifiers (after any renaming). It is an error if any of the listed identifiers are not found in the original set.
- An `except` form produces a subset of the given ⟨import set⟩, excluding the listed identifiers (after any renaming). It is an error if any of the listed identifiers are not found in the original set.
- A `rename` form modifies the given ⟨import set⟩, replacing each instance of ⟨$identifier_1$⟩ with ⟨$identifier_2$⟩. It is an error if any of the listed ⟨$identifier_1$⟩s are not found in the original set.
- A `prefix` form automatically renames all identifiers in the given ⟨import set⟩, prefixing each with the specified ⟨identifier⟩.

`import` declarations may also be made at the top-level of a program. In a module declaration, it is an error to import the same identifier more than once with different bindings, to redefine or mutate and imported binding with `define`, `define-syntax` or `set!`.

The `begin`, `include`, and `include-ci` declarations are used to specify the commands and definitions that make up the body of the module. `begin` takes a list of forms

to be spliced literally, analogous to the top-level `begin`. `include` and `include-ci` both take one or more filenames, read all top-level forms from the files and include the results into the module body as though wrapped in a `begin`. `include-ci` uses a case-folding reader when reading the forms from the file.

The `cond-expand` module declaration provides a way to statically expand different module declarations depending on the platform or implementation under which the module is being loaded. A ⟨cond-expand clause⟩ must be of the following form:

`(`⟨feature requirement⟩ ⟨module declaration⟩ `...)`

The last clause may be an "else clause," which has the form

`(else` ⟨module declaration⟩ `...)`

A ⟨feature requirement⟩ must be one of the following forms:

- ⟨feature identifier⟩
- `(module` ⟨module name⟩`)`
- `(and` ⟨feature requirement⟩ `...)`
- `(or` ⟨feature requirement⟩ `...)`
- `(not` ⟨feature requirement⟩`)`

Each implementation maintains a list of feature identifiers which are present, as well as a list of modules which can be imported. The value of a ⟨feature requirement⟩ can be determined by replacing each ⟨feature identifier⟩ and `(module` ⟨module name⟩`)` on the implementation's lists with `#t`, and all other feature identifiers and module names with `#f`, then evaluating the resulting expression as a Scheme boolean expression under the normal interpretation of `and`, `or`, and `not`.

A `cond-expand` form is then expanded by evaluating the ⟨feature requirement⟩s of successive ⟨cond-expand clause⟩ in order, until one of them returns `#t`. When a true clause is found, the corresponding ⟨module declaration⟩s are spliced into the current module definition and the remaining clauses are ignored. If none of the ⟨feature requirement⟩s evaluate to `#t`, then if there is an `else` clause its ⟨module declaration⟩s are included, otherwise the `cond-expand` has no effect.

The exact features provided are implementation-defined, but for portability a set of recommended features is given in appendix B.

After all `cond-expand` forms are expanded, a new environment is constructed for the module consisting of all imported bindings. The forms from all `begin`, `include` and `include-ci` declarations are expanded in that environment in the order in which they occur in the module declaration.

The top-level forms in a module are executed in the order in which they occur when the module is loaded. A module is loaded zero or more times when it is imported by a program or by another module which is about to be loaded, but must be loaded at least once per program in which it is so imported.

5.5.2. Module Examples

```
(module (stack)
  (export make push! pop! empty!)
  (import (scheme base))
  (begin
   (define (make) (list ()))
   (define (push! s v)
     (set-car! s (cons v (car s))))
   (define (pop! s) (let ((v (caar s)))
                      (set-car! s (cdar s))
                      v))
   (define (empty! s) (set-car! s ()))))

(module (balloons)
  (export make push pop)
  (import (scheme))
  (begin
   (define (make w h) (cons w h))
   (define (push b amt)
     (cons (- (car b) amt) (+ (cdr b) amt)))
   (define (pop b) (display "Boom! ")
                   (display (* (car b) (cdr b)))
                   (newline))))

(module (party)
  ;; Total exports:
  ;; make, push, push!, make-party, pop!
  (export (rename (balloon:make make)
                  (balloon:push push))
          push!
          make-party
          (rename (party-pop! pop!)))
  (import
   (scheme base)
   (only (stack) make push! pop!) ; not empty!
   (prefix (balloons) balloon:))
  (begin
   ;; Creates a party as a stack of balloons,
   ;; starting with two balloons
   (define (make-party)
     (let ((s (make))) ; from stack
       (push! s (balloon:make 10 10))
       (push! s (balloon:make 12 9))
       s))
   (define (party-pop! p)
     (balloon:pop (pop! p)))))

(module (main)
  (export)
  (import (scheme base) (party))
```

```
(begin
 (define p (make-party))
 (pop! p)        ; displays "Boom! 108"
 (push! p (push (make 5 5) 1))
 (pop! p)))      ; displays "Boom! 24"
```

6. Standard procedures

This chapter describes Scheme's built-in procedures. The initial (or "top level") Scheme environment is empty, and bindings must be introduced with `import`.

Implementations may provide an interactive session called a *REPL* (Read-Eval-Print-Loop), where Scheme expressions are entered and evaluated one at a time. For convenience and ease of use, the "top-level" Scheme environment in an interactive session is not empty, but must start out with a number of variables bound to locations containing at least the bindings provided by the `(scheme base)` module. This module includes mostly core syntax and primitive procedures that manipulate data. For example, the variable `abs` is bound to (a location initially containing) a procedure of one argument that computes the absolute value of a number, and the variable `+` is bound to a procedure that computes sums. The full list of `(scheme base)` bindings can be found in chapter A.

A program may use a top-level definition to bind any variable. It may subsequently alter any such binding by an assignment (see 4.1.6). These operations do not modify the behavior of Scheme's built-in procedures, or any procedure defined in a module (see section 5.5). Altering any top-level binding that has not been introduced by a definition has an unspecified effect on the behavior of the built-in procedures.

6.1. Equivalence predicates

A *predicate* is a procedure that always returns a boolean value (`#t` or `#f`). An *equivalence predicate* is the computational analogue of a mathematical equivalence relation (it is symmetric, reflexive, and transitive). Of the equivalence predicates described in this section, `eq?` is the finest or most discriminating, and `equal?` is the coarsest. `Eqv?` is slightly less discriminating than `eq?`.

`(eqv?` *obj*$_1$ *obj*$_2$`)` procedure

The `eqv?` procedure defines a useful equivalence relation on objects. Briefly, it returns `#t` if *obj*$_1$ and *obj*$_2$ should normally be regarded as the same object. This relation is left slightly open to interpretation, but the following partial specification of `eqv?` holds for all implementations of Scheme.

The `eqv?` procedure returns `#t` if:

- *obj*$_1$ and *obj*$_2$ are both `#t` or both `#f`.
- *obj*$_1$ and *obj*$_2$ are both symbols and

```
(string=? (symbol->string obj1)
          (symbol->string obj2))
                              ⟹  #t
```

  Note: This assumes that neither *obj*$_1$ nor *obj*$_2$ is an "uninterned symbol" as alluded to in section 6.3.3. This report does not presume to specify the behavior of `eqv?` on implementation-dependent extensions.

- *obj*$_1$ and *obj*$_2$ are both numbers, are numerically equal (see `=`, section 6.2), and are either both exact or both inexact.
- *obj*$_1$ and *obj*$_2$ are both characters and are the same character according to the `char=?` procedure (section 6.3.4).
- both *obj*$_1$ and *obj*$_2$ are the empty list.
- *obj*$_1$ and *obj*$_2$ are pairs, vectors, bytevectors, records, or strings that denote the same locations in the store (section 3.4).

The `eqv?` procedure returns `#f` if:

- *obj*$_1$ and *obj*$_2$ are of different types (section 3.2).
- one of *obj*$_1$ and *obj*$_2$ is `#t` but the other is `#f`.
- *obj*$_1$ and *obj*$_2$ are symbols but

```
(string=? (symbol->string obj1)
          (symbol->string obj2))
                              ⟹  #f
```

- one of *obj*$_1$ and *obj*$_2$ is an exact number but the other is an inexact number.
- *obj*$_1$ and *obj*$_2$ are numbers for which the `=` procedure returns `#f`.
- *obj*$_1$ and *obj*$_2$ are characters for which the `char=?` procedure returns `#f`.
- one of *obj*$_1$ and *obj*$_2$ is the empty list but the other is not.
- *obj*$_1$ and *obj*$_2$ are pairs, vectors, bytevectors, records, or strings that denote distinct locations.
- *obj*$_1$ and *obj*$_2$ are procedures that would behave differently (return different value(s) or have different side effects) for some arguments.

```
(eqv? 'a 'a)                    ⟹  #t
(eqv? 'a 'b)                    ⟹  #f
(eqv? 2 2)                      ⟹  #t
(eqv? '() '())                  ⟹  #t
(eqv? 100000000 100000000)      ⟹  #t
(eqv? (cons 1 2) (cons 1 2))    ⟹  #f
(eqv? (lambda () 1)
      (lambda () 2))            ⟹  #f
(eqv? #f 'nil)                  ⟹  #f
```

The following examples illustrate cases in which the above rules do not fully specify the behavior of `eqv?`. All that can be said about such cases is that the value returned by `eqv?` must be a boolean.

```
(eqv? "" "")                    ⟹  unspecified
(eqv? '#() '#())                ⟹  unspecified
(eqv? (lambda (x) x)
      (lambda (x) x))           ⟹  unspecified
(let ((p (lambda (x) x)))
  (eqv? p p))                   ⟹  unspecified
(eqv? (lambda (x) x)
      (lambda (y) y))           ⟹  unspecified
```

The next set of examples shows the use of `eqv?` with procedures that have local state. `Gen-counter` must return a distinct procedure every time, since each procedure has its own internal counter. `Gen-loser`, however, returns equivalent procedures each time, since the local state does not affect the value or side effects of the procedures. However, `eqv?` may or may not detect this equivalence.

```
(define gen-counter
  (lambda ()
    (let ((n 0))
      (lambda () (set! n (+ n 1)) n))))
(let ((g (gen-counter)))
  (eqv? g g))                   ⟹  #t
(eqv? (gen-counter) (gen-counter))
                                ⟹  #f
(define gen-loser
  (lambda ()
    (let ((n 0))
      (lambda () (set! n (+ n 1)) 27))))
(let ((g (gen-loser)))
  (eqv? g g))                   ⟹  #t
(eqv? (gen-loser) (gen-loser))
                                ⟹  unspecified

(letrec ((f (lambda () (if (eqv? f g) 'both 'f)))
         (g (lambda () (if (eqv? f g) 'both 'g))))
  (eqv? f g))
                                ⟹  unspecified

(letrec ((f (lambda () (if (eqv? f g) 'f 'both)))
         (g (lambda () (if (eqv? f g) 'g 'both))))
  (eqv? f g))
                                ⟹  #f
```

Since it is an error to modify constant objects (those returned by literal expressions), implementations are permitted, though not required, to share structure between constants where appropriate. Thus the value of `eqv?` on constants is sometimes implementation-dependent.

```
(eqv? '(a) '(a))                ⟹  unspecified
(eqv? "a" "a")                  ⟹  unspecified
(eqv? '(b) (cdr '(a b)))        ⟹  unspecified
(let ((x '(a)))
  (eqv? x x))                   ⟹  #t
```

Rationale: The above definition of `eqv?` allows implementations latitude in their treatment of procedures and literals: implementations are free either to detect or to fail to detect that two procedures or two literals are equivalent to each other, and can decide whether or not to merge representations of equivalent objects by using the same pointer or bit pattern to represent both.

`(eq?` obj_1 obj_2`)` procedure

`Eq?` is similar to `eqv?` except that in some cases it is capable of discerning distinctions finer than those detectable by `eqv?`.

`Eq?` and `eqv?` are guaranteed to have the same behavior on symbols, booleans, the empty list, pairs, procedures, and non-empty strings and vectors. `Eq?`'s behavior on numbers and characters is implementation-dependent, but it will always return either true or false, and will return true only when `eqv?` would also return true. `Eq?` may also behave differently from `eqv?` on empty vectors and empty strings.

```
(eq? 'a 'a)                     ⟹  #t
(eq? '(a) '(a))                 ⟹  unspecified
(eq? (list 'a) (list 'a))       ⟹  #f
(eq? "a" "a")                   ⟹  unspecified
(eq? "" "")                     ⟹  unspecified
(eq? '() '())                   ⟹  #t
(eq? 2 2)                       ⟹  unspecified
(eq? #\A #\A)                   ⟹  unspecified
(eq? car car)                   ⟹  #t
(let ((n (+ 2 3)))
  (eq? n n))                    ⟹  unspecified
(let ((x '(a)))
  (eq? x x))                    ⟹  #t
(let ((x '#()))
  (eq? x x))                    ⟹  #t
(let ((p (lambda (x) x)))
  (eq? p p))                    ⟹  #t
```

Rationale: It will usually be possible to implement `eq?` much more efficiently than `eqv?`, for example, as a simple pointer comparison instead of as some more complicated operation. One reason is that it may not be possible to compute `eqv?` of two numbers in constant time, whereas `eq?` implemented as pointer comparison will always finish in constant time. `Eq?` may be used like `eqv?` in applications using procedures to implement objects with state since it obeys the same constraints as `eqv?`.

`(equal?` obj_1 obj_2`)` procedure

`Equal?` recursively compares the contents of pairs, vectors, and strings, applying `eqv?` on other objects such as numbers and symbols. A rule of thumb is that objects are generally `equal?` if they print the same. `Equal?` must always terminate, even if its arguments are circular data structures.

```
(equal? 'a 'a)                ⟹  #t
(equal? '(a) '(a))            ⟹  #t
(equal? '(a (b) c)
        '(a (b) c))           ⟹  #t
(equal? "abc" "abc")          ⟹  #t
(equal? 2 2)                  ⟹  #t
(equal? (make-vector 5 'a)
        (make-vector 5 'a))   ⟹  #t
(equal? (lambda (x) x)
        (lambda (y) y))       ⟹  unspecified
```

6.2. Numbers

It is important to distinguish between mathematical numbers, the Scheme numbers that attempt to model them, the machine representations used to implement the Scheme numbers, and notations used to write numbers. This report uses the types *number*, *complex*, *real*, *rational*, and *integer* to refer to both mathematical numbers and Scheme numbers. Machine representations such as fixed point and floating point are referred to by names such as *fixnum* and *flonum*. Fixnums are integers with a limited and machine-dependent range; flonums are real numbers with a limited and machine-dependent range and precision.

6.2.1. Numerical types

Mathematically, numbers may be arranged into a tower of subtypes in which each level is a subset of the level above it:

number
complex
real
rational
integer

For example, 3 is an integer. Therefore 3 is also a rational, a real, and a complex. The same is true of the Scheme numbers that model 3. For Scheme numbers, these types are defined by the predicates `number?`, `complex?`, `real?`, `rational?`, and `integer?`.

There is no simple relationship between a number's type and its representation inside a computer. Although most implementations of Scheme will offer at least two different representations of 3, these different representations denote the same integer.

Scheme's numerical operations treat numbers as abstract data, as independent of their representation as possible. Although an implementation of Scheme may use fixnum, flonum, and perhaps other representations for numbers, this should not be apparent to a casual programmer writing simple programs.

It is necessary, however, to distinguish between numbers that are represented exactly and those that may not be. For example, indexes into data structures must be known exactly, as must some polynomial coefficients in a symbolic algebra system. On the other hand, the results of measurements are inherently inexact, and irrational numbers may be approximated by rational and therefore inexact approximations. In order to catch uses of inexact numbers where exact numbers are required, Scheme explicitly distinguishes exact from inexact numbers. This distinction is orthogonal to the dimension of type.

6.2.2. Exactness

Scheme numbers are either *exact* or *inexact*. A number is exact if it was written as an exact constant or was derived from exact numbers using only exact operations. A number is inexact if it was written as an inexact constant, if it was derived using inexact ingredients, or if it was derived using inexact operations. Thus inexactness is a contagious property of a number.

If two implementations produce exact results for a computation that did not involve inexact intermediate results, the two ultimate results will be mathematically equivalent. This is generally not true of computations involving inexact numbers since approximate methods such as floating point arithmetic may be used, but it is the duty of each implementation to make the result as close as practical to the mathematically ideal result.

Rational operations such as `+` should always produce exact results when given exact arguments. If the operation is unable to produce an exact result, then it may either report the violation of an implementation restriction or it may silently coerce its result to an inexact value. See section 6.2.3.

With the exception of `inexact->exact`, the operations described in this section must generally return inexact results when given any inexact arguments. An operation may, however, return an exact result if it can prove that the value of the result is unaffected by the inexactness of its arguments. For example, multiplication of any number by an exact zero may produce an exact zero result, even if the other argument is inexact.

6.2.3. Implementation restrictions

Implementations of Scheme are not required to implement the whole tower of subtypes given in section 6.2.1, but

they must implement a coherent subset consistent with both the purposes of the implementation and the spirit of the Scheme language. For example, an implementation in which all numbers are real may still be quite useful.

Implementations may also support only a limited range of numbers of any type, subject to the requirements of this section. The supported range for exact numbers of any type may be different from the supported range for inexact numbers of that type. For example, an implementation that uses flonums to represent all its inexact real numbers may support a practically unbounded range of exact integers and rationals while limiting the range of inexact reals (and therefore the range of inexact integers and rationals) to the dynamic range of the flonum format. Furthermore the gaps between the representable inexact integers and rationals are likely to be very large in such an implementation as the limits of this range are approached.

An implementation of Scheme must support exact integers throughout the range of numbers that may be used for indexes of lists, vectors, and strings or that may result from computing the length of a list, vector, or string. The `length`, `vector-length`, and `string-length` procedures must return an exact integer, and it is an error to use anything but an exact integer as an index. Furthermore any integer constant within the index range, if expressed by an exact integer syntax, will indeed be read as an exact integer, regardless of any implementation restrictions that may apply outside this range. Finally, the procedures listed below will always return exact integer results provided all their arguments are exact integers and the mathematically expected results are representable as exact integers within the implementation:

```
+                   -                   *
quotient            remainder           modulo
max                 min                 abs
numerator           denominator         gcd
lcm                 floor               ceiling
truncate            round               rationalize
expt                exact-integer-sqrt
floor/              ceiling/            truncate/
round/              euclidean/
floor-quotient      floor-remainder
ceiling-quotient    ceiling-remainder
truncate-quotient   truncate-remainder
round-quotient      round-remainder
euclidean-quotient  euclidean-remainder
```

Implementations are encouraged, but not required, to support exact integers and exact rationals of practically unlimited size and precision, and to implement the above procedures and the `/` procedure in such a way that they always return exact results when given exact arguments. If one of these procedures is unable to deliver an exact result when given exact arguments, then it may either report a violation of an implementation restriction or it may silently coerce its result to an inexact number. Such a coercion may cause an error later.

An implementation may use floating point and other approximate representation strategies for inexact numbers. This report recommends, but does not require, that the IEEE 754 standard be followed by implementations that use flonum representations, and that implementations using other representations should match or exceed the precision achievable using these floating point standards [14]. In particular, the description of transcendental functions in IEEE 754:2008 should be followed by such implementations, particularly with respect to infinities and NaNs.

In particular, implementations that use flonum representations must follow these rules: A flonum result must be represented with at least as much precision as is used to express any of the inexact arguments to that operation. It is desirable (but not required) for potentially inexact operations such as `sqrt`, when applied to exact arguments, to produce exact answers whenever possible (for example the square root of an exact 4 should be an exact 2). If, however, an exact number is operated upon so as to produce an inexact result (as by `sqrt`), and if the result is represented as a flonum, then the most precise flonum format available must be used; but if the result is represented in some other way then the representation must have at least as much precision as the most precise flonum format available.

In addition, implementations that use flonum representations may distinguish special number objects called positive infinity, negative infinity, and NaN.

Positive infinity is regarded as an inexact real (but not rational) number object that represents an indeterminate number greater than the numbers represented by all rational number objects. Negative infinity is regarded as an inexact real (but not rational) number object that represents an indeterminate number less than the numbers represented by all rational numbers.

A NaN is regarded as an inexact real (but not rational) number object so indeterminate that it might represent any real number, including positive or negative infinity, and might even be greater than positive infinity or less than negative infinity. It might even represent no number at all, as in the case of `(asin 2.0)`.

Note that either the real or the imaginary part of a complex number can be an infinity or NaN.

Although Scheme allows a variety of written notations for numbers, any particular implementation may support only some of them. For example, an implementation in which all numbers are real need not support the rectangular and polar notations for complex numbers. If an implementation encounters an exact numerical constant that it cannot represent as an exact number, then it may either report a violation of an implementation restriction or it may silently represent the constant by an inexact number.

6.2.4. Syntax of numerical constants

The syntax of the written representations for numbers is described formally in section 7.1.1. Note that case is not significant in numerical constants.

A number may be written in binary, octal, decimal, or hexadecimal by the use of a radix prefix. The radix prefixes are `#b` (binary), `#o` (octal), `#d` (decimal), and `#x` (hexadecimal). With no radix prefix, a number is assumed to be expressed in decimal.

A numerical constant may be specified to be either exact or inexact by a prefix. The prefixes are `#e` for exact, and `#i` for inexact. An exactness prefix may appear before or after any radix prefix that is used. If the written representation of a number has no exactness prefix, the constant may be either inexact or exact. It is inexact if it contains a decimal point, an exponent, or a "`#`" character in the place of a digit, otherwise it is exact.

In systems with inexact numbers of varying precisions it may be useful to specify the precision of a constant. For this purpose, implementations may accept numerical constants written with an exponent marker that indicates the desired precision of the inexact representation. The letters `s`, `f`, `d`, and `l`, meaning *short*, *single*, *double*, and *long* precision respectively, are acceptable in place of `e`. The default precision has at least as much precision as *double*, but implementations may wish to allow this default to be set by the user.

```
3.14159265358979F0
        Round to single — 3.141593
0.6L0
        Extend to long — .600000000000000
```

The numbers positive infinity, negative infinity and NaN are written `+inf.0`, `-inf.0` and `+nan.0` respectively. Implementations are not required to support them, but if they do, they must be in conformance with IEEE 754. However, implementations are not required to support signaling NaNs, or provide a way to distinguish between different NaNs.

6.2.5. Numerical operations

The reader is referred to section 1.3.3 for a summary of the naming conventions used to specify restrictions on the types of arguments to numerical routines. The examples used in this section assume that any numerical constant written using an exact notation is indeed represented as an exact number. Some examples also assume that certain numerical constants written using an inexact notation can be represented without loss of accuracy; the inexact constants were chosen so that this is likely to be true in implementations that use flonums to represent inexact numbers.

`(number?` *obj*`)` procedure
`(complex?` *obj*`)` procedure
`(real?` *obj*`)` procedure
`(rational?` *obj*`)` procedure
`(integer?` *obj*`)` procedure

These numerical type predicates can be applied to any kind of argument, including non-numbers. They return `#t` if the object is of the named type, and otherwise they return `#f`. In general, if a type predicate is true of a number then all higher type predicates are also true of that number. Consequently, if a type predicate is false of a number, then all lower type predicates are also false of that number.

If z is a complex number, then `(real?` z`)` is true if and only if `(zero? (imag-part z))` and `(exact? (imag-part z))` are both true. If x is an inexact real number, then `(integer?` x`)` is true if and only if `(=` x `(round` x`))`.

```
(complex? 3+4i)            ⟹  #t
(complex? 3)               ⟹  #t
(real? 3)                  ⟹  #t
(real? -2.5+0.0i)          ⟹  #t
(real? #e1e10)             ⟹  #t
(rational? 6/10)           ⟹  #t
(rational? 6/3)            ⟹  #t
(integer? 3+0i)            ⟹  #t
(integer? 3.0)             ⟹  #t
(integer? 8/4)             ⟹  #t
```

Note: The behavior of these type predicates on inexact numbers is unreliable, since any inaccuracy may affect the result.

Note: In many implementations the `complex?` procedure will be the same as `number?`, but unusual implementations may be able to represent some irrational numbers exactly or may extend the number system to support some kind of non-complex numbers.

`(exact?` z`)` procedure
`(inexact?` z`)` procedure

These numerical predicates provide tests for the exactness of a quantity. For any Scheme number, precisely one of these predicates is true.

```
(exact? 3.0)               ⟹  #f
(exact? #e3.0)             ⟹  #t
(inexact? 3.)              ⟹  #t
```

`(exact-integer?` z`)` procedure

The conjunction of `exact?` and `integer?`, returns `#t` if z is both exact and an integer. Otherwise, `#f` is returned.

`(finite?` z`)` inexact module procedure

`Finite` returns `#t` on all real numbers except `+inf.0`, `-inf.0`, and `+nan.0`, and on complex numbers if their real

and imaginary parts are both finite. Otherwise it returns `#f`.

```
(finite? 3)                ⟹  #t
(finite? +inf.0)           ⟹  #f
(finite? 3.0+inf.0i)       ⟹  #f
```

`(nan?` z`)` inexact module procedure

`Nan` returns `#t` on `+nan.0`, and on any complex number if its real part or its imaginary part or both are `+nan.0`. Otherwise it returns `#f`.

```
(nan? +nan.0)              ⟹  #t
(nan? +nan.0+5.0i)         ⟹  #t
```

`(=` z_1 z_2 z_3 `...)` procedure
`(<` x_1 x_2 x_3 `...)` procedure
`(>` x_1 x_2 x_3 `...)` procedure
`(<=` x_1 x_2 x_3 `...)` procedure
`(>=` x_1 x_2 x_3 `...)` procedure

These procedures return `#t` if their arguments are (respectively): equal, monotonically increasing, monotonically decreasing, monotonically nondecreasing, or monotonically nonincreasing.

These predicates are required to be transitive.

Note: The traditional implementations of these predicates in Lisp-like languages are not transitive.

Note: While it is not an error to compare inexact numbers using these predicates, the results may be unreliable because a small inaccuracy may affect the result; this is especially true of `=` and `zero?`. When in doubt, consult a numerical analyst.

`(zero?` z`)` procedure
`(positive?` x`)` procedure
`(negative?` x`)` procedure
`(odd?` n`)` procedure
`(even?` n`)` procedure

These numerical predicates test a number for a particular property, returning `#t` or `#f`. See note above.

`(max` x_1 x_2 `...)` procedure
`(min` x_1 x_2 `...)` procedure

These procedures return the maximum or minimum of their arguments.

```
(max 3 4)                  ⟹  4    ; exact
(max 3.9 4)                ⟹  4.0  ; inexact
```

Note: If any argument is inexact, then the result will also be inexact (unless the procedure can prove that the inaccuracy is not large enough to affect the result, which is possible only in unusual implementations). If `min` or `max` is used to compare numbers of mixed exactness, and the numerical value of the result cannot be represented as an inexact number without loss of accuracy, then the procedure may report a violation of an implementation restriction.

`(+` z_1 `...)` procedure
`(*` z_1 `...)` procedure

These procedures return the sum or product of their arguments.

```
(+ 3 4)                    ⟹  7
(+ 3)                      ⟹  3
(+)                        ⟹  0
(* 4)                      ⟹  4
(*)                        ⟹  1
```

`(-` z_1 z_2`)` procedure
`(-` z`)` procedure
`(-` z_1 z_2 `...)` procedure
`(/` z_1 z_2`)` procedure
`(/` z`)` procedure
`(/` z_1 z_2 `...)` procedure

With two or more arguments, these procedures return the difference or quotient of their arguments, associating to the left. With one argument, however, they return the additive or multiplicative inverse of their argument.

```
(- 3 4)                    ⟹  -1
(- 3 4 5)                  ⟹  -6
(- 3)                      ⟹  -3
(/ 3 4 5)                  ⟹  3/20
(/ 3)                      ⟹  1/3
```

`(abs` x`)` procedure

`Abs` returns the absolute value of its argument.

```
(abs -7)                   ⟹  7
```

`(floor/` n_1 n_2`)` procedure
`(floor-quotient` n_1 n_2`)` procedure
`(floor-remainder` n_1 n_2`)` procedure
`(ceiling/` n_1 n_2`)` procedure
`(ceiling-quotient` n_1 n_2`)` procedure
`(ceiling-remainder` n_1 n_2`)` procedure
`(truncate/` n_1 n_2`)` procedure
`(truncate-quotient` n_1 n_2`)` procedure
`(truncate-remainder` n_1 n_2`)` procedure
`(round/` n_1 n_2`)` procedure
`(round-quotient` n_1 n_2`)` procedure

`(round-remainder` n_1 n_2`)` procedure
`(euclidean/` n_1 n_2`)` procedure
`(euclidean-quotient` n_1 n_2`)` procedure
`(euclidean-remainder` n_1 n_2`)` procedure

These procedures, all in the division module, implement number-theoretic (integer) division. n_2 should be non-zero. The procedures ending in `/` return two integers; the other procedures return an integer. All the procedures compute a quotient n_q and remainder n_r such that $n_1 = n_2 n_q + n_r$. For each of the five division operators, there are three procedures defined as follows:

```
(⟨operator⟩/ n1 n2)            ⟹ nq nr
(⟨operator⟩-quotient n1 n2)    ⟹ nq
(⟨operator⟩-remainder n1 n2)   ⟹ nr
```

The remainder n_r is determined by the choice of integer n_q: $n_r = n_1 - n_2 n_q$. Each set of operators uses a different choice of n_q:

ceiling $n_q = \lceil n_1/n_2 \rceil$
floor $n_q = \lfloor n_1/n_2 \rfloor$
truncate $n_q = \text{truncate}(n_1/n_2)$
round $n_q = [n_1/n_2]$
euclidean if $n_2 > 0, n_q = \lfloor n_1/n_2 \rfloor$; if $n_2 < 0, n_q = \lceil n_1/n_2 \rceil$

For any of the operators, and for integers n_1 and n_2 with n_2 not equal to 0,

```
(= n1 (+ (* n2 (⟨operator⟩-quotient n1 n2))
         (⟨operator⟩-remainder n1 n2)))
                              ⟹ #t
```

provided all numbers involved in that computation are exact.

`(quotient` n_1 n_2`)` procedure
`(remainder` n_1 n_2`)` procedure
`(modulo` n_1 n_2`)` procedure

`Quotient` and `remainder` are equivalent to `truncate-quotient` and `truncate-remainder` respectively. `Modulo` is equivalent to `floor-remainder`.

```
(modulo 13 4)              ⟹ 1
(remainder 13 4)           ⟹ 1

(modulo -13 4)             ⟹ 3
(remainder -13 4)          ⟹ -1

(modulo 13 -4)             ⟹ -3
(remainder 13 -4)          ⟹ 1

(modulo -13 -4)            ⟹ -1
(remainder -13 -4)         ⟹ -1

(remainder -13 -4.0)       ⟹ -1.0  ; inexact
```

Note: These procedures are provided for backward compatibility with earlier versions of this report.

`(gcd` n_1 ...`)` procedure
`(lcm` n_1 ...`)` procedure

These procedures return the greatest common divisor or least common multiple of their arguments. The result is always non-negative.

```
(gcd 32 -36)               ⟹ 4
(gcd)                      ⟹ 0
(lcm 32 -36)               ⟹ 288
(lcm 32.0 -36)             ⟹ 288.0  ; inexact
(lcm)                      ⟹ 1
```

`(numerator` q`)` procedure
`(denominator` q`)` procedure

These procedures return the numerator or denominator of their argument; the result is computed as if the argument was represented as a fraction in lowest terms. The denominator is always positive. The denominator of 0 is defined to be 1.

```
(numerator (/ 6 4))        ⟹ 3
(denominator (/ 6 4))      ⟹ 2
(denominator
  (exact->inexact (/ 6 4))) ⟹ 2.0
```

`(floor` x`)` procedure
`(ceiling` x`)` procedure
`(truncate` x`)` procedure
`(round` x`)` procedure

These procedures return integers. `Floor` returns the largest integer not larger than x. `Ceiling` returns the smallest integer not smaller than x. `Truncate` returns the integer closest to x whose absolute value is not larger than the absolute value of x. `Round` returns the closest integer to x, rounding to even when x is halfway between two integers.

Rationale: `Round` rounds to even for consistency with the default rounding mode specified by the IEEE floating point standard.

Note: If the argument to one of these procedures is inexact, then the result will also be inexact. If an exact value is needed, the result should be passed to the `inexact->exact` procedure.

```
(floor -4.3)               ⟹ -5.0
(ceiling -4.3)             ⟹ -4.0
(truncate -4.3)            ⟹ -4.0
(round -4.3)               ⟹ -4.0

(floor 3.5)                ⟹ 3.0
(ceiling 3.5)              ⟹ 4.0
(truncate 3.5)             ⟹ 3.0
```

```
(round 3.5)                ⟹  4.0  ; inexact

(round 7/2)                ⟹  4    ; exact
(round 7)                  ⟹  7
```

`(rationalize` x y`)` procedure

`Rationalize` returns the *simplest* rational number differing from x by no more than y. A rational number r_1 is *simpler* than another rational number r_2 if $r_1 = p_1/q_1$ and $r_2 = p_2/q_2$ (in lowest terms) and $|p_1| \leq |p_2|$ and $|q_1| \leq |q_2|$. Thus 3/5 is simpler than 4/7. Although not all rationals are comparable in this ordering (consider 2/7 and 3/5) any interval contains a rational number that is simpler than every other rational number in that interval (the simpler 2/5 lies between 2/7 and 3/5). Note that $0 = 0/1$ is the simplest rational of all.

```
(rationalize
  (inexact->exact .3) 1/10) ⟹ 1/3    ; exact
(rationalize .3 1/10)       ⟹ #i1/3  ; inexact
```

`(exp` z`)`	inexact module procedure
`(log` z`)`	inexact module procedure
`(sin` z`)`	inexact module procedure
`(cos` z`)`	inexact module procedure
`(tan` z`)`	inexact module procedure
`(asin` z`)`	inexact module procedure
`(acos` z`)`	inexact module procedure
`(atan` z`)`	inexact module procedure
`(atan` y x`)`	inexact module procedure

These procedures compute the usual transcendental functions. `Log` computes the natural logarithm of z (not the base ten logarithm). `Asin`, `acos`, and `atan` compute arcsine ($\sin^{-1}$), arccosine ($\cos^{-1}$), and arctangent ($\tan^{-1}$), respectively. The two-argument variant of `atan` computes `(angle (make-rectangular` x y`))` (see below), even in implementations that don't support general complex numbers.

In general, the mathematical functions log, arcsine, arccosine, and arctangent are multiply defined. The value of $\log z$ is defined to be the one whose imaginary part lies in the range from $-\pi$ (exclusive) to π (inclusive). $\log 0$ is undefined. With log defined this way, the values of $\sin^{-1} z$, $\cos^{-1} z$, and $\tan^{-1} z$ are according to the following formulæ:

$$\sin^{-1} z = -i \log(iz + \sqrt{1 - z^2})$$

$$\cos^{-1} z = \pi/2 - \sin^{-1} z$$

$$\tan^{-1} z = (\log(1 + iz) - \log(1 - iz))/(2i)$$

The above specification follows [29], which in turn cites [21]; refer to these sources for more detailed discussion of branch cuts, boundary conditions, and implementation of these functions. When it is possible these procedures produce a real result from a real argument.

`(sqrt` z`)` inexact module procedure

Returns the principal square root of z. The result will have either positive real part, or zero real part and non-negative imaginary part.

`(exact-integer-sqrt` k`)` procedure

Returns two non-negative exact integers s and r where $k = s^2 + r$ and $k < (s + 1)^2$.

```
(exact-integer-sqrt 4)      ⟹ 2 0
(exact-integer-sqrt 5)      ⟹ 2 1
```

`(expt` z_1 z_2`)` procedure

Returns z_1 raised to the power z_2. For nonzero z1, this is

$$z_1{}^{z_2} = e^{z_2 \log z_1}$$

0.0^z is 1.0 if $z = 0.0$, and 0.0 if `(real-part z)` is positive. For other cases in which the first argument is zero, either an error is signalled or an unspecified number is returned.

`(make-rectangular` x_1 x_2`)`	complex module procedure
`(make-polar` x_3 x_4`)`	complex module procedure
`(real-part` z`)`	complex module procedure
`(imag-part` z`)`	complex module procedure
`(magnitude` z`)`	complex module procedure
`(angle` z`)`	complex module procedure

Suppose x_1, x_2, x_3, and x_4 are real numbers and z is a complex number such that

$$z = x_1 + x_2 i = x_3 \cdot e^{ix_4}$$

Then

```
(make-rectangular x1 x2)    ⟹ z
(make-polar x3 x4)          ⟹ z
(real-part z)               ⟹ x1
(imag-part z)               ⟹ x2
(magnitude z)               ⟹ |x3|
(angle z)                   ⟹ x_angle
```

where $-\pi < x_{angle} \leq \pi$ with $x_{angle} = x_4 + 2\pi n$ for some integer n.

`Make-polar` may return an inexact complex number even if its arguments are exact.

Rationale: `Magnitude` is the same as `abs` for a real argument, but `abs` is in the `base` module, whereas `magnitude` is in the optional `complex` module.

`(exact->inexact` *z*`)` procedure
`(inexact->exact` *z*`)` procedure

`Exact->inexact` returns an inexact representation of *z*. The value returned is the inexact number that is numerically closest to the argument. For inexact arguments, the result is the same as the argument. For exact complex numbers, the result is a complex number whose real and imaginary parts are the result of applying `exact->inexact` to the real and imaginary parts of the argument, respectively. If an exact argument has no reasonably close inexact equivalent, then a violation of an implementation restriction may be reported.

`Inexact->exact` returns an exact representation of *z*. The value returned is the exact number that is numerically closest to the argument. For exact arguments, the result is the same as the argument. For inexact non-integral real arguments, the implementation may return a rational approximation, or may report an implementation violation. For inexact complex arguments, the result is a complex number whose real and imaginary parts are result of applying `inexact->exact` to the real and imaginary parts of the argument, respectively. If an inexact argument has no reasonably close exact equivalent, then a violation of an implementation restriction may be reported.

These procedures implement the natural one-to-one correspondence between exact and inexact integers throughout an implementation-dependent range. See section 6.2.3.

Note: The names `exact->inexact` and `inexact->exact` are historical anomalies; the argument to each of these procedures may be either exact or inexact.

6.2.6. Numerical input and output

`(number->string` *z*`)` procedure
`(number->string` *z radix*`)` procedure

Radix must be an exact integer, either 2, 8, 10, or 16. If omitted, *radix* defaults to 10. The procedure `number->string` takes a number and a radix and returns as a string an external representation of the given number in the given radix such that

```
(let ((number number)
      (radix radix))
  (eqv? number
        (string->number (number->string number
                                        radix)
                        radix)))
```

is true. It is an error if no possible result makes this expression true.

If *z* is inexact, the radix is 10, and the above expression can be satisfied by a result that contains a decimal point, then the result contains a decimal point and is expressed using the minimum number of digits (exclusive of exponent and trailing zeroes) needed to make the above expression true [5, 7]; otherwise the format of the result is unspecified.

The result returned by `number->string` never contains an explicit radix prefix.

Note: The error case can occur only when *z* is not a complex number or is a complex number with a non-rational real or imaginary part.

Rationale: If *z* is an inexact number represented using flonums, and the radix is 10, then the above expression is normally satisfied by a result containing a decimal point. The unspecified case allows for infinities, NaNs, and non-flonum representations.

`(string->number` *string*`)` procedure
`(string->number` *string radix*`)` procedure

Returns a number of the maximally precise representation expressed by the given *string*. *Radix* must be an exact integer, either 2, 8, 10, or 16. If supplied, *radix* is a default radix that may be overridden by an explicit radix prefix in *string* (e.g. `"#o177"`). If *radix* is not supplied, then the default radix is 10. If *string* is not a syntactically valid notation for a number, then `string->number` returns `#f`.

```
(string->number "100")      ⟹  100
(string->number "100" 16)   ⟹  256
(string->number "1e2")      ⟹  100.0
(string->number "15##")     ⟹  1500.0
```

Note: The domain of `string->number` may be restricted by implementations in the following ways. `String->number` is permitted to return `#f` whenever *string* contains an explicit radix prefix. If all numbers supported by an implementation are real, then `string->number` is permitted to return `#f` whenever *string* uses the polar or rectangular notations for complex numbers. If all numbers are integers, then `string->number` may return `#f` whenever the fractional notation is used. If all numbers are exact, then `string->number` may return `#f` whenever an exponent marker or explicit exactness prefix is used, or if a `#` appears in place of a digit. If all inexact numbers are integers, then `string->number` may return `#f` whenever a decimal point is used.

6.3. Other data types

This section describes operations on some of Scheme's non-numeric data types: booleans, pairs, lists, symbols, characters, strings and vectors.

6.3.1. Booleans

The standard boolean objects for true and false are written as `#t` and `#f`. What really matters, though, are the objects

that the Scheme conditional expressions (`if`, `cond`, `and`, `or`, `do`) treat as true or false. The phrase "a true value" (or sometimes just "true") means any object treated as true by the conditional expressions, and the phrase "a false value" (or "false") means any object treated as false by the conditional expressions.

Of all the standard Scheme values, only `#f` counts as false in conditional expressions. Except for `#f`, all standard Scheme values, including `#t`, pairs, the empty list, symbols, numbers, strings, vectors, bytevectors, records, and procedures, count as true.

Note: Programmers accustomed to other dialects of Lisp should be aware that Scheme distinguishes both `#f` and the empty list from the symbol `nil`.

Boolean constants evaluate to themselves, so they do not need to be quoted in programs.

```
#t                      ⟹  #t
#f                      ⟹  #f
'#f                     ⟹  #f
```

`(not` *obj*`)` procedure

`Not` returns `#t` if *obj* is false, and returns `#f` otherwise.

```
(not #t)                ⟹  #f
(not 3)                 ⟹  #f
(not (list 3))          ⟹  #f
(not #f)                ⟹  #t
(not '())               ⟹  #f
(not (list))            ⟹  #f
(not 'nil)              ⟹  #f
```

`(boolean?` *obj*`)` procedure

`Boolean?` returns `#t` if *obj* is either `#t` or `#f` and returns `#f` otherwise.

```
(boolean? #f)           ⟹  #t
(boolean? 0)            ⟹  #f
(boolean? '())          ⟹  #f
```

6.3.2. Pairs and lists

A *pair* (sometimes called a *dotted pair*) is a record structure with two fields called the car and cdr fields (for historical reasons). Pairs are created by the procedure `cons`. The car and cdr fields are accessed by the procedures `car` and `cdr`. The car and cdr fields are assigned by the procedures `set-car!` and `set-cdr!`.

Pairs are used primarily to represent lists. A *list* can be defined recursively as either the empty list or a pair whose cdr is a list. More precisely, the set of lists is defined as the smallest set X such that

- The empty list is in X.
- If *list* is in X, then any pair whose cdr field contains *list* is also in X.

The objects in the car fields of successive pairs of a list are the elements of the list. For example, a two-element list is a pair whose car is the first element and whose cdr is a pair whose car is the second element and whose cdr is the empty list. The length of a list is the number of elements, which is the same as the number of pairs.

The empty list is a special object of its own type (it is not a pair); it has no elements and its length is zero.

Note: The above definitions imply that all lists have finite length and are terminated by the empty list.

The most general notation (external representation) for Scheme pairs is the "dotted" notation `(`c_1 `.` c_2`)` where c_1 is the value of the car field and c_2 is the value of the cdr field. For example `(4 . 5)` is a pair whose car is 4 and whose cdr is 5. Note that `(4 . 5)` is the external representation of a pair, not an expression that evaluates to a pair.

A more streamlined notation can be used for lists: the elements of the list are simply enclosed in parentheses and separated by spaces. The empty list is written `()` . For example,

```
(a b c d e)
```

and

```
(a . (b . (c . (d . (e . ()))))))
```

are equivalent notations for a list of symbols.

A chain of pairs not ending in the empty list is called an *improper list*. Note that an improper list is not a list. The list and dotted notations can be combined to represent improper lists:

```
(a b c . d)
```

is equivalent to

```
(a . (b . (c . d)))
```

Whether a given pair is a list depends upon what is stored in the cdr field. When the `set-cdr!` procedure is used, an object can be a list one moment and not the next:

```
(define x (list 'a 'b 'c))
(define y x)
y                       ⟹  (a b c)
(list? y)               ⟹  #t
(set-cdr! x 4)          ⟹  unspecified
x                       ⟹  (a . 4)
(eqv? x y)              ⟹  #t
y                       ⟹  (a . 4)
(list? y)               ⟹  #f
(set-cdr! x x)          ⟹  unspecified
(list? x)               ⟹  #f
```

Within literal expressions and representations of objects read by the `read` procedure, the forms `'`⟨datum⟩, `` `⟨datum⟩, `,`⟨datum⟩, and `,@`⟨datum⟩ denote two-element lists whose first elements are the symbols `quote`, `quasiquote`, `unquote`, and `unquote-splicing`, respectively. The second element in each case is ⟨datum⟩. This convention is supported so that arbitrary Scheme programs may be represented as lists. That is, according to Scheme's grammar, every ⟨expression⟩ is also a ⟨datum⟩ (see section 7.1.2). Among other things, this permits the use of the `read` procedure to parse Scheme programs. See section 3.3.

`(pair?` *obj*`)` procedure

`Pair?` returns `#t` if *obj* is a pair, and otherwise returns `#f`.

```
(pair? '(a . b))          ⟹  #t
(pair? '(a b c))          ⟹  #t
(pair? '())               ⟹  #f
(pair? '#(a b))           ⟹  #f
```

`(cons` obj_1 obj_2`)` procedure

Returns a newly allocated pair whose car is obj_1 and whose cdr is obj_2. The pair is guaranteed to be different (in the sense of `eqv?`) from every existing object.

```
(cons 'a '())             ⟹  (a)
(cons '(a) '(b c d))      ⟹  ((a) b c d)
(cons "a" '(b c))         ⟹  ("a" b c)
(cons 'a 3)               ⟹  (a . 3)
(cons '(a b) 'c)          ⟹  ((a b) . c)
```

`(car` *pair*`)` procedure

Returns the contents of the car field of *pair*. Note that it is an error to take the car of the empty list.

```
(car '(a b c))            ⟹  a
(car '((a) b c d))        ⟹  (a)
(car '(1 . 2))            ⟹  1
(car '())                 ⟹  error
```

`(cdr` *pair*`)` procedure

Returns the contents of the cdr field of *pair*. Note that it is an error to take the cdr of the empty list.

```
(cdr '((a) b c d))        ⟹  (b c d)
(cdr '(1 . 2))            ⟹  2
(cdr '())                 ⟹  error
```

`(set-car!` *pair obj*`)` procedure

Stores *obj* in the car field of *pair*. The value returned by `set-car!` is unspecified.

```
(define (f) (list 'not-a-constant-list))
(define (g) '(constant-list))
(set-car! (f) 3)          ⟹  unspecified
(set-car! (g) 3)          ⟹  error
```

`(set-cdr!` *pair obj*`)` procedure

Stores *obj* in the cdr field of *pair*. The value returned by `set-cdr!` is unspecified.

`(caar` *pair*`)` procedure
`(cadr` *pair*`)` procedure
⋮ ⋮
`(cdddar` *pair*`)` procedure
`(cddddr` *pair*`)` procedure

These procedures are compositions of `car` and `cdr`, where for example `caddr` could be defined by

```
(define caddr (lambda (x) (car (cdr (cdr x)))))
```

Arbitrary compositions, up to four deep, are provided. There are twenty-eight of these procedures in all.

`(null?` *obj*`)` procedure

Returns `#t` if *obj* is the empty list, otherwise returns `#f`.

`(list?` *obj*`)` procedure

Returns `#t` if *obj* is a list, otherwise returns `#f`. By definition, all lists have finite length and are terminated by the empty list.

```
(list? '(a b c))          ⟹  #t
(list? '())               ⟹  #t
(list? '(a . b))          ⟹  #f
(let ((x (list 'a)))
  (set-cdr! x x)
  (list? x))              ⟹  #f
```

`(make-list` *k*`)` procedure
`(make-list` *k fill*`)` procedure

Returns a newly allocated list of *k* elements. If a second argument is given, then each element is initialized to *fill*. Otherwise the initial contents of each element is unspecified.

`(list` *obj ...*`)` procedure

Returns a newly allocated list of its arguments.

```
(list 'a (+ 3 4) 'c)      ⟹  (a 7 c)
(list)                    ⟹  ()
```

`(length` *list*`)` procedure

Returns the length of *list*.

```
(length '(a b c))          ⟹  3
(length '(a (b) (c d e)))  ⟹  3
(length '())               ⟹  0
```

(append *list* ...) procedure

Returns a list consisting of the elements of the first *list* followed by the elements of the other *list*s.

```
(append '(x) '(y))          ⟹  (x y)
(append '(a) '(b c d))      ⟹  (a b c d)
(append '(a (b)) '((c)))    ⟹  (a (b) (c))
```

The resulting list is always newly allocated, except that it shares structure with the last *list* argument. The last argument may actually be any object; an improper list results if the last argument is not a proper list.

```
(append '(a b) '(c . d))    ⟹  (a b c . d)
(append '() 'a)             ⟹  a
```

(reverse *list*) procedure

Returns a newly allocated list consisting of the elements of *list* in reverse order.

```
(reverse '(a b c))          ⟹  (c b a)
(reverse '(a (b c) d (e (f))))
          ⟹  ((e (f)) d (b c) a)
```

(list-tail *list k*) procedure

Returns the sublist of *list* obtained by omitting the first *k* elements. It is an error if *list* has fewer than *k* elements. `List-tail` could be defined by

```
(define list-tail
  (lambda (x k)
    (if (zero? k)
        x
        (list-tail (cdr x) (- k 1)))))
```

(list-ref *list k*) procedure

Returns the *k*th element of *list*. (This is the same as the car of (list-tail *list k*).) It is an error if *list* has fewer than *k* elements.

```
(list-ref '(a b c d) 2)     ⟹  c
(list-ref '(a b c d)
          (inexact->exact (round 1.8)))
          ⟹  c
```

(list-set! *list k obj*) procedure

k must be a valid index of *list*. `List-set!` stores *obj* in element *k* of *list*. The value returned by `list-set!` is unspecified.

```
(let ((lst (list 0 '(2 2 2 2) "Anna")))
  (list-set! lst 1 '("Sue" "Sue"))
  vec)
          ⟹  (0 ("Sue" "Sue") "Anna")

(list-set! '(0 1 2) 1 "doe")
          ⟹  error  ; constant list
```

(memq *obj list*) procedure
(memv *obj list*) procedure
(member *obj list*) procedure
(member *obj list compare*) procedure

These procedures return the first sublist of *list* whose car is *obj*, where the sublists of *list* are the non-empty lists returned by (list-tail *list k*) for *k* less than the length of *list*. If *obj* does not occur in *list*, then `#f` (not the empty list) is returned. `Memq` uses `eq?` to compare *obj* with the elements of *list*, while `memv` uses `eqv?` and `member` uses *compare* if given and `equal?` otherwise.

```
(memq 'a '(a b c))          ⟹  (a b c)
(memq 'b '(a b c))          ⟹  (b c)
(memq 'a '(b c d))          ⟹  #f
(memq (list 'a) '(b (a) c)) ⟹  #f
(member (list 'a)
        '(b (a) c))         ⟹  ((a) c)
(member "B"
        '("a" "b" "c")
        string-ci=?)        ⟹  ("b" "c")
(memq 101 '(100 101 102))   ⟹  unspecified
(memv 101 '(100 101 102))   ⟹  (101 102)
```

(assq *obj alist*) procedure
(assv *obj alist*) procedure
(assoc *obj alist*) procedure
(assoc *obj alist compare*) procedure

Alist (for "association list") must be a list of pairs. These procedures find the first pair in *alist* whose car field is *obj*, and returns that pair. If no pair in *alist* has *obj* as its car, then `#f` (not the empty list) is returned. `Assq` uses `eq?` to compare *obj* with the car fields of the pairs in *alist*, while `assv` uses `eqv?` and `assoc` uses *compare* if given and `equal?` otherwise.

```
(define e '((a 1) (b 2) (c 3)))
(assq 'a e)                 ⟹  (a 1)
(assq 'b e)                 ⟹  (b 2)
(assq 'd e)                 ⟹  #f
(assq (list 'a) '(((a)) ((b)) ((c))))
                            ⟹  #f
(assoc (list 'a) '(((a)) ((b)) ((c))))
                            ⟹  ((a))
(assq 5 '((2 3) (5 7) (11 13)))
                            ⟹  unspecified
(assv 5 '((2 3) (5 7) (11 13)))
                            ⟹  (5 7)
```

Rationale: Although they are ordinarily used as predicates, `memq`, `memv`, `member`, `assq`, `assv`, and `assoc` do not have question marks in their names because they return values that may be useful rather than just `#t` or `#f`.

`(list-copy` *list*`)` procedure

Returns a newly allocated copy of the given *list*.

6.3.3. Symbols

Symbols are objects whose usefulness rests on the fact that two symbols are identical (in the sense of `eqv?`) if and only if their names are spelled the same way. For instance, they may be used the way enumerated values are used in other languages.

The rules for writing a symbol are exactly the same as the rules for writing an identifier; see sections 2.1 and 7.1.1.

It is guaranteed that any symbol that has been returned as part of a literal expression, or read using the `read` procedure, and subsequently written out using the `write` procedure, will read back in as the identical symbol (in the sense of `eqv?`).

Note: Some implementations have values known as "uninterned symbols," which defeat write/read invariance, and also generate exceptions to the rule that two symbols are the same if and only if their names are spelled the same.

`(symbol?` *obj*`)` procedure

Returns `#t` if *obj* is a symbol, otherwise returns `#f`.

```
(symbol? 'foo)              ⟹  #t
(symbol? (car '(a b)))      ⟹  #t
(symbol? "bar")             ⟹  #f
(symbol? 'nil)              ⟹  #t
(symbol? '())               ⟹  #f
(symbol? #f)                ⟹  #f
```

`(symbol->string` *symbol*`)` procedure

Returns the name of *symbol* as a string. It is an error to apply mutation procedures like `string-set!` to strings returned by this procedure.

```
(symbol->string 'flying-fish)
                            ⟹  "flying-fish"
(symbol->string 'Martin)    ⟹  "Martin"
(symbol->string
   (string->symbol "Malvina"))
                            ⟹  "Malvina"
```

`(string->symbol` *string*`)` procedure

Returns the symbol whose name is *string*. This procedure can create symbols with names containing special characters that would require escaping when written.

```
(string->symbol "mISSISSIppi")
          ⟹  'mISSISSIppi
(eq? 'bitBlt (string->symbol "bitBlt"))
          ⟹  #t
(eq? 'JollyWog
     (string->symbol
       (symbol->string 'JollyWog)))
          ⟹  #t
(string=? "K. Harper, M.D."
          (symbol->string
            (string->symbol "K. Harper, M.D.")))
          ⟹  #t
```

6.3.4. Characters

Characters are objects that represent printed characters such as letters and digits. All Scheme implementations must support at least the ASCII character repertoire: that is, Unicode characters U+0000 through U+007F. Implementations may support any other Unicode characters they see fit, and may also support non-Unicode characters as well. Except as otherwise specified, the result of applying any of the following procedures to a non-Unicode character is implementation-dependent. Characters are written using the notation `#\`⟨character⟩ or `#\`⟨character name⟩ or `#\x`⟨hex scalar value⟩. For example:

```
#\a           ; lower case letter
#\A           ; upper case letter
#\(           ; left parenthesis
#\            ; the space character
#\space       ; the preferred way to write a space
#\tab         ; the tab character, U+0009
#\newline     ; the linefeed character, U+000A
#\return      ; the return character, U+000D
#\null        ; the null character, U+0000
#\alarm       ; U+0007
#\backspace   ; U+0008
#\escape      ; U+001B
#\delete      ; U+007F
#\x03BB       ; λ (if supported)
```

Case is significant in `#\`⟨character⟩, and in `#\`⟨character name⟩, but not in `#\x`⟨hex scalar value⟩. If ⟨character⟩ in `#\`⟨character⟩ is alphabetic, then the character following ⟨character⟩ must be a delimiter character such as a space or parenthesis. This rule resolves the ambiguous case where, for example, the sequence of characters "`#\space`" could be taken to be either a representation of the space character or a representation of the character "`#\s`" followed by a representation of the symbol "`pace`."

Characters written in the `#\` notation are self-evaluating. That is, they do not have to be quoted in programs.

Some of the procedures that operate on characters ignore the difference between upper case and lower case. The procedures that ignore case have "`-ci`" (for "case insensitive") embedded in their names.

`(char?` *obj*`)` procedure

Returns `#t` if *obj* is a character, otherwise returns `#f`.

`(char=?` $char_1$ $char_2$ $char_3$ ...`)` procedure
`(char<?` $char_1$ $char_2$ $char_3$ ...`)` procedure
`(char>?` $char_1$ $char_2$ $char_3$ ...`)` procedure
`(char<=?` $char_1$ $char_2$ $char_3$ ...`)` procedure
`(char>=?` $char_1$ $char_2$ $char_3$ ...`)` procedure

These procedures return `#t` if their arguments are (respectively): equal, monotonically increasing, monotonically decreasing, monotonically nondecreasing, or monotonically nonincreasing.

These predicates are required to be transitive.

These procedures impose a total ordering on the set of characters which is the same as the Unicode code point ordering. This is true independent of whether the implementation uses the Unicode representation internally.

`(char-ci=?` $char_1$ $char_2$ $char_3$ ...`)`
char module procedure
`(char-ci<?` $char_1$ $char_2$ $char_3$ ...`)`
char module procedure
`(char-ci>?` $char_1$ $char_2$ $char_3$ ...`)`
char module procedure
`(char-ci<=?` $char_1$ $char_2$ $char_3$ ...`)`
char module procedure
`(char-ci>=?` $char_1$ $char_2$ $char_3$ ...`)`
char module procedure

These procedures are similar to `char=?` et cetera, but they treat upper case and lower case letters as the same. For example, `(char-ci=? #\A #\a)` returns `#t`.

Specifically, these procedures behave as if `char-foldcase` were applied to their arguments before comparing them.

`(char-alphabetic?` *char*`)` char module procedure
`(char-numeric?` *char*`)` char module procedure
`(char-whitespace?` *char*`)` char module procedure
`(char-upper-case?` *letter*`)` char module procedure
`(char-lower-case?` *letter*`)` char module procedure

These procedures return `#t` if their arguments are alphabetic, numeric, whitespace, upper case, or lower case characters, respectively, otherwise they return `#f`.

Specifically, they must return `#t` when applied to characters with the Unicode properties Alphabetic, Numeric_Digit, White_Space, Uppercase, and Lowercase respectively, and `#f` when applied to any other Unicode characters. Note that many Unicode characters are alphabetic but neither upper nor lower case.

`(char->integer` *char*`)` procedure
`(integer->char` *n*`)` procedure

Given a Unicode character, `char->integer` returns an exact integer between 0 and `#xD7FF` or between `#xE000` and `#x10FFFF` which is equal to the Unicode code point of that character. Given a non-Unicode character, it returns an exact integer greater than `#x10FFFF`. This is true independent of whether the implementation uses the Unicode representation internally.

Given an exact integer that is the value returned by a character when `char->integer` is applied to it, `integer->char` returns that character.

`(char-upcase` *char*`)` char module procedure
`(char-downcase` *char*`)` char module procedure
`(char-foldcase` *char*`)` char module procedure

The `char-upcase` procedure, given an argument that forms the lowercase part of a Unicode casing pair, returns the uppercase member of the pair, provided that both characters are supported by the Scheme implementation. Note that Turkic casing pairs are not used. If the argument is not the lowercase part of such a pair, it is returned.

The `char-downcase` procedure, given an argument that forms the uppercase part of a Unicode casing pair, returns the lowercase member of the pair, provided that both characters are supported by the Scheme implementation. Note that Turkic casing pairs are not used. If the argument is not the uppercase part of such a pair, it is returned.

The `char-foldcase` procedure applies the Unicode simple case-folding algorithm to its argument and returns the result. Note that Turkic-specific folding is not used. If the argument is an uppercase letter, the result will be a lowercase letter.

Note that many Unicode lowercase characters do not have uppercase equivalents.

6.3.5. Strings

Strings are sequences of characters. Implementations may support characters that they do not allow to appear in strings. Strings are written as sequences of characters enclosed within doublequotes (`"`). Within a string literal, various escape sequences represent characters other than themselves. Escape sequences always start with a backslash (`\`):

- `\a` : alarm, U+0007
- `\b` : backspace, U+0008
- `\t` : character tabulation, U+0009
- `\n` : linefeed, U+000A

- `\r` : return, U+000D
- `\"` : doublequote, U+0022
- `\\` : backslash, U+005C
- `\`⟨intraline whitespace⟩⟨line ending⟩ ⟨intraline whitespace⟩ : nothing
- `\x`⟨hex scalar value⟩`;` : specified character (note the terminating semi-colon).

The result is unspecified if any other character in a string occurs after a backslash.

Except for a line ending, any character outside of an escape sequence stands for itself in the string literal. A line ending which is preceded by `\`⟨intraline whitespace⟩ expands to nothing (along with any trailing intraline whitespace), and can be used to indent strings for improved legibility. Any other line ending has the same effect as inserting a `\n` character into the string.

Example:

```
"The word \"recursion\" has many meanings."
```

The *length* of a string is the number of characters that it contains. This number is an exact, non-negative integer that is fixed when the string is created. The *valid indexes* of a string are the exact non-negative integers less than the length of the string. The first character of a string has index 0, the second has index 1, and so on.

In phrases such as "the characters of *string* beginning with index *start* and ending with index *end*," it is understood that the index *start* is inclusive and the index *end* is exclusive. Thus if *start* and *end* are the same index, a null substring is referred to, and if *start* is zero and *end* is the length of *string*, then the entire string is referred to.

Some of the procedures that operate on strings ignore the difference between upper and lower case. The versions that ignore case have "`-ci`" (for "case insensitive") embedded in their names.

`(string?` *obj*`)` procedure

Returns `#t` if *obj* is a string, otherwise returns `#f`.

`(make-string` *k*`)` procedure
`(make-string` *k char*`)` procedure

`Make-string` returns a newly allocated string of length *k*. If *char* is given, then all elements of the string are initialized to *char*, otherwise the contents of the string are unspecified.

`(string` *char* ...`)` procedure

Returns a newly allocated string composed of the arguments.

`(string-length` *string*`)` procedure

Returns the number of characters in the given *string*.

`(string-ref` *string k*`)` procedure

k must be a valid index of *string*. `String-ref` returns character *k* of *string* using zero-origin indexing.

`(string-set!` *string k char*`)` procedure

k must be a valid index of *string*. `String-set!` stores *char* in element *k* of *string* and returns an unspecified value.

```
(define (f) (make-string 3 #\*))
(define (g) "***")
(string-set! (f) 0 #\?)      ⟹  unspecified
(string-set! (g) 0 #\?)      ⟹  error
(string-set! (symbol->string 'immutable)
             0
             #\?)            ⟹  error
```

`(string=?` $char_1$ $char_2$ $char_3$ ...`)` procedure

Returns `#t` if all the strings are the same length and contain exactly the same characters in the same positions, otherwise returns `#f`.

`(string-ci=?` $char_1$ $char_2$ $char_3$ ...`)` procedure

Returns `#t` if, after case-folding, all the strings are the same length and contain the same characters in the same positions, otherwise returns `#f`. Specifically, these procedures behave as if `string-foldcase` were applied to their arguments before comparing them.

`(string-ni=?` $char_1$ $char_2$ $char_3$ ...`)` procedure

Returns `#t` if, after an implementation-defined normalization, all the strings are the same length and contain the same characters in the same positions, otherwise returns `#f`. The intent is to provide a means of comparing strings that should be considered equivalent in some situations but may be represented by a different sequence of characters.

Specifically, an implementation which supports Unicode should consider using Unicode normalization NFC or NFD as specified by Unicode TR#15. Implementations which only support ASCII or some other character set which provides no ambiguous representations of character sequences may define the normalization to be the identity operation, in which case `string-ni=?` is equivalent to `string=?`.

`(string<?` $string_1$ $string_2$ $string_3$ ...`)` procedure
`(string-ci<?` $string_1$ $string_2$ $string_3$ ...`)` procedure
`(string-ni<?` $string_1$ $string_2$ $string_3$ ...`)` procedure
`(string>?` $string_1$ $string_2$ $string_3$ ...`)` procedure
`(string-ci>?` $string_1$ $string_2$ $string_3$ ...`)` procedure

`(string-ni>?` *string*$_1$ *string*$_2$ *string*$_3$ ...`)` procedure
`(string<=?` *string*$_1$ *string*$_2$ *string*$_3$ ...`)` procedure
`(string-ci<=?` *string*$_1$ *string*$_2$ *string*$_3$ ...`)` procedure
`(string-ni<=?` *string*$_1$ *string*$_2$ *string*$_3$ ...`)` procedure
`(string>=?` *string*$_1$ *string*$_2$ *string*$_3$ ...`)` procedure
`(string-ci>=?` *string*$_1$ *string*$_2$ *string*$_3$ ...`)` procedure
`(string-ni>=?` *string*$_1$ *string*$_2$ *string*$_3$ ...`)` procedure

These procedures return `#t` if their arguments are (respectively): equal, monotonically increasing, monotonically decreasing, monotonically nondecreasing, or monotonically nonincreasing.

These predicates are required to be transitive.

These procedures compare strings in an implementation-defined way. One approach is to make them the lexicographic extensions to strings of the corresponding orderings on characters. In that case, `string<?` would be the lexicographic ordering on strings induced by the ordering `char<?` on characters, and if the two strings differ in length but are the same up to the length of the shorter string, the shorter string would be considered to be lexicographically less than the longer string. However, it is also permitted to use the natural ordering imposed by the internal representation of strings, or a more complex locale-specific ordering.

In all cases, a pair of strings must satisfy exactly one of `string<?`, `string=?`, and `string>?`, and must satisfy `string<=?` if and only if they do not satisfy `string>?` and `string>=?` if and only if they do not satisfy `string<?`.

The "`-ci`" procedures behave as if they applied `string-foldcase` to their arguments before invoking the corresponding procedures without "`-ci`".

The "`-ni`" procedures behave as if they applied the implementation-defined normalization used by `string-ni=?` to their arguments before invoking the corresponding procedures without "`-ni`".

`(string-upcase` *string*`)` char module procedure
`(string-downcase` *string*`)` char module procedure
`(string-foldcase` *string*`)` char module procedure

These procedures apply the Unicode full string uppercasing, lowercasing, and case-folding algorithms to their arguments and return the result. Note that Turkic-specific mappings and foldings are not used. The result may differ in length from the argument. What is more, a few characters have case-mappings that depend on the surrounding context. For example, Greek capital sigma normally lowercases to Greek small sigma, but at the end of a word it downcases to Greek small final sigma instead.

`(substring` *string start end*`)` procedure

String must be a string, and *start* and *end* must be exact integers satisfying

$$0 \le start \le end \le \texttt{(string-length } string\texttt{)}.$$

`Substring` returns a newly allocated string formed from the characters of *string* beginning with index *start* (inclusive) and ending with index *end* (exclusive).

`(string-append` *string* ...`)` procedure

Returns a newly allocated string whose characters form the concatenation of the given strings.

`(string->list` *string*`)` procedure
`(list->string` *list*`)` procedure

`String->list` returns a newly allocated list of the characters that make up the given string. `List->string` returns a newly allocated string formed from the characters in the list *list*, which must be a list of characters. `String->list` and `list->string` are inverses so far as `equal?` is concerned.

`(string-copy` *string*`)` procedure

Returns a newly allocated copy of the given *string*.

`(string-fill!` *string char*`)` procedure

Stores *char* in every element of the given *string* and returns an unspecified value.

6.3.6. Vectors

Vectors are heterogenous structures whose elements are indexed by integers. A vector typically occupies less space than a list of the same length, and the average time required to access a randomly chosen element is typically less for the vector than for the list.

The *length* of a vector is the number of elements that it contains. This number is a non-negative integer that is fixed when the vector is created. The *valid indexes* of a vector are the exact non-negative integers less than the length of the vector. The first element in a vector is indexed by zero, and the last element is indexed by one less than the length of the vector.

Vectors are written using the notation `#(`*obj* ...`)`. For example, a vector of length 3 containing the number zero in element 0, the list `(2 2 2 2)` in element 1, and the string `"Anna"` in element 2 can be written as following:

```
#(0 (2 2 2 2) "Anna")
```

Note that this is the external representation of a vector, not an expression evaluating to a vector. Like list constants, vector constants must be quoted:

```
'#(0 (2 2 2 2) "Anna")
        ⟹  #(0 (2 2 2 2) "Anna")
```

`(vector?` *obj*`)` procedure

Returns `#t` if *obj* is a vector, otherwise returns `#f`.

`(make-vector` *k*`)` procedure
`(make-vector` *k fill*`)` procedure

Returns a newly allocated vector of *k* elements. If a second argument is given, then each element is initialized to *fill*. Otherwise the initial contents of each element is unspecified.

`(vector` *obj ...*`)` procedure

Returns a newly allocated vector whose elements contain the given arguments. Analogous to `list`.

```
(vector 'a 'b 'c)           ⟹  #(a b c)
```

`(vector-length` *vector*`)` procedure

Returns the number of elements in *vector* as an exact integer.

`(vector-ref` *vector k*`)` procedure

k must be a valid index of *vector*. `Vector-ref` returns the contents of element *k* of *vector*.

```
(vector-ref '#(1 1 2 3 5 8 13 21)
            5)
          ⟹  8
(vector-ref '#(1 1 2 3 5 8 13 21)
            (let ((i (round (* 2 (acos -1)))))
              (if (inexact? i)
                  (inexact->exact i)
                  i)))
          ⟹ 13
```

`(vector-set!` *vector k obj*`)` procedure

k must be a valid index of *vector*. `Vector-set!` stores *obj* in element *k* of *vector*. The value returned by `vector-set!` is unspecified.

```
(let ((vec (vector 0 '(2 2 2 2) "Anna")))
  (vector-set! vec 1 '("Sue" "Sue"))
  vec)
          ⟹  #(0 ("Sue" "Sue") "Anna")

(vector-set! '#(0 1 2) 1 "doe")
          ⟹  error  ; constant vector
```

`(vector->list` *vector*`)` procedure
`(list->vector` *list*`)` procedure

`Vector->list` returns a newly allocated list of the objects contained in the elements of *vector*. `List->vector` returns a newly created vector initialized to the elements of the list *list*.

```
(vector->list '#(dah dah didah))
          ⟹  (dah dah didah)
(list->vector '(dididit dah))
          ⟹  #(dididit dah)
```

`(vector->string` *string*`)` procedure
`(string->vector` *vector*`)` procedure

`Vector->string` returns a newly allocated string of the objects contained in the elements of *vector*, which must be characters. `String->vector` returns a newly created vector initialized to the elements of the string *string*.

`(vector-copy` *vector*`)` procedure

Returns a newly allocated copy of the given *vector*.

`(vector-fill!` *vector fill*`)` procedure

Stores *fill* in every element of *vector*. The value returned by `vector-fill!` is unspecified.

6.3.7. Bytevectors

Bytevectors are a disjoint type for representing blocks of binary data. Conceptually, bytevectors can be thought of as homogenous vectors of 8-bit bytes, but they typically occupy less space than a vector. A byte is an exact integer in the range [0..255].

The *length* of a bytevector is the number of elements that it contains. This number is a non-negative integer that is fixed when the bytevector is created. The *valid indexes* of a bytevector are the exact non-negative integers less than the length of the bytevector, starting at index zero as with vectors.

`(bytevector?` *obj*`)` procedure

Returns `#t` if *obj* is a bytevector. Otherwise, `#f` is returned.

`(make-bytevector` *k*`)` procedure
`(make-bytevector` *k byte*`)` procedure

`Make-bytevector` returns a newly allocated bytevector of length *k*. If *byte* is given, then all elements of the bytevector are initialized to *byte*, otherwise the contents of each element are unspecified.

`(bytevector-length` *bytevector*`)` procedure

Returns the length of *bytevector* in bytes as an exact integer.

`(bytevector-u8-ref` *bytevector k*`)` procedure

Returns the *k*th byte of *bytevector*.

`(bytevector-u8-set!` *bytevector k byte*`)` procedure

Stores *byte* as the *k*th byte of *bytevector*. The value returned by `bytevector-u8-set!` is unspecified.

`(bytevector-copy` *bytevector*`)` procedure

Returns a newly allocated bytevector containing the same bytes as *bytevector*.

`(bytevector-copy!` *from to*`)` procedure

Copy the bytes of bytevector *from* to bytevector *to*, which must not be shorter. The value returned by `bytevector-copy!` is unspecified.

`(bytevector-copy-partial` *bytevector start end*`)` procedure

Returns a newly allocated bytevector containing the bytes in *bytevector* between *start* (inclusive) and *end* (exclusive).

`(bytevector-copy-partial!` *from start end to at*`)` procedure

Copy the bytes of bytevector *from* between *start* and *end* to bytevector *to*, starting at *at*. The order in which bytes are copied is unspecified, except that if the source and destination overlap, copying takes place as if the source is first copied into a temporary bytevector and then into the destination. This can be achieved without allocating storage by making sure to copy in the correct direction in such circumstances.

The inequality `(>= (- (bytevector-length` *to*`) at) (-` *end start*`))` must be true. The value returned by `partial-bytevector-copy!` is unspecified.

6.4. Control features

This chapter describes various primitive procedures which control the flow of program execution in special ways. The `procedure?` predicate is also described here.

`(procedure?` *obj*`)` procedure

Returns `#t` if *obj* is a procedure, otherwise returns `#f`.

```
(procedure? car)            ⟹  #t
(procedure? 'car)           ⟹  #f
(procedure? (lambda (x) (* x x)))
                            ⟹  #t
(procedure? '(lambda (x) (* x x)))
                            ⟹  #f
(call-with-current-continuation procedure?)
                            ⟹  #t
```

`(apply` *proc* arg_1 ... *args*`)` procedure

Proc must be a procedure and *args* must be a list. Calls *proc* with the elements of the list `(append (list` arg_1 ...`)` *args*`)` as the actual arguments.

```
(apply + (list 3 4))        ⟹  7

(define compose
  (lambda (f g)
    (lambda args
      (f (apply g args)))))

((compose sqrt *) 12 75)    ⟹  30
```

`(map` *proc* $list_1$ $list_2$ ...`)` procedure

The *list*s must be lists, and *proc* must be a procedure taking as many arguments as there are *list*s and returning a single value. If more than one *list* is given and not all lists have the same length, `map` terminates when the shortest list runs out. `Map` applies *proc* element-wise to the elements of the *list*s and returns a list of the results, in order. It is an error for *proc* to mutate any of the lists. The dynamic order in which *proc* is applied to the elements of the *list*s is unspecified. If multiple returns occur from `map`, the values returned by earlier returns are not mutated.

```
(map cadr '((a b) (d e) (g h)))
          ⟹  (b e h)

(map (lambda (n) (expt n n))
     '(1 2 3 4 5))
          ⟹  (1 4 27 256 3125)

(map + '(1 2 3) '(4 5 6))   ⟹  (5 7 9)

(let ((count 0))
  (map (lambda (ignored)
         (set! count (+ count 1))
         count)
       '(a b)))               ⟹  (1 2) or (2 1)
```

`(string-map` *proc* $string_1$ $string_2$ ...`)` procedure

The *string*s must be strings, and *proc* must be a procedure taking as many arguments as there are *string*s and returning a single value. If more than one *string* is given and not all strings have the same length, `string-map` terminates when the shortest list runs out. `String-map` applies *proc* element-wise to the elements of the *string*s and returns a string of the results, in order. The dynamic order in which *proc* is applied to the elements of the *string*s is unspecified. If multiple returns occur from `string-map`, the values returned by earlier returns are not mutated.

```
(string-map char-foldcase "AbdEgH")
          ⟹  "abdegh"

(string-map
 (lambda (c)
   (integer->char (+ 1 (char->integer c))))
 "HAL")
          ⟹  "IBM"

(string-map
 (lambda (c k)
   (if (eqv? k #\u)
       (char-upcase c)
       (char-downcase c)))
 "studlycaps"
 "ululululul")
          ⟹   "StUdLyCaPs"
```

`(vector-map` *proc vector*$_1$ *vector*$_2$ ...`)` procedure

The *vector*s must be vectors, and *proc* must be a procedure taking as many arguments as there are *vector*s and returning a single value. If more than one *vector* is given and not all vectors have the same length, `vector-map` terminates when the shortest list runs out. `Vector-map` applies *proc* element-wise to the elements of the *vector*s and returns a vector of the results, in order. The dynamic order in which *proc* is applied to the elements of the *vector*s is unspecified. If multiple returns occur from `vector-map`, the values returned by earlier returns are not mutated.

```
(vector-map cadr '#((a b) (d e) (g h)))
          ⟹  #(b e h)

(vector-map (lambda (n) (expt n n))
            '#(1 2 3 4 5))
          ⟹  #(1 4 27 256 3125)

(vector-map + '#(1 2 3) '#(4 5 6) '#(5 7 9))

(let ((count 0))
  (vector-map
   (lambda (ignored)
     (set! count (+ count 1))
     count)
   '#(a b)))                ⟹  #(1 2) or #(2 1)
```

`(for-each` *proc list*$_1$ *list*$_2$ ...`)` procedure

The arguments to `for-each` are like the arguments to `map`, but `for-each` calls *proc* for its side effects rather than for its values. Unlike `map`, `for-each` is guaranteed to call *proc* on the elements of the *list*s in order from the first element(s) to the last, and the value returned by `for-each` is unspecified. It is an error for *proc* to mutate any of the lists. If more than one *list* is given and not all lists have the same length, `for-each` terminates when the shortest list runs out.

```
(let ((v (make-vector 5)))
  (for-each (lambda (i)
              (vector-set! v i (* i i)))
            '(0 1 2 3 4))
  v)                        ⟹  #(0 1 4 9 16)
```

`(string-for-each` *proc string*$_1$ *string*$_2$ ...`)` procedure

The arguments to `string-for-each` are like the arguments to `string-map`, but `string-for-each` calls *proc* for its side effects rather than for its values. Unlike `string-map`, `string-for-each` is guaranteed to call *proc* on the elements of the *list*s in order from the first element(s) to the last, and the value returned by `string-for-each` is unspecified. If more than one *string* is given and not all strings have the same length, `string-for-each` terminates when the shortest string runs out.

```
(let ((v '()))
  (string-for-each
   (lambda (c) (set! v (cons (char->integer c) v))
   "abcde")
  v)                        ⟹  (101 100 99 98 97)
```

`(vector-for-each` *proc vector*$_1$ *vector*$_2$ ...`)` procedure

The arguments to `vector-for-each` are like the arguments to `vector-map`, but `vector-for-each` calls *proc* for its side effects rather than for its values. Unlike `vector-map`, `vector-for-each` is guaranteed to call *proc* on the elements of the *vector*s in order from the first element(s) to the last, and the value returned by `vector-for-each` is unspecified. If more than one *vector* is given and not all vectors have the same length, `vector-for-each` terminates when the shortest vector runs out.

```
(let ((v (make-list 5)))
  (vector-for-each
   (lambda (i) (list-set! v i (* i i)))
   '#(0 1 2 3 4))
  v)                        ⟹  (0 1 4 9 16)
```

`(force` *promise*`)` lazy module procedure

Forces the value of *promise* (see `delay` and `lazy`, section 4.2.5). If a value which is not a promise has already been computed, this value is returned. Otherwise, the promise is first evaluated, then overwritten by the obtained promise or value, and then `force` is again applied (iteratively) to the promise.

```
(force (delay (+ 1 2)))     ⟹  3
(let ((p (delay (+ 1 2))))
  (list (force p) (force p)))
                            ⟹  (3 3)
```

```
(define integers
  (letrec ((next
            (lambda (n)
              (delay (cons n (next (+ n 1)))))))
    (next 0)))
(define head
  (lambda (stream) (car (force stream))))
(define tail
  (lambda (stream) (cdr (force stream))))

(head (tail (tail integers)))
                              ⟹  2
```

The following example is a mechanical transformation of a lazy stream-filtering algorithm into Scheme. Each call to a constructor is wrapped in `delay`, and each argument passed to a deconstructor is wrapped in `force`. The use of `(lazy ...)` instead of `(delay (force ...))` around the body of the procedure ensures that an ever-growing sequence of pending promises does not build up until the heap is exhausted.

```
(define (stream-filter p? s)
  (lazy
   (if (null? (force s)) (delay '())
       (let ((h (car (force s)))
             (t (cdr (force s))))
         (if (p? h)
             (delay (cons h (stream-filter p? t)))
             (stream-filter p? t))))))

(head (tail (tail (stream-filter? odd? integers))))
                              ⟹ 5
```

`Delay`, `lazy`, and `force` are mainly intended for programs written in functional style. The following examples should not be considered to illustrate good programming style, but they illustrate the property that only one value is computed for a promise, no matter how many times it is forced.

```
(define count 0)
(define p
  (delay (begin (set! count (+ count 1))
                (if (> count x)
                    count
                    (force p)))))
(define x 5)
p                             ⟹  a promise
(force p)                     ⟹  6
p                             ⟹  a promise, still
(begin (set! x 10)
       (force p))             ⟹  6
```

Here is a possible implementation of `delay`, `force` and `lazy`. We define the expression

`(lazy` ⟨expression⟩`)`

to have the same meaning as the procedure call

`(make-promise #f (lambda ()` ⟨expression⟩`))`

as follows

```
(define-syntax lazy
  (syntax-rules ()
    ((lazy expression)
     (make-promise #f (lambda () expression)))))
```

and we define the expression

`(delay` ⟨expression⟩`)`

to have the same meaning as the following `lazy` form:

`(lazy (make-promise #t` ⟨expression⟩`))`

as follows

```
(define-syntax delay
  (syntax-rules ()
    ((delay expression)
     (lazy (make-promise #t expression)))))
```

where `make-promise` is defined as follows:

```
(define make-promise
  (lambda (done? proc)
    (list (cons done? proc))))
```

Finally, we define `force` to iteratively call the procedure expressions in promises using a trampoline technique until a non-lazy result (i.e. a value created by `delay` instead of `lazy`) is returned, as follows:

```
(define (force promise)
  (if (promise-done? promise)
      (promise-value promise)
      (let ((promise* ((promise-value promise))))
        (unless (promise-done? promise)
          (promise-update! promise* promise))
        (force promise))))
```

with the following promise accessors:

```
(define promise-done?
  (lambda (x) (car (car x))))
(define promise-value
  (lambda (x) (cdr (car x))))
(define promise-update!
  (lambda (new old)
    (set-car! (car old) (promise-done? new))
    (set-cdr! (car old) (promise-value new))
    (set-car! new (car old))))
```

Various extensions to this semantics of `delay`, `force` and `lazy` are supported in some implementations:

- Calling `force` on an object that is not a promise may simply return the object.
- It may be the case that there is no means by which a promise can be operationally distinguished from its forced value. That is, expressions like the following may evaluate to either `#t` or to `#f`, depending on the implementation:

```
(eqv? (delay 1) 1)           ⟹  unspecified
(pair? (delay (cons 1 2)))   ⟹  unspecified
```

- Some implementations may implement "implicit forcing," where the value of a promise is forced by primitive procedures like `cdr` and `+`:

```
(+ (delay (* 3 7)) 13)       ⟹  34
```

`(call-with-current-continuation` *proc*`)` procedure
`(call/cc` *proc*`)` procedure

Proc must be a procedure of one argument. The procedure `call-with-current-continuation` (or its equivalent abbreviation `call/cc`) packages up the current continuation (see the rationale below) as an "escape procedure" and passes it as an argument to *proc*. The escape procedure is a Scheme procedure that, if it is later called, will abandon whatever continuation is in effect at that later time and will instead use the continuation that was in effect when the escape procedure was created. Calling the escape procedure may cause the invocation of *before* and *after* thunks installed using `dynamic-wind`.

The escape procedure accepts the same number of arguments as the continuation to the original call to `call-with-current-continuation`. Except for continuations created by the `call-with-values` procedure (including the initialization expressions of `let-values` and `let*-values` syntax forms), all continuations take exactly one value. The effect of passing no value or more than one value to continuations that were not created by `call-with-values` is unspecified.

However, the continuations of all non-final expressions within a sequence of expressions, such as in `lambda`, `case-lambda`, `begin`, `let`, `let*`, `letrec`, `letrec*`, `let-values`, `let*-values`, `let-syntax`, `letrec-syntax`, `parameterize`, `guard`, `case`, `cond`, `when` and `unless` forms, take an arbitrary number of values, because they discard the values passed to them in any event.

The escape procedure that is passed to *proc* has unlimited extent just like any other procedure in Scheme. It may be stored in variables or data structures and may be called as many times as desired.

The following examples show only the most common ways in which `call-with-current-continuation` is used. If all real uses were as simple as these examples, there would be no need for a procedure with the power of `call-with-current-continuation`.

```
(call-with-current-continuation
  (lambda (exit)
    (for-each (lambda (x)
                (if (negative? x)
                    (exit x)))
              '(54 0 37 -3 245 19))
    #t))                          ⟹  -3

(define list-length
  (lambda (obj)
    (call-with-current-continuation
      (lambda (return)
        (letrec ((r
                  (lambda (obj)
                    (cond ((null? obj) 0)
                          ((pair? obj)
                           (+ (r (cdr obj)) 1))
                          (else (return #f))))))
          (r obj))))))

(list-length '(1 2 3 4))    ⟹  4

(list-length '(a b . c))    ⟹  #f
```

Rationale:

A common use of `call-with-current-continuation` is for structured, non-local exits from loops or procedure bodies, but in fact `call-with-current-continuation` is extremely useful for implementing a wide variety of advanced control structures.

Whenever a Scheme expression is evaluated there is a *continuation* wanting the result of the expression. The continuation represents an entire (default) future for the computation. If the expression is evaluated at top level, for example, then the continuation might take the result, print it on the screen, prompt for the next input, evaluate it, and so on forever. Most of the time the continuation includes actions specified by user code, as in a continuation that will take the result, multiply it by the value stored in a local variable, add seven, and give the answer to the top level continuation to be printed. Normally these ubiquitous continuations are hidden behind the scenes and programmers do not think much about them. On rare occasions, however, a programmer may need to deal with continuations explicitly. `Call-with-current-continuation` allows Scheme programmers to do that by creating a procedure that acts just like the current continuation.

Most programming languages incorporate one or more special-purpose escape constructs with names like `exit`, `return`, or even `goto`. In 1965, however, Peter Landin [18] invented a general purpose escape operator called the J-operator. John Reynolds [26] described a simpler but equally powerful construct in 1972. The `catch` special form described by Sussman and Steele in the 1975 report on Scheme is exactly the same as Reynolds's construct, though its name came from a less general construct in MacLisp. Several Scheme implementors noticed that the full power of the `catch` construct could be provided by a procedure instead of by a special syntactic construct, and the name `call-with-current-continuation` was coined in 1982. This name is descriptive, but opinions differ on the merits of such a long name, and some people use the name `call/cc` instead.

`(values` *obj* ...`)` procedure

Delivers all of its arguments to its continuation. Except for continuations created by the `call-with-values` procedure, all continuations take exactly one value. `Values` might be defined as follows:

```
(define (values . things)
  (call-with-current-continuation
    (lambda (cont) (apply cont things))))
```

`(call-with-values` *producer consumer*`)` procedure

Calls its *producer* argument with no values and a continuation that, when passed some values, calls the *consumer* procedure with those values as arguments. The continuation for the call to *consumer* is the continuation of the call to `call-with-values`.

```
(call-with-values (lambda () (values 4 5))
                  (lambda (a b) b))
                            ⟹  5

(call-with-values * -)      ⟹  -1
```

`(dynamic-wind` *before thunk after*`)` procedure

Calls *thunk* without arguments, returning the result(s) of this call. *Before* and *after* are called, also without arguments, as required by the following rules (note that in the absence of calls to continuations captured using `call-with-current-continuation` the three arguments are called once each, in order). *Before* is called whenever execution enters the dynamic extent of the call to *thunk* and *after* is called whenever it exits that dynamic extent. The dynamic extent of a procedure call is the period between when the call is initiated and when it returns. *Before* and *after* are excluded from the dynamic extent. In Scheme, because of `call-with-current-continuation`, the dynamic extent of a call may not be a single, connected time period. It is defined as follows:

- The dynamic extent is entered when execution of the body of the called procedure begins.
- The dynamic extent is also entered when execution is not within the dynamic extent and a continuation is invoked that was captured (using `call-with-current-continuation`) during the dynamic extent.
- It is exited when the called procedure returns.
- It is also exited when execution is within the dynamic extent and a continuation is invoked that was captured while not within the dynamic extent.

If a second call to `dynamic-wind` occurs within the dynamic extent of the call to *thunk* and then a continuation is invoked in such a way that the *after*s from these two invocations of `dynamic-wind` are both to be called, then the *after* associated with the second (inner) call to `dynamic-wind` is called first.

If a second call to `dynamic-wind` occurs within the dynamic extent of the call to *thunk* and then a continuation is invoked in such a way that the *before*s from these two invocations of `dynamic-wind` are both to be called, then the *before* associated with the first (outer) call to `dynamic-wind` is called first.

If invoking a continuation requires calling the *before* from one call to `dynamic-wind` and the *after* from another, then the *after* is called first.

The effect of using a captured continuation to enter or exit the dynamic extent of a call to *before* or *after* is undefined.

```
(let ((path '())
      (c #f))
  (let ((add (lambda (s)
               (set! path (cons s path)))))
    (dynamic-wind
      (lambda () (add 'connect))
      (lambda ()
        (add (call-with-current-continuation
               (lambda (c0)
                 (set! c c0)
                 'talk1))))
      (lambda () (add 'disconnect)))
    (if (< (length path) 4)
        (c 'talk2)
        (reverse path))))

        ⟹ (connect talk1 disconnect
            connect talk2 disconnect)
```

`(make-parameter` *init*`)` procedure
`(make-parameter` *init converter*`)` procedure

Returns a new parameter object which is associated with the value returned by the call `(`*converter init*`)`. If the conversion procedure *converter* is not specified the identity function is used instead.

A parameter object is a procedure which accepts zero arguments and returns its associated value. The associated value may be changed with `parameterize`. The effect of passing arguments to a parameter object is explicitly implementation-dependent.

Here is a possible implementation of `make-parameter` and `parameterize` suitable for an implementation with no threads. Parameter objects are implemented here as procedures, using two arbitrary unique objects `<param-set!>` and `<param-convert>`:

```
(define (make-parameter init . o)
  (let* ((converter (if (pair? o)
                        (car o)
                        (lambda (x) x)))
         (value (converter init)))
    (lambda args
      (if (pair? args)
          (cond
           ((eq? (car args) <param-set!>)
            (set! value (cadr args)))
           ((eq? (car args) <param-convert>)
            converter)
           (else
            (error "bad parameter syntax")))
          value))))
```

`Parameterize` then uses `dynamic-wind` to dynamically rebind the associated value:

```
(define-syntax parameterize
  (syntax-rules ()
    ((parameterize ("step")
                   ((param value p old new) ...)
                   ()
                   body)
     (let ((p param) ...)
       (let ((old (p)) ...
             (new ((p <param-convert>) value)) ...)
         (dynamic-wind
          (lambda () (p <param-set!> new) ...)
          (lambda () . body)
          (lambda () (p <param-set!> old) ...)))))
    ((parameterize ("step")
                   args
                   ((param value) . rest)
                   body)
     (parameterize ("step")
                   ((param value p old new) . args)
                   rest
                   body))
    ((parameterize ((param value) ...) . body)
     (parameterize ("step")
                   ()
                   ((param value) ...)
                   body))))
```

Parameter objects can be used to specify configurable settings for a computation without the need to explicitly pass the value to every procedure in the call chain.

```
(define radix
  (make-parameter
   10
   (lambda (x)
     (if (and (integer? x) (<= 2 x 16))
         x
         (error "invalid radix")))))

(define (f n) (number->string n (radix)))

(f 12)                          ⟹ "12"
(parameterize ((radix 16))
  (f 12))                       ⟹ "C"
(f 12)                          ⟹ "12"

(radix 16)                      ⟹ unspecified

(parameterize ((radix 0))
  (f 12))                       ⟹ error
```

6.5. Exceptions

This section describes Scheme's exception-handling and exception-raising procedures. See also 4.2.7 for the `guard` syntax.

Exception handlers are one-argument procedures that determine the action the program takes when an exceptional situation is signalled. The system implicitly maintains a current exception handler.

The program raises an exception by invoking the current exception handler, passing it an object encapsulating information about the exception. Any procedure accepting one argument may serve as an exception handler and any object may be used to represent an exception.

`(with-exception-handler` *handler thunk*`)` procedure

Handler must be a procedure and should accept one argument. *Thunk* must be a procedure that accepts zero arguments. The `with-exception-handler` procedure returns the results of invoking *thunk*. *Handler* is installed as the current exception handler for the dynamic extent (as determined by `dynamic-wind`) of the invocation of *thunk*.

`(raise` *obj*`)` procedure

Raises a exception by invoking the current exception handler on *obj*. The handler is called with a continuation whose dynamic extent is that of the call to `raise`, except that the current exception handler is the one that was in place when the handler being called was installed. If the handler returns, an exception is raised in the same dynamic extent as the handler.

`(raise-continuable` *obj*`)` procedure

Raises an exception by invoking the current exception handler on *obj*. The handler is called with a continuation that is equivalent to the continuation of the call to `raise-continuable`, with these two exceptions: (1) the current exception handler is the one that was in place when the handler being called was installed, and (2) if the handler being called returns, then it will again become the

current exception handler. If the handler returns, the values it returns become the values returned by the call to `raise-continuable`.

`(error` *message obj ...*`)` procedure

Message should be a string. `raise` on a newly created implementation-defined object which encapsulates the information provided by *message*, as well as any *obj*s, known as the irritants. The procedure `error-object?` must return `#t` on such objects.

```
(with-exception-handler
  (lambda (con)
    (cond
      ((string? con)
       (display con))
      (else
       (display "a warning has been issued")))
    42)
  (lambda ()
    (+ (raise-continuable "should be a number")
       23)))
  prints: should be a number
                              ⟹ 65
```

`(error-object?` *obj*`)` procedure

Returns `#t` if *obj* is an object created by `error`, otherwise returns `#f`.

`(error-object-message` *error-object*`)` procedure

Returns the message encapsulated by *error-object*.

`(error-object-irritants` *error-object*`)` procedure

Returns a list of the irritants encapsulated by *error-object*.

6.6. Eval

`(eval` *expression environment-specifier*`)`
eval module procedure

Evaluates *expression* in the specified environment and returns its value. *Expression* must be a valid Scheme expression represented as data, and *environment-specifier* must be a value returned by one of the four procedures described below. Implementations may extend `eval` to allow non-expression programs (definitions) as the first argument and to allow other values as environments, with the restriction that `eval` is not allowed to create new bindings in the environments returned by `null-environment` or `scheme-report-environment`.

```
(eval '(* 7 3) (scheme-report-environment 5))
                                ⟹  21

(let ((f (eval '(lambda (f x) (f x x))
               (null-environment 5))))
  (f + 10))
                                ⟹  20
```

`(scheme-report-environment` *version*`)`
eval module procedure

`(null-environment` *version*`)` eval module procedure

Version must be the exact integer `7`, corresponding to this revision of the Scheme report (the Revised[7] Report on Scheme). `Scheme-report-environment` returns a specifier for an environment that is empty except for all bindings defined in this report that are either required or both optional and supported by the implementation. `Null-environment` returns a specifier for an environment that is empty except for the (syntactic) bindings for all syntactic keywords defined in this report that are either required or both optional and supported by the implementation.

Other values of *version* can be used to specify environments matching past revisions of this report, but their support is not required. If *version* is neither `7` nor another value supported by the implementation, an error is signalled.

The effect of assigning (through the use of `eval`) a variable bound in a `scheme-report-environment` (for example `car`) is unspecified. Thus the environments specified by `scheme-report-environment` may be immutable.

`(environment` *list*$_1$ *...*`)` eval module procedure

This procedure returns a specifier for the environment that results by starting with an empty environment and then importing each *list*, considered as an import set, into it. The bindings of the environment represented by the specifier are immutable.

`(interaction-environment)` repl module procedure

This procedure returns a specifier for the environment that contains implementation-defined bindings, typically a superset of those listed in the report. The intent is that this procedure will return the environment in which the implementation would evaluate expressions dynamically typed by the user.

6.7. Input and output

6.7.1. Ports

Ports represent input and output devices. To Scheme, an input port is a Scheme object that can deliver data upon

command, while an output port is a Scheme object that can accept data. Whether the input and output port types are disjoint is implementation-dependent.

Different *port types* operate on different data. Scheme implementations are required to support *character ports* and *binary ports*, but may also provide other port types.

A character port supports reading or writing of individual characters from or to a backing store containing characters using `read-char` and `write-char` below, as well as operations defined in terms of characters such as `read` and `write`.

A binary port supports reading or writing of individual bytes from or to a backing store containing bytes using `read-u8` and `write-u8`, below. Whether the character and binary port types are disjoint is implementation-dependent.

Ports can be used to access files, devices, and similar things on the host system in which the Scheme program is running.

`(call-with-input-file` *string proc*`)` file module procedure
`(call-with-output-file` *string proc*`)` file module procedure

Proc should be a procedure that accepts one argument. For `call-with-input-file`, the file named by *string* should already exist; for `call-with-output-file`, the effect is unspecified if the file already exists. These procedures call *proc* with one argument: the character port obtained by opening the named file for input or output as if by `open-input-file` or `open-output-file`. If the file cannot be opened, an error is signalled. If *proc* returns, then the port is closed automatically and the value(s) yielded by the *proc* is(are) returned. If *proc* does not return, then the port will not be closed automatically unless it is possible to prove that the port will never again be used for a read or write operation. *Rationale:* Because Scheme's escape procedures have unlimited extent, it is possible to escape from the current continuation but later to escape back in. If implementations were permitted to close the port on any escape from the current continuation, then it would be impossible to write portable code using both `call-with-current-continuation` and `call-with-input-file` or `call-with-output-file`.

`(call-with-port` *port proc*`)` io module procedure

Proc must accept one argument. The `call-with-port` procedure calls *proc* with *port* as an argument. If *proc* returns, *port* is closed automatically and the values returned by *proc* are returned.

`(input-port?` *obj*`)` io module procedure
`(output-port?` *obj*`)` io module procedure
`(character-port?` *obj*`)` io module procedure
`(binary-port?` *obj*`)` io module procedure
`(port?` *obj*`)` io module procedure

Returns `#t` if *obj* is an input port, output port, character port, binary port, or any kind of port, respectively, otherwise returns `#f`.

`(port-open?` *port*`)` io module procedure

Returns `#t` if *port* is still open and capable of performing input or output, and `#f` otherwise.

`(current-input-port)` io module procedure
`(current-output-port)` io module procedure
`(current-error-port)` io module procedure

Returns the current default input port, output port, or error port (an output port), respectively. These are parameter objects, which can be overridden with `parameterize` (see section 6.4). The initial bindings for each of these are bound to system defined binary ports.

`(with-input-from-file` *string thunk*`)` file module procedure
`(with-output-to-file` *string thunk*`)` file module procedure

Thunk should be a procedure of no arguments. For `with-input-from-file`, the file named by *string* should already exist; for `with-output-to-file`, the effect is unspecified if the file already exists. The file is opened for input or output as if by `open-input-file` or `open-output-file`, and the return port is made the default value returned by `current-input-port` or `current-output-port` (and is used by `(read)`, `(write` *obj*`)`, and so forth). The *thunk* is then called with no arguments. When the *thunk* returns, the port is closed and the previous default is restored. `With-input-from-file` and `with-output-to-file` return(s) the value(s) yielded by *thunk*. If an escape procedure is used to escape from the continuation of these procedures, their behavior is implementation-dependent.

`(open-input-file` *string*`)` file module procedure
`(open-binary-input-file` *string*`)` file module procedure

Takes a *string* for an existing file and returns a character input port or binary input port capable of delivering data from the file. If the file cannot be opened, an error is signalled.

`(open-output-file` *string*`)` file module procedure
`(open-binary-output-file` *string*`)` file module procedure

Takes a *string* naming an output file to be created and returns a character output port or binary output port capable of writing data to a new file by that name. If the file

cannot be opened, an error is signalled. If a file with the given name already exists, the effect is unspecified.

`(close-port` *port*`)`
io module procedure`(close-input-port` *port*`)`
io module procedure
`(close-output-port` *port*`)` io module procedure

Closes the file associated with *port*, rendering the *port* incapable of delivering or accepting data. It is an error to apply the first two procedures to a port which is not an input or output port, respectively. Scheme implementations may provide ports which are simultaneously input and output ports, such as sockets; the `close-input-port` and `close-output-port` procedures may be used to close the input and output sides of the port independently.

These routines have no effect if the file has already been closed. The value returned is unspecified.

`(open-input-string` *string*`)` io module procedure

Takes a string and returns a character input port that delivers characters from the string.

`(open-output-string)` io module procedure

Returns a character output port that will accumulate characters for retrieval by `get-output-string`.

`(get-output-string` *port*`)` io module procedure

Given an output port created by `open-output-string`, returns a string consisting of the characters that have been output to the port so far.

`(open-input-bytevector` *bytevector*`)`
io module procedure

Takes a bytevector and returns a binary input port that delivers bytes from the bytevector.

`(open-output-bytevector)` io module procedure

Returns a binary output port that will accumulate bytes for retrieval by `get-output-bytevector`.

`(get-output-bytevector` *port*`)` io module procedure

Given an output port created by `open-output-bytevector`, returns a bytevector consisting of the bytes that have been output to the port so far.

6.7.2. Input

`(read)` read module procedure
`(read` *port*`)` read module procedure

`Read` converts external representations of Scheme objects into the objects themselves. That is, it is a parser for the nonterminal ⟨datum⟩ (see sections 7.1.2 and 6.3.2). `Read` returns the next object parsable from the given character input *port*, updating *port* to point to the first character past the end of the external representation of the object.

If an end of file is encountered in the input before any characters are found that can begin an object, then an end of file object is returned. The port remains open, and further attempts to read will also return an end of file object. If an end of file is encountered after the beginning of an object's external representation, but the external representation is incomplete and therefore not parsable, an error is signalled.

The *port* argument may be omitted, in which case it defaults to the value returned by `current-input-port`. It is an error to read from a closed port.

`(read-char)` io module procedure
`(read-char` *port*`)` io module procedure

Returns the next character available from the character input *port*, updating the *port* to point to the following character. If no more characters are available, an end of file object is returned. *Port* may be omitted, in which case it defaults to the value returned by `current-input-port`.

`(peek-char)` io module procedure
`(peek-char` *port*`)` io module procedure

Returns the next character available from the character input *port*, *without* updating the *port* to point to the following character. If no more characters are available, an end of file object is returned. *Port* may be omitted, in which case it defaults to the value returned by `current-input-port`.

Note: The value returned by a call to `peek-char` is the same as the value that would have been returned by a call to `read-char` with the same *port*. The only difference is that the very next call to `read-char` or `peek-char` on that *port* will return the value returned by the preceding call to `peek-char`. In particular, a call to `peek-char` on an interactive port will hang waiting for input whenever a call to `read-char` would have hung.

`(read-line)` io module procedure
`(read-line` *port*`)` io module procedure

Returns the next line of text available from the character input *port*, updating the *port* to point to the following character. If an end of line is read, a string containing all of the text up to (but not including) the end of line is returned, and the port is updated to point just past

the end of line. If an end of file is encountered before any linefeed character is read, but some characters have been read, a string containing those characters is returned. If an end of file is encountered before any characters are read, an end-of-file object is returned. For the purpose of this procedure, an end of line consists of either a linefeed character, a carriage return character, or a sequence of a carriage return character followed by a linefeed character. *Port* may be omitted, in which case it defaults to the value returned by `current-input-port`.

`(eof-object?` *obj*`)` io module procedure

Returns `#t` if *obj* is an end of file object, otherwise returns `#f`. The precise set of end of file objects will vary among implementations, but in any case no end of file object will ever be an object that can be read in using `read`.

`(char-ready?)` io module procedure
`(char-ready?` *port*`)` io module procedure

Returns `#t` if a character is ready on the character input *port* and returns `#f` otherwise. If `char-ready` returns `#t` then the next `read-char` operation on the given *port* is guaranteed not to hang. If the *port* is at end of file then `char-ready?` returns `#t`. *Port* may be omitted, in which case it defaults to the value returned by `current-input-port`.

Rationale: `Char-ready?` exists to make it possible for a program to accept characters from interactive ports without getting stuck waiting for input. Any input editors associated with such ports must ensure that characters whose existence has been asserted by `char-ready?` cannot be rubbed out. If `char-ready?` were to return `#f` at end of file, a port at end of file would be indistinguishable from an interactive port that has no ready characters.

`(read-u8)` io module procedure
`(read-u8` *port*`)` io module procedure

Returns the next byte available from the binary input *port*, updating the *port* to point to the following byte. If no more bytes are available, an end of file object is returned. *Port* may be omitted, in which case it defaults to the value returned by `current-input-port`.

`(peek-u8)` io module procedure
`(peek-u8` *port*`)` io module procedure

Returns the next byte available from the binary input *port*, *without* updating the *port* to point to the following byte. If no more bytes are available, an end of file object is returned. *Port* may be omitted, in which case it defaults to the value returned by `current-input-port`.

`(u8-ready?)` io module procedure
`(u8-ready?` *port*`)` io module procedure

Returns `#t` if a byte is ready on the binary input *port* and returns `#f` otherwise. If `u8-ready?` returns `#t` then the next `read-u8` operation on the given *port* is guaranteed not to hang. If the *port* is at end of file then `u8-ready?` returns `#t`. *Port* may be omitted, in which case it defaults to the value returned by `current-input-port`.

Note: If `u8-ready?` returns `#t`, a subsequent `read-char` operation may still hang.

6.7.3. Output

`(write` *obj*`)` write module procedure
`(write` *obj port*`)` write module procedure

Writes a written representation of *obj* to the given character output *port*. Strings that appear in the written representation are enclosed in doublequotes, and within those strings backslash and doublequote characters are escaped by backslashes. Symbols that contain non-ASCII characters are escaped either with inline hex escapes or with vertical bars. Character objects are written using the `#\` notation. Shared list structure is represented using labels. `Write` returns an unspecified value. The *port* argument may be omitted, in which case it defaults to the value returned by `current-output-port`.

`(write-simple` *obj*`)` write module procedure
`(write-simple` *obj port*`)` write module procedure

`Write-simple` is the same as `write`, except that shared structure is not represented using labels. This may cause `write-simple` not to terminate if *obj* contains circular structure.

`(display` *obj*`)` write module procedure
`(display` *obj port*`)` write module procedure

Writes a representation of *obj* to the given character output *port*. Strings that appear in the written representation are not enclosed in doublequotes, and no characters are escaped within those strings. Symbols are not escaped. Character objects appear in the representation as if written by `write-char` instead of by `write`. `Display` returns an unspecified value. The *port* argument may be omitted, in which case it defaults to the value returned by `current-output-port`.

Rationale: `Write` is intended for producing machine-readable output and `display` is for producing human-readable output.

`(newline)` io module procedure
`(newline` *port*`)` io module procedure

Writes an end of line to character output *port*. Exactly how this is done differs from one operating system to another.

Returns an unspecified value. The *port* argument may be omitted, in which case it defaults to the value returned by `current-output-port`.

`(write-char` *char*`)` io module procedure
`(write-char` *char port*`)` io module procedure

Writes the character *char* (not an external representation of the character) to the given character output *port* and returns an unspecified value. The *port* argument may be omitted, in which case it defaults to the value returned by `current-output-port`.

`(write-u8` *byte*`)` io module procedure
`(write-u8` *byte port*`)` io module procedure

Writes the *byte* to the given binary output *port* and returns an unspecified value. The *port* argument may be omitted, in which case it defaults to the value returned by `current-output-port`.

`(flush-output-port)` io module procedure
`(flush-output-port` *port*`)` io module procedure

Flushes any buffered output from the buffer of output-port to the underlying file or device and returns an unspecified value. The *port* argument may be omitted, in which case it defaults to the value returned by `current-output-port`.

6.7.4. System interface

Questions of system interface generally fall outside of the domain of this report. However, the following operations are important enough to deserve description here.

`(load` *filename*`)` load module procedure

An implementation-dependent operation is used to transform *filename* into the name of an existing file containing Scheme source code. The `load` procedure reads expressions and definitions from the file and evaluates them sequentially. It is unspecified whether the results of the expressions are printed. The `load` procedure does not affect the values returned by `current-input-port` and `current-output-port`. `Load` returns an unspecified value.

`(include` *filename*`)` load module syntax
`(include-ci` *filename*`)` load module syntax

`Include` and `include-ci` are similar to `load` except that they are syntax which expands into the expressions and definitions from the file as though wrapped in a `begin` form. Thus it can be used to include internal definitions and otherwise interact with the current lexical scope. In these forms the *filename* must be a string literal.

Rationale: For portability, `load` and `include` must operate on source files. Their operation on other kinds of files necessarily varies among implementations.

`(file-exists?` *filename*`)` file module procedure

Filename must be a string. The `file-exists?` procedure returns `#t` if the named file exists at the time the procedure is called, `#f` otherwise.

`(delete-file` *filename*`)` file module procedure

Filename must be a string. The `delete-file` procedure deletes the named file if it exists and can be deleted, and returns an unspecified value. If the file does not exist or cannot be deleted, an error is signalled.

`(command-line)` process-context module procedure

Returns the command line arguments passed to the process as a list of strings.

`(exit)` process-context module procedure
`(exit` *obj*`)` process-context module procedure

Exits the running program and communicates an exit value to the operating system. If no argument is supplied, the `exit` procedure should communicate to the operating system that the program exited normally. If an argument is supplied, the exit procedure should translate the argument into an appropriate exit value for the operating system. If obj is `#f`, the exit is assumed to be abnormal.

`(get-environment-variable` *name*`)`
process-context module procedure

Most operating systems provide each running process with an *environment* consisting of *environment variables*. Both the name and value of an environment variable are strings. `Get-environment-variable` returns the value of the environment variable *name*, or `#f` if the named environment variable is not found. `Get-environment-variable` may use locale-setting information to encode the name and decode the value of the environment variable. It is an error if `get-environment-variable` can't decode the value.

```
(get-environment-variable "PATH")
        ⟹ "/usr/local/bin:/usr/bin:/bin"
```

`(get-environment-variables)`
process-context module procedure

Returns the names and values of all the environment variables as an a-list, where the car of each entry is the name of an environment variable and the cdr is its value. The order of the list is unspecified.

```
(get-environment-variables)
          ⟹ (("USER" . "root") ("HOME" . "/"))
```

`(current-second)` time module procedure

Returns an inexact number representing time on the International Atomic Time (TAI) scale. The value 0.0 represents ten seconds after midnight on January 1, 1970 TAI (equivalent to midnight Universal Time) and the value 1.0 represents one TAI second later. High-accuracy values are not required; in particular, returning Coordinated Universal Time plus a suitable constant may be the best an implementation can do.

`(current-jiffy)` time module procedure

Returns an exact integer representing the number of jiffies (arbitrary elapsed time units) since an arbitrary epoch which may vary between runs of a program.

`(jiffies-per-second)` time module procedure

Returns an exact integer representing the number of jiffies per SI second. This value is an implementation-specified constant.

7. Formal syntax and semantics

This chapter provides formal descriptions of what has already been described informally in previous chapters of this report.

7.1. Formal syntax

This section provides a formal syntax for Scheme written in an extended BNF.

All spaces in the grammar are for legibility. Case is insignificant; for example, `#x1A` and `#X1a` are equivalent. ⟨empty⟩ stands for the empty string.

The following extensions to BNF are used to make the description more concise: ⟨thing⟩* means zero or more occurrences of ⟨thing⟩; and ⟨thing⟩$^+$ means at least one ⟨thing⟩.

7.1.1. Lexical structure

This section describes how individual tokens (identifiers, numbers, etc.) are formed from sequences of characters. The following sections describe how expressions and programs are formed from sequences of tokens.

⟨Intertoken space⟩ may occur on either side of any token, but not within a token.

Identifiers, dot, numbers, characters, and booleans must be terminated by a ⟨delimiter⟩ or by the end of the input.

The following four characters from the ASCII repertoire are reserved for future extensions to the language: `[ ] { }`

In addition to the identifier characters of the ASCII repertoire specified below, Scheme implementations may permit any additional repertoire of Unicode characters to be employed in symbols (and therefore identifiers), provided that each such character has a Unicode general category of Lu, Ll, Lt, Lm, Lo, Mn, Mc, Me, Nd, Nl, No, Pd, Pc, Po, Sc, Sm, Sk, So, or Co, or is U+200C or U+200D (the zero-width non-joiner and joiner, respectively, which are needed for correct spelling in Persian, Hindi, and other languages). No non-Unicode characters may be used explicitly (that is, other than by specifying an escape) in symbols or identifiers.

All Scheme implementations must permit the sequence `\x<hexdigits>;` to appear in Scheme symbols (and therefore identifiers). If the character with the given Unicode scalar value is supported by the implementation, identifiers containing such a sequence are equivalent to identifiers containing the corresponding character. The `symbol->string` procedure may return the actual character or the escape sequence at the implementation's option, but any leading zeros must be removed from the escape sequence.

⟨token⟩ ⟶ ⟨identifier⟩ | ⟨boolean⟩ | ⟨number⟩
    | ⟨character⟩ | ⟨string⟩
    | `(` | `)` | `#(` | `'` | `` ` `` | `,` | `,@` | `.`
⟨delimiter⟩ ⟶ ⟨whitespace⟩ | `(` | `)` | `"` | `;`
⟨intraline whitespace⟩ ⟶ ⟨space or tab⟩
⟨whitespace⟩ ⟶ ⟨intraline whitespace or newline or return⟩
⟨comment⟩ ⟶ `;` ⟨all subsequent characters up to a line break⟩
    | ⟨nested comment⟩
    | `#;` ⟨atmosphere⟩ ⟨datum⟩
⟨nested comment⟩ ⟶ `#|` ⟨comment text⟩ ⟨comment cont⟩* `|#`
⟨comment text⟩ ⟶ ⟨character sequence not containing `#|` or `|#`⟩
⟨comment cont⟩ ⟶ ⟨nested comment⟩ ⟨comment text⟩
⟨atmosphere⟩ ⟶ ⟨whitespace⟩ | ⟨comment⟩
⟨intertoken space⟩ ⟶ ⟨atmosphere⟩*

⟨identifier⟩ ⟶ ⟨initial⟩ ⟨subsequent⟩*
    | ⟨vertical bar⟩ ⟨symbol element⟩* ⟨vertical bar⟩
    | ⟨peculiar identifier⟩
⟨initial⟩ ⟶ ⟨letter⟩ | ⟨special initial⟩
    | ⟨inline hex escape⟩
⟨letter⟩ ⟶ `a` | `b` | `c` | `...` | `z`
    | `A` | `B` | `C` | `...` | `Z`
⟨special initial⟩ ⟶ `!` | `$` | `%` | `&` | `*` | `/` | `:` | `<` | `=`
    | `>` | `?` | `^` | `_` | `~`
⟨subsequent⟩ ⟶ ⟨initial⟩ | ⟨digit⟩
    | ⟨special subsequent⟩
⟨digit⟩ ⟶ `0` | `1` | `2` | `3` | `4` | `5` | `6` | `7` | `8` | `9`
⟨hex digit⟩ ⟶ ⟨digit⟩
    | `a` | `A` | `b` | `B` | `c` | `C` | `d` | `D` | `e` | `E` | `f` | `F`
⟨explicit sign⟩ ⟶ `+` | `-`
⟨special subsequent⟩ ⟶ ⟨explicit sign⟩ | `.` | `@`
⟨inline hex escape⟩ ⟶ `\x`⟨hex scalar value⟩`;`
⟨hex scalar value⟩ ⟶ ⟨hex digit⟩$^+$
⟨peculiar identifier⟩ ⟶ ⟨explicit sign⟩
    | ⟨explicit sign⟩ ⟨sign subsequent⟩ ⟨subsequent⟩*
    | ⟨explicit sign⟩ `.` ⟨dot subsequent⟩ ⟨subsequent⟩*
    | `.` ⟨non-digit⟩ ⟨subsequent⟩*
⟨non-digit⟩ ⟶ ⟨dot subsequent⟩ | ⟨explicit sign⟩
⟨dot subsequent⟩ ⟶ ⟨sign subsequent⟩ | `.`
⟨sign subsequent⟩ ⟶ ⟨initial⟩ | ⟨explicit sign⟩ | `@`
⟨symbol element⟩ ⟶
    ⟨any character other than ⟨vertical bar⟩ or \⟩
⟨syntactic keyword⟩ ⟶ ⟨expression keyword⟩
    | `else` | `=>` | `define`
    | `unquote` | `unquote-splicing`
⟨expression keyword⟩ ⟶ `quote` | `lambda` | `if`
    | `set!` | `begin` | `cond` | `and` | `or` | `case`
    | `let` | `let*` | `letrec` | `do` | `delay`
    | `quasiquote`

⟨variable⟩ ⟶ ⟨any ⟨identifier⟩ that isn't also a ⟨syntactic keyword⟩⟩

⟨boolean⟩ ⟶ `#t` | `#f`
⟨character⟩ ⟶ `#\` ⟨any character⟩
    | `#\` ⟨character name⟩
    | `#\x`⟨hex scalar value⟩
⟨character name⟩ ⟶ `null` | `alarm` | `backspace` | `tab`
    | `newline` | `return` | `escape` | `space` | `delete`

⟨string⟩ ⟶ `"` ⟨string element⟩* `"`
⟨string element⟩ ⟶ ⟨any character other than " or \⟩
    | `\a` | `\b` | `\t` | `\n` | `\r` | `\"` | `\\`
    | `\`⟨intraline whitespace⟩⟨line ending⟩ ⟨intraline whitespace⟩
    | ⟨inline hex escape⟩
⟨bytevector⟩ ⟶ `#u8(`⟨byte⟩*`)`
⟨byte⟩ ⟶ ⟨any exact integer between 0 and 255⟩

⟨number⟩ ⟶ ⟨num 2⟩ | ⟨num 8⟩
    | ⟨num 10⟩ | ⟨num 16⟩

The following rules for ⟨num R⟩, ⟨complex R⟩, ⟨real R⟩, ⟨ureal R⟩, ⟨uinteger R⟩, and ⟨prefix R⟩ should be replicated for $R = 2, 8, 10$, and 16. There are no rules for ⟨decimal 2⟩, ⟨decimal 8⟩, and ⟨decimal 16⟩, which means that numbers containing decimal points or exponents must be in decimal radix.

⟨num R⟩ ⟶ ⟨prefix R⟩ ⟨complex R⟩
⟨complex R⟩ ⟶ ⟨real R⟩ | ⟨real R⟩ `@` ⟨real R⟩
    | ⟨real R⟩ `+` ⟨ureal R⟩ `i` | ⟨real R⟩ `-` ⟨ureal R⟩ `i`
    | ⟨real R⟩ `+ i` | ⟨real R⟩ `- i`
    | `+` ⟨ureal R⟩ `i` | `-` ⟨ureal R⟩ `i` | `+ i` | `- i`
⟨real R⟩ ⟶ ⟨sign⟩ ⟨ureal R⟩
    | ⟨infinity⟩
⟨ureal R⟩ ⟶ ⟨uinteger R⟩
    | ⟨uinteger R⟩ `/` ⟨uinteger R⟩
    | ⟨decimal R⟩
⟨decimal 10⟩ ⟶ ⟨uinteger 10⟩ ⟨suffix⟩
    | `.` ⟨digit 10⟩$^+$ `#`* ⟨suffix⟩
    | ⟨digit 10⟩$^+$ `.` ⟨digit 10⟩* `#`* ⟨suffix⟩
    | ⟨digit 10⟩$^+$ `#`$^+$ `.` `#`* ⟨suffix⟩
⟨uinteger R⟩ ⟶ ⟨digit R⟩$^+$ `#`*
⟨prefix R⟩ ⟶ ⟨radix R⟩ ⟨exactness⟩
    | ⟨exactness⟩ ⟨radix R⟩
⟨infinity⟩ ⟶ `+inf.0` | `-inf.0` | `+nan.0`

⟨suffix⟩ ⟶ ⟨empty⟩
    | ⟨exponent marker⟩ ⟨sign⟩ ⟨digit 10⟩$^+$
⟨exponent marker⟩ ⟶ `e` | `s` | `f` | `d` | `l`
⟨sign⟩ ⟶ ⟨empty⟩ | `+` | `-`
⟨exactness⟩ ⟶ ⟨empty⟩ | `#i` | `#e`
⟨radix 2⟩ ⟶ `#b`
⟨radix 8⟩ ⟶ `#o`
⟨radix 10⟩ ⟶ ⟨empty⟩ | `#d`
⟨radix 16⟩ ⟶ `#x`
⟨digit 2⟩ ⟶ `0` | `1`

⟨digit 8⟩ ⟶ `0` | `1` | `2` | `3` | `4` | `5` | `6` | `7`
⟨digit 10⟩ ⟶ ⟨digit⟩
⟨digit 16⟩ ⟶ ⟨digit 10⟩ | `a` | `b` | `c` | `d` | `e` | `f`

7.1.2. External representations

⟨Datum⟩ is what the `read` procedure (section 6.7.2) successfully parses. Note that any string that parses as an ⟨expression⟩ will also parse as a ⟨datum⟩.

⟨datum⟩ ⟶ ⟨simple datum⟩ | ⟨compound datum⟩
    | ⟨label⟩ `=` ⟨datum⟩ | ⟨label⟩ `#`
⟨simple datum⟩ ⟶ ⟨boolean⟩ | ⟨number⟩
    | ⟨character⟩ | ⟨string⟩ | ⟨symbol⟩ | ⟨bytevector⟩
⟨symbol⟩ ⟶ ⟨identifier⟩
⟨compound datum⟩ ⟶ ⟨list⟩ | ⟨vector⟩
⟨list⟩ ⟶ `(`⟨datum⟩*`)` | `(`⟨datum⟩$^+$ `.` ⟨datum⟩`)`
    | ⟨abbreviation⟩
⟨abbreviation⟩ ⟶ ⟨abbrev prefix⟩ ⟨datum⟩
⟨abbrev prefix⟩ ⟶ `'` | `` ` `` | `,` | `,@`
⟨vector⟩ ⟶ `#(`⟨datum⟩*`)`
⟨label⟩ ⟶ `#` ⟨digit 10⟩$^+$

7.1.3. Expressions

⟨expression⟩ ⟶ ⟨variable⟩
    | ⟨literal⟩
    | ⟨procedure call⟩
    | ⟨lambda expression⟩
    | ⟨conditional⟩
    | ⟨assignment⟩
    | ⟨derived expression⟩
    | ⟨macro use⟩
    | ⟨macro block⟩

⟨literal⟩ ⟶ ⟨quotation⟩ | ⟨self-evaluating⟩
⟨self-evaluating⟩ ⟶ ⟨boolean⟩ | ⟨number⟩
    | ⟨character⟩ | ⟨string⟩
⟨quotation⟩ ⟶ `'`⟨datum⟩ | `(quote` ⟨datum⟩`)`
⟨procedure call⟩ ⟶ `(`⟨operator⟩ ⟨operand⟩*`)`
⟨operator⟩ ⟶ ⟨expression⟩
⟨operand⟩ ⟶ ⟨expression⟩

⟨lambda expression⟩ ⟶ `(lambda` ⟨formals⟩ ⟨body⟩`)`
⟨formals⟩ ⟶ `(`⟨variable⟩*`)` | ⟨variable⟩
    | `(`⟨variable⟩$^+$ `.` ⟨variable⟩`)`
⟨body⟩ ⟶ ⟨syntax definition⟩* ⟨definition⟩* ⟨sequence⟩
⟨sequence⟩ ⟶ ⟨command⟩* ⟨expression⟩
⟨command⟩ ⟶ ⟨expression⟩

⟨conditional⟩ ⟶ `(if` ⟨test⟩ ⟨consequent⟩ ⟨alternate⟩`)`
⟨test⟩ ⟶ ⟨expression⟩
⟨consequent⟩ ⟶ ⟨expression⟩
⟨alternate⟩ ⟶ ⟨expression⟩ | ⟨empty⟩

⟨assignment⟩ ⟶ `(set!` ⟨variable⟩ ⟨expression⟩`)`

⟨derived expression⟩ ⟶
      `(cond` ⟨cond clause⟩$^+$`)`
    | `(cond` ⟨cond clause⟩* `(else` ⟨sequence⟩`))`
    | `(case` ⟨expression⟩
      ⟨case clause⟩$^+$`)`
    | `(case` ⟨expression⟩
      ⟨case clause⟩*
      `(else` ⟨sequence⟩`))`
    | `(case` ⟨expression⟩
      ⟨case clause⟩*
      `(else =>` ⟨recipient⟩`))`
    | `(and` ⟨test⟩*`)`
    | `(or` ⟨test⟩*`)`
    | `(when` ⟨expression⟩ ⟨body⟩`)`
    | `(unless` ⟨expression⟩ ⟨body⟩`)`
    | `(let (`⟨binding spec⟩*`)` ⟨body⟩`)`
    | `(let` ⟨variable⟩ `(`⟨binding spec⟩*`)` ⟨body⟩`)`
    | `(let*` `(`⟨binding spec⟩*`)` ⟨body⟩`)`
    | `(letrec (`⟨binding spec⟩*`)` ⟨body⟩`)`
    | `(letrec*` `(`⟨binding spec⟩*`)` ⟨body⟩`)`
    | `(let-values (`⟨formals⟩*`)` ⟨body⟩`)`
    | `(let*-values (`⟨formals⟩*`)` ⟨body⟩`)`
    | `(case-lambda` ⟨case-lambda clause⟩*`)`
    | `(begin` ⟨sequence⟩`)`
    | `(do (`⟨iteration spec⟩*`)`
          `(`⟨test⟩ ⟨do result⟩`)`
        ⟨command⟩*`)`
    | `(delay` ⟨expression⟩`)`
    | `(lazy` ⟨expression⟩`)`
    | ⟨quasiquotation⟩

⟨cond clause⟩ ⟶ `(`⟨test⟩ ⟨sequence⟩`)`
    | `(`⟨test⟩`)`
    | `(`⟨test⟩ `=>` ⟨recipient⟩`)`
⟨recipient⟩ ⟶ ⟨expression⟩
⟨case clause⟩ ⟶ `((`⟨datum⟩*`)` ⟨sequence⟩`)`
    | `((`⟨datum⟩*`) =>` ⟨recipient⟩`)`
⟨binding spec⟩ ⟶ `(`⟨variable⟩ ⟨expression⟩`)`
⟨iteration spec⟩ ⟶ `(`⟨variable⟩ ⟨init⟩ ⟨step⟩`)`
    | `(`⟨variable⟩ ⟨init⟩`)`
⟨case-lambda clause⟩ ⟶ `(`⟨formals⟩ ⟨body⟩`)`
⟨init⟩ ⟶ ⟨expression⟩
⟨step⟩ ⟶ ⟨expression⟩
⟨do result⟩ ⟶ ⟨sequence⟩ | ⟨empty⟩

⟨macro use⟩ ⟶ `(`⟨keyword⟩ ⟨datum⟩*`)`
⟨keyword⟩ ⟶ ⟨identifier⟩

⟨macro block⟩ ⟶
    `(let-syntax (`⟨syntax spec⟩*`)` ⟨body⟩`)`
    | `(letrec-syntax (`⟨syntax spec⟩*`)` ⟨body⟩`)`
⟨syntax spec⟩ ⟶ `(`⟨keyword⟩ ⟨transformer spec⟩`)`

7.1.4. Quasiquotations

The following grammar for quasiquote expressions is not context-free. It is presented as a recipe for generating an infinite number of production rules. Imagine a copy of the following rules for $D = 1, 2, 3, \ldots$. D keeps track of the nesting depth.

⟨quasiquotation⟩ ⟶ ⟨quasiquotation 1⟩
⟨qq template 0⟩ ⟶ ⟨expression⟩
⟨quasiquotation D⟩ ⟶ `⟨qq template D⟩
    | (quasiquote ⟨qq template D⟩)
⟨qq template D⟩ ⟶ ⟨simple datum⟩
    | ⟨list qq template D⟩
    | ⟨vector qq template D⟩
    | ⟨unquotation D⟩
⟨list qq template D⟩ ⟶ (⟨qq template or splice D⟩*)
    | (⟨qq template or splice D⟩$^+$. ⟨qq template D⟩)
    | '⟨qq template D⟩
    | ⟨quasiquotation $D+1$⟩
⟨vector qq template D⟩ ⟶ #(⟨qq template or splice D⟩*)
⟨unquotation D⟩ ⟶ ,⟨qq template $D-1$⟩
    | (unquote ⟨qq template $D-1$⟩)
⟨qq template or splice D⟩ ⟶ ⟨qq template D⟩
    | ⟨splicing unquotation D⟩
⟨splicing unquotation D⟩ ⟶ ,@⟨qq template $D-1$⟩
    | (unquote-splicing ⟨qq template $D-1$⟩)

In ⟨quasiquotation⟩s, a ⟨list qq template D⟩ can sometimes be confused with either an ⟨unquotation D⟩ or a ⟨splicing unquotation D⟩. The interpretation as an ⟨unquotation⟩ or ⟨splicing unquotation D⟩ takes precedence.

7.1.5. Transformers

⟨transformer spec⟩ ⟶
    (syntax-rules (⟨identifier⟩*) ⟨syntax rule⟩*)
    | (syntax-rules ⟨identifier⟩ (⟨identifier⟩*)
        ⟨syntax rule⟩*)
⟨syntax rule⟩ ⟶ (⟨pattern⟩ ⟨template⟩)
⟨pattern⟩ ⟶ ⟨pattern identifier⟩
    | ⟨underscore⟩
    | (⟨pattern⟩*)
    | (⟨pattern⟩$^+$. ⟨pattern⟩)
    | (⟨pattern⟩* ⟨pattern⟩ ⟨ellipsis⟩ ⟨pattern⟩*)
    | (⟨pattern⟩* ⟨pattern⟩ ⟨ellipsis⟩ ⟨pattern⟩*
        . ⟨pattern⟩)
    | #(⟨pattern⟩*)
    | #(⟨pattern⟩* ⟨pattern⟩ ⟨ellipsis⟩ ⟨pattern⟩*)
    | ⟨pattern datum⟩
⟨pattern datum⟩ ⟶ ⟨string⟩
    | ⟨character⟩
    | ⟨boolean⟩
    | ⟨number⟩
⟨template⟩ ⟶ ⟨pattern identifier⟩
    | (⟨template element⟩*)
    | (⟨template element⟩$^+$. ⟨template⟩)
    | #(⟨template element⟩*)
    | ⟨template datum⟩
⟨template element⟩ ⟶ ⟨template⟩
    | ⟨template⟩ ⟨ellipsis⟩
⟨template datum⟩ ⟶ ⟨pattern datum⟩
⟨pattern identifier⟩ ⟶ ⟨any identifier except ...⟩
⟨ellipsis⟩ ⟶ ⟨an identifier defaulting to ...⟩
⟨underscore⟩ ⟶ ⟨the identifier _⟩

7.1.6. Programs and definitions

⟨program⟩ ⟶ ⟨command or definition⟩*
⟨command or definition⟩ ⟶ ⟨command⟩
    | ⟨definition⟩
    | ⟨syntax definition⟩
    | (import ⟨import set⟩$^+$)
    | (begin ⟨command or definition⟩$^+$)
⟨definition⟩ ⟶ (define ⟨variable⟩ ⟨expression⟩)
    | (define (⟨variable⟩ ⟨def formals⟩) ⟨body⟩)
    | (define-record-type ⟨variable⟩
        ⟨constructor⟩ ⟨variable⟩ ⟨field spec⟩*)
    | (begin ⟨definition⟩*)
⟨def formals⟩ ⟶ ⟨variable⟩*
    | ⟨variable⟩* . ⟨variable⟩
⟨constructor⟩ ⟶ (⟨variable⟩ ⟨field name⟩*)
⟨field spec⟩ ⟶ (⟨field name⟩ ⟨variable⟩)
    | (⟨field name⟩ ⟨variable⟩ ⟨variable⟩)
⟨field name⟩ ⟶ ⟨identifier⟩
⟨syntax definition⟩ ⟶
    (define-syntax ⟨keyword⟩ ⟨transformer spec⟩)
    | (begin ⟨syntax definition⟩*)

7.1.7. Modules

⟨module⟩ ⟶
    (module ⟨module name⟩ ⟨module declaration⟩*)
⟨module name⟩ ⟶ (⟨module name part⟩$^+$)
⟨module name part⟩ ⟶ ⟨identifier⟩ | ⟨uinteger 10⟩
⟨module declaration⟩ ⟶ (export ⟨export spec⟩*)
    | (import ⟨import set⟩*)
    | (begin ⟨command or definition⟩*)
    | (include ⟨string⟩$^+$)
    | (include-ci ⟨string⟩$^+$)
    | (cond-expand ⟨cond-expand clause⟩*)
    | (cond-expand ⟨cond-expand clause⟩*
            (else ⟨module declaration⟩*))
⟨export spec⟩ ⟶ ⟨identifier⟩
    | (rename ⟨identifier⟩ ⟨identifier⟩)

⟨import set⟩ ⟶ ⟨module name⟩
| `(only` ⟨import set⟩ ⟨identifier⟩$^+$`)`
| `(except` ⟨import set⟩ ⟨identifier⟩$^+$`)`
| `(prefix` ⟨import set⟩ ⟨identifier⟩`)`
| `(rename` ⟨import set⟩ ⟨export spec⟩$^+$`)`
⟨cond-expand clause⟩ ⟶
`(`⟨feature requirement⟩ ⟨module declaration⟩*`)`
⟨feature requirement⟩ ⟶ ⟨identifier⟩
| ⟨module name⟩
| `(and` ⟨feature requirement⟩*`)`
| `(or` ⟨feature requirement⟩*`)`
| `(not` ⟨feature requirement⟩`)`

7.2. Formal semantics

This section provides a formal denotational semantics for the primitive expressions of Scheme and selected built-in procedures. The concepts and notation used here are described in [31]; the notation is summarized below:

$\langle \ldots \rangle$	sequence formation
$s \downarrow k$	kth member of the sequence s (1-based)
$\#s$	length of sequence s
$s \S t$	concatenation of sequences s and t
$s \dagger k$	drop the first k members of sequence s
$t \to a, b$	McCarthy conditional "if t then a else b"
$\rho[x/i]$	substitution "ρ with x for i"
x in D	injection of x into domain D
$x \mid$ D	projection of x to domain D

The reason that expression continuations take sequences of values instead of single values is to simplify the formal treatment of procedure calls and multiple return values.

The boolean flag associated with pairs, vectors, and strings will be true for mutable objects and false for immutable objects.

The order of evaluation within a call is unspecified. We mimic that here by applying arbitrary permutations *permute* and *unpermute*, which must be inverses, to the arguments in a call before and after they are evaluated. This is not quite right since it suggests, incorrectly, that the order of evaluation is constant throughout a program (for any given number of arguments), but it is a closer approximation to the intended semantics than a left-to-right evaluation would be.

The storage allocator *new* is implementation-dependent, but it must obey the following axiom: if $new\,\sigma \in$ L, then $\sigma\,(new\,\sigma \mid \text{L}) \downarrow 2 = false$.

The definition of $\mathcal{K}$ is omitted because an accurate definition of $\mathcal{K}$ would complicate the semantics without being very interesting.

If P is a program in which all variables are defined before being referenced or assigned, then the meaning of P is

$$\mathcal{E}[\![\texttt{((lambda (I*) P')} \langle\text{undefined}\rangle \texttt{ ...)}]\!]$$

where I* is the sequence of variables defined in P, P′ is the sequence of expressions obtained by replacing every definition in P by an assignment, ⟨undefined⟩ is an expression that evaluates to *undefined*, and $\mathcal{E}$ is the semantic function that assigns meaning to expressions.

7.2.1. Abstract syntax

K ∈ Con	constants, including quotations
I ∈ Ide	identifiers (variables)
E ∈ Exp	expressions
Γ ∈ Com = Exp	commands

Exp ⟶ K | I | `(`E_0 E*`)`
| `(lambda (I*)` Γ* E_0`)`
| `(lambda (I* . I)` Γ* E_0`)`
| `(lambda` I Γ* E_0`)`
| `(if` E_0 E_1 E_2`)` | `(if` E_0 E_1`)`
| `(set!` I E`)`

7.2.2. Domain equations

$\alpha \in$ L		locations
$\nu \in$ N		natural numbers
T	$= \{false, true\}$	booleans
Q		symbols
H		characters
R		numbers
E_p	$= \text{L} \times \text{L} \times \text{T}$	pairs
E_v	$= \text{L}^* \times \text{T}$	vectors
E_s	$= \text{L}^* \times \text{T}$	strings
M	$= \{false, true, null, undefined, unspecified\}$	miscellaneous
$\phi \in$ F	$= \text{L} \times (\text{E}^* \to \text{K} \to \text{C})$	procedure values
$\epsilon \in$ E	$= \text{Q} + \text{H} + \text{R} + \text{E}_\text{p} + \text{E}_\text{v} + \text{E}_\text{s} + \text{M} + \text{F}$	expressed values
$\sigma \in$ S	$= \text{L} \to (\text{E} \times \text{T})$	stores
$\rho \in$ U	$= \text{Ide} \to \text{L}$	environments
$\theta \in$ C	$= \text{S} \to \text{A}$	command continuations
$\kappa \in$ K	$= \text{E}^* \to \text{C}$	expression continuations
A		answers
X		errors

7.2.3. Semantic functions

$\mathcal{K}$: Con → E
$\mathcal{E}$: Exp → U → K → C
$\mathcal{E}^*$: Exp* → U → K → C
$\mathcal{C}$: Com* → U → C → C

Definition of $\mathcal{K}$ deliberately omitted.

$\mathcal{E}[\![\mathrm{K}]\!] = \lambda\rho\kappa\,.\,\mathit{send}\,(\mathcal{K}[\![\mathrm{K}]\!])\,\kappa$

$\mathcal{E}[\![\mathrm{I}]\!] = \lambda\rho\kappa\,.\,\mathit{hold}\,(\mathit{lookup}\,\rho\,\mathrm{I})$
$\quad(\mathit{single}(\lambda\epsilon\,.\,\epsilon = \mathit{undefined} \to$
$\quad\quad\mathit{wrong}\text{ “undefined variable”},$
$\quad\quad\mathit{send}\,\epsilon\,\kappa))$

$\mathcal{E}[\![\texttt{(}\mathrm{E}_0\ \mathrm{E}^*\texttt{)}]\!] =$
$\quad\lambda\rho\kappa\,.\,\mathcal{E}^*(\mathit{permute}(\langle\mathrm{E}_0\rangle \S \mathrm{E}^*))$
$\quad\quad\rho$
$\quad\quad(\lambda\epsilon^*\,.\,((\lambda\epsilon^*\,.\,\mathit{applicate}\,(\epsilon^* \downarrow 1)\,(\epsilon^* \dagger 1)\,\kappa)$
$\quad\quad\quad(\mathit{unpermute}\,\epsilon^*)))$

$\mathcal{E}[\![\texttt{(lambda (I*) }\Gamma^*\ \mathrm{E}_0\texttt{)}]\!] =$
$\quad\lambda\rho\kappa\,.\,\lambda\sigma\,.$
$\quad\quad\mathit{new}\,\sigma \in \mathtt{L} \to$
$\quad\quad\mathit{send}\,(\langle\mathit{new}\,\sigma \mid \mathtt{L},$
$\quad\quad\quad\lambda\epsilon^*\kappa'\,.\,\#\epsilon^* = \#\mathrm{I}^* \to$
$\quad\quad\quad\quad\mathit{tievals}(\lambda\alpha^*\,.\,(\lambda\rho'\,.\,\mathcal{C}[\![\Gamma^*]\!]\rho'(\mathcal{E}[\![\mathrm{E}_0]\!]\rho'\kappa'))$
$\quad\quad\quad\quad\quad\quad(\mathit{extends}\,\rho\,\mathrm{I}^*\,\alpha^*))$
$\quad\quad\quad\quad\quad\epsilon^*,$
$\quad\quad\quad\quad\mathit{wrong}\text{ “wrong number of arguments”}\rangle$
$\quad\quad\quad\text{in }\mathtt{E})$
$\quad\quad\quad\kappa$
$\quad\quad\quad(\mathit{update}\,(\mathit{new}\,\sigma \mid \mathtt{L})\,\mathit{unspecified}\,\sigma),$
$\quad\quad\mathit{wrong}\text{ “out of memory”}\,\sigma$

$\mathcal{E}[\![\texttt{(lambda (I* . I) }\Gamma^*\ \mathrm{E}_0\texttt{)}]\!] =$
$\quad\lambda\rho\kappa\,.\,\lambda\sigma\,.$
$\quad\quad\mathit{new}\,\sigma \in \mathtt{L} \to$
$\quad\quad\mathit{send}\,(\langle\mathit{new}\,\sigma \mid \mathtt{L},$
$\quad\quad\quad\lambda\epsilon^*\kappa'\,.\,\#\epsilon^* \geq \#\mathrm{I}^* \to$
$\quad\quad\quad\quad\mathit{tievalsrest}$
$\quad\quad\quad\quad\quad(\lambda\alpha^*\,.\,(\lambda\rho'\,.\,\mathcal{C}[\![\Gamma^*]\!]\rho'(\mathcal{E}[\![\mathrm{E}_0]\!]\rho'\kappa'))$
$\quad\quad\quad\quad\quad\quad(\mathit{extends}\,\rho\,(\mathrm{I}^* \S \langle\mathrm{I}\rangle)\,\alpha^*))$
$\quad\quad\quad\quad\quad\epsilon^*$
$\quad\quad\quad\quad\quad(\#\mathrm{I}^*),$
$\quad\quad\quad\quad\mathit{wrong}\text{ “too few arguments”}\rangle\text{ in }\mathtt{E})$
$\quad\quad\quad\kappa$
$\quad\quad\quad(\mathit{update}\,(\mathit{new}\,\sigma \mid \mathtt{L})\,\mathit{unspecified}\,\sigma),$
$\quad\quad\mathit{wrong}\text{ “out of memory”}\,\sigma$

$\mathcal{E}[\![\texttt{(lambda I }\Gamma^*\ \mathrm{E}_0\texttt{)}]\!] = \mathcal{E}[\![\texttt{(lambda (. I) }\Gamma^*\ \mathrm{E}_0\texttt{)}]\!]$

$\mathcal{E}[\![\texttt{(if }\mathrm{E}_0\ \mathrm{E}_1\ \mathrm{E}_2\texttt{)}]\!] =$
$\quad\lambda\rho\kappa\,.\,\mathcal{E}[\![\mathrm{E}_0]\!]\,\rho\,(\mathit{single}\,(\lambda\epsilon\,.\,\mathit{truish}\,\epsilon \to \mathcal{E}[\![\mathrm{E}_1]\!]\rho\kappa,$
$\quad\quad\quad\quad\quad\quad\mathcal{E}[\![\mathrm{E}_2]\!]\rho\kappa))$

$\mathcal{E}[\![\texttt{(if }\mathrm{E}_0\ \mathrm{E}_1\texttt{)}]\!] =$
$\quad\lambda\rho\kappa\,.\,\mathcal{E}[\![\mathrm{E}_0]\!]\,\rho\,(\mathit{single}\,(\lambda\epsilon\,.\,\mathit{truish}\,\epsilon \to \mathcal{E}[\![\mathrm{E}_1]\!]\rho\kappa,$
$\quad\quad\quad\quad\quad\quad\mathit{send}\,\mathit{unspecified}\,\kappa))$

Here and elsewhere, any expressed value other than *undefined* may be used in place of *unspecified*.

$\mathcal{E}[\![\texttt{(set! I E)}]\!] =$
$\quad\lambda\rho\kappa\,.\,\mathcal{E}[\![\mathrm{E}]\!]\,\rho\,(\mathit{single}(\lambda\epsilon\,.\,\mathit{assign}\,(\mathit{lookup}\,\rho\,\mathrm{I})$
$\quad\quad\quad\quad\quad\quad\quad\quad\epsilon$
$\quad\quad\quad\quad\quad\quad\quad\quad(\mathit{send}\,\mathit{unspecified}\,\kappa)))$

$\mathcal{E}^*[\![\,]\!] = \lambda\rho\kappa\,.\,\kappa\langle\,\rangle$

$\mathcal{E}^*[\![\mathrm{E}_0\ \mathrm{E}^*]\!] =$
$\quad\lambda\rho\kappa\,.\,\mathcal{E}[\![\mathrm{E}_0]\!]\,\rho\,(\mathit{single}(\lambda\epsilon_0\,.\,\mathcal{E}^*[\![\mathrm{E}^*]\!]\,\rho\,(\lambda\epsilon^*\,.\,\kappa\,(\langle\epsilon_0\rangle \S \epsilon^*))))$

$\mathcal{C}[\![\,]\!] = \lambda\rho\theta\,.\,\theta$

$\mathcal{C}[\![\Gamma_0\ \Gamma^*]\!] = \lambda\rho\theta\,.\,\mathcal{E}[\![\Gamma_0]\!]\,\rho\,(\lambda\epsilon^*\,.\,\mathcal{C}[\![\Gamma^*]\!]\rho\theta)$

7.2.4. Auxiliary functions

$\mathit{lookup} : \mathtt{U} \to \mathrm{Ide} \to \mathtt{L}$
$\mathit{lookup} = \lambda\rho\mathrm{I}\,.\,\rho\mathrm{I}$

$\mathit{extends} : \mathtt{U} \to \mathrm{Ide}^* \to \mathtt{L}^* \to \mathtt{U}$
$\mathit{extends} =$
$\quad\lambda\rho\mathrm{I}^*\alpha^*\,.\,\#\mathrm{I}^* = 0 \to \rho,$
$\quad\quad\mathit{extends}\,(\rho[(\alpha^* \downarrow 1)/(\mathrm{I}^* \downarrow 1)])\,(\mathrm{I}^* \dagger 1)\,(\alpha^* \dagger 1)$

$\mathit{wrong} : \mathtt{X} \to \mathtt{C}$ [implementation-dependent]

$\mathit{send} : \mathtt{E} \to \mathtt{K} \to \mathtt{C}$
$\mathit{send} = \lambda\epsilon\kappa\,.\,\kappa\langle\epsilon\rangle$

$\mathit{single} : (\mathtt{E} \to \mathtt{C}) \to \mathtt{K}$
$\mathit{single} =$
$\quad\lambda\psi\epsilon^*\,.\,\#\epsilon^* = 1 \to \psi(\epsilon^* \downarrow 1),$
$\quad\quad\mathit{wrong}\text{ “wrong number of return values”}$

$\mathit{new} : \mathtt{S} \to (\mathtt{L} + \{\mathit{error}\})$ [implementation-dependent]

$\mathit{hold} : \mathtt{L} \to \mathtt{K} \to \mathtt{C}$
$\mathit{hold} = \lambda\alpha\kappa\sigma\,.\,\mathit{send}\,(\sigma\alpha \downarrow 1)\kappa\sigma$

$\mathit{assign} : \mathtt{L} \to \mathtt{E} \to \mathtt{C} \to \mathtt{C}$
$\mathit{assign} = \lambda\alpha\epsilon\theta\sigma\,.\,\theta(\mathit{update}\,\alpha\epsilon\sigma)$

$\mathit{update} : \mathtt{L} \to \mathtt{E} \to \mathtt{S} \to \mathtt{S}$
$\mathit{update} = \lambda\alpha\epsilon\sigma\,.\,\sigma[\langle\epsilon, \mathit{true}\rangle/\alpha]$

$\mathit{tievals} : (\mathtt{L}^* \to \mathtt{C}) \to \mathtt{E}^* \to \mathtt{C}$
$\mathit{tievals} =$
$\quad\lambda\psi\epsilon^*\sigma\,.\,\#\epsilon^* = 0 \to \psi\langle\,\rangle\sigma,$
$\quad\quad\mathit{new}\,\sigma \in \mathtt{L} \to \mathit{tievals}\,(\lambda\alpha^*\,.\,\psi(\langle\mathit{new}\,\sigma \mid \mathtt{L}\rangle \S \alpha^*))$
$\quad\quad\quad\quad\quad\quad\quad\quad(\epsilon^* \dagger 1)$
$\quad\quad\quad\quad\quad\quad\quad\quad(\mathit{update}(\mathit{new}\,\sigma \mid \mathtt{L})(\epsilon^* \downarrow 1)\sigma),$
$\quad\quad\quad\quad\mathit{wrong}\text{ “out of memory”}\sigma$

$\mathit{tievalsrest} : (\mathtt{L}^* \to \mathtt{C}) \to \mathtt{E}^* \to \mathtt{N} \to \mathtt{C}$
$\mathit{tievalsrest} =$
$\quad\lambda\psi\epsilon^*\nu\,.\,\mathit{list}\,(\mathit{dropfirst}\,\epsilon^*\nu)$
$\quad\quad\quad(\mathit{single}(\lambda\epsilon\,.\,\mathit{tievals}\,\psi\,((\mathit{takefirst}\,\epsilon^*\nu) \S \langle\epsilon\rangle))))$

$\mathit{dropfirst} = \lambda ln\,.\,n = 0 \to l, \mathit{dropfirst}\,(l \dagger 1)(n-1)$

$\mathit{takefirst} = \lambda ln\,.\,n = 0 \to \langle\,\rangle, \langle l \downarrow 1\rangle \S (\mathit{takefirst}\,(l \dagger 1)(n-1))$

$\mathit{truish} : \mathtt{E} \to \mathtt{T}$
$\mathit{truish} = \lambda\epsilon\,.\,\epsilon = \mathit{false} \to \mathit{false}, \mathit{true}$

$\mathit{permute} : \mathrm{Exp}^* \to \mathrm{Exp}^*$ [implementation-dependent]

$\mathit{unpermute} : \mathtt{E}^* \to \mathtt{E}^*$ [inverse of *permute*]

$\mathit{applicate} : \mathtt{E} \to \mathtt{E}^* \to \mathtt{K} \to \mathtt{C}$
$\mathit{applicate} =$
$\quad\lambda\epsilon\epsilon^*\kappa\,.\,\epsilon \in \mathtt{F} \to (\epsilon \mid \mathtt{F} \downarrow 2)\epsilon^*\kappa, \mathit{wrong}\text{ “bad procedure”}$

$$
\begin{array}{l}
\mathit{onearg} : (\mathtt{E} \to \mathtt{K} \to \mathtt{C}) \to (\mathtt{E}^* \to \mathtt{K} \to \mathtt{C}) \\
\mathit{onearg} = \\
\quad \lambda\zeta\epsilon^*\kappa \,.\, \#\epsilon^* = 1 \to \zeta(\epsilon^* \downarrow 1)\kappa, \\
\qquad\qquad \mathit{wrong}\ \text{“wrong number of arguments”}
\end{array}
$$

$$
\begin{array}{l}
\mathit{twoarg} : (\mathtt{E} \to \mathtt{E} \to \mathtt{K} \to \mathtt{C}) \to (\mathtt{E}^* \to \mathtt{K} \to \mathtt{C}) \\
\mathit{twoarg} = \\
\quad \lambda\zeta\epsilon^*\kappa \,.\, \#\epsilon^* = 2 \to \zeta(\epsilon^* \downarrow 1)(\epsilon^* \downarrow 2)\kappa, \\
\qquad\qquad \mathit{wrong}\ \text{“wrong number of arguments”}
\end{array}
$$

$$
\begin{array}{l}
\mathit{list} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{list} = \\
\quad \lambda\epsilon^*\kappa \,.\, \#\epsilon^* = 0 \to \mathit{send}\,\mathit{null}\,\kappa, \\
\qquad\qquad \mathit{list}\,(\epsilon^* \dagger 1)(\mathit{single}(\lambda\epsilon \,.\, \mathit{cons}\langle\epsilon^* \downarrow 1, \epsilon\rangle\kappa))
\end{array}
$$

$$
\begin{array}{l}
\mathit{cons} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{cons} = \\
\quad \mathit{twoarg}\,(\lambda\epsilon_1\epsilon_2\kappa\sigma \,.\, \mathit{new}\,\sigma \in \mathtt{L} \to \\
\qquad\qquad\qquad (\lambda\sigma' \,.\, \mathit{new}\,\sigma' \in \mathtt{L} \to \\
\qquad\qquad\qquad\qquad \mathit{send}\,(\langle \mathit{new}\,\sigma \mid \mathtt{L}, \mathit{new}\,\sigma' \mid \mathtt{L}, \mathit{true}\rangle \\
\qquad\qquad\qquad\qquad\qquad \text{in } \mathtt{E}) \\
\qquad\qquad\qquad\qquad\qquad \kappa \\
\qquad\qquad\qquad\qquad\qquad (\mathit{update}(\mathit{new}\,\sigma' \mid \mathtt{L})\epsilon_2\sigma'), \\
\qquad\qquad\qquad\qquad \mathit{wrong}\ \text{“out of memory”}\sigma') \\
\qquad\qquad\qquad (\mathit{update}(\mathit{new}\,\sigma \mid \mathtt{L})\epsilon_1\sigma), \\
\qquad\qquad\qquad \mathit{wrong}\ \text{“out of memory”}\sigma)
\end{array}
$$

$$
\begin{array}{l}
\mathit{less} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{less} = \\
\quad \mathit{twoarg}\,(\lambda\epsilon_1\epsilon_2\kappa \,.\, (\epsilon_1 \in \mathtt{R} \wedge \epsilon_2 \in \mathtt{R}) \to \\
\qquad\qquad\qquad \mathit{send}\,(\epsilon_1 \mid \mathtt{R} < \epsilon_2 \mid \mathtt{R} \to \mathit{true}, \mathit{false})\kappa, \\
\qquad\qquad\qquad \mathit{wrong}\ \text{“non-numeric argument to }\texttt{<}\text{”})
\end{array}
$$

$$
\begin{array}{l}
\mathit{add} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{add} = \\
\quad \mathit{twoarg}\,(\lambda\epsilon_1\epsilon_2\kappa \,.\, (\epsilon_1 \in \mathtt{R} \wedge \epsilon_2 \in \mathtt{R}) \to \\
\qquad\qquad\qquad \mathit{send}\,((\epsilon_1 \mid \mathtt{R} + \epsilon_2 \mid \mathtt{R}) \text{ in } \mathtt{E})\kappa, \\
\qquad\qquad\qquad \mathit{wrong}\ \text{“non-numeric argument to }\texttt{+}\text{”})
\end{array}
$$

$$
\begin{array}{l}
\mathit{car} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{car} = \\
\quad \mathit{onearg}\,(\lambda\epsilon\kappa \,.\, \epsilon \in \mathtt{E_p} \to \mathit{hold}\,(\epsilon \mid \mathtt{E_p} \downarrow 1)\kappa, \\
\qquad\qquad\qquad\qquad \mathit{wrong}\ \text{“non-pair argument to }\texttt{car}\text{”})
\end{array}
$$

$\mathit{cdr} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C}$ [similar to *car*]

$$
\begin{array}{l}
\mathit{setcar} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{setcar} = \\
\quad \mathit{twoarg}\,(\lambda\epsilon_1\epsilon_2\kappa \,.\, \epsilon_1 \in \mathtt{E_p} \to \\
\qquad\qquad\qquad (\epsilon_1 \mid \mathtt{E_p} \downarrow 3) \to \mathit{assign}\,(\epsilon_1 \mid \mathtt{E_p} \downarrow 1) \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\quad \epsilon_2 \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\quad (\mathit{send}\,\mathit{unspecified}\,\kappa), \\
\qquad\qquad\qquad \mathit{wrong}\ \text{“immutable argument to }\texttt{set-car!}\text{”}, \\
\qquad\qquad\qquad \mathit{wrong}\ \text{“non-pair argument to }\texttt{set-car!}\text{”})
\end{array}
$$

$$
\begin{array}{l}
\mathit{eqv} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{eqv} = \\
\quad \mathit{twoarg}\,(\lambda\epsilon_1\epsilon_2\kappa \,.\, (\epsilon_1 \in \mathtt{M} \wedge \epsilon_2 \in \mathtt{M}) \to \\
\qquad\qquad\qquad\quad \mathit{send}\,(\epsilon_1 \mid \mathtt{M} = \epsilon_2 \mid \mathtt{M} \to \mathit{true}, \mathit{false})\kappa, \\
\qquad\qquad\qquad (\epsilon_1 \in \mathtt{Q} \wedge \epsilon_2 \in \mathtt{Q}) \to \\
\qquad\qquad\qquad\quad \mathit{send}\,(\epsilon_1 \mid \mathtt{Q} = \epsilon_2 \mid \mathtt{Q} \to \mathit{true}, \mathit{false})\kappa, \\
\qquad\qquad\qquad (\epsilon_1 \in \mathtt{H} \wedge \epsilon_2 \in \mathtt{H}) \to \\
\qquad\qquad\qquad\quad \mathit{send}\,(\epsilon_1 \mid \mathtt{H} = \epsilon_2 \mid \mathtt{H} \to \mathit{true}, \mathit{false})\kappa, \\
\qquad\qquad\qquad (\epsilon_1 \in \mathtt{R} \wedge \epsilon_2 \in \mathtt{R}) \to \\
\qquad\qquad\qquad\quad \mathit{send}\,(\epsilon_1 \mid \mathtt{R} = \epsilon_2 \mid \mathtt{R} \to \mathit{true}, \mathit{false})\kappa, \\
\qquad\qquad\qquad (\epsilon_1 \in \mathtt{E_p} \wedge \epsilon_2 \in \mathtt{E_p}) \to \\
\qquad\qquad\qquad\quad \mathit{send}\,((\lambda p_1 p_2 \,.\, ((p_1 \downarrow 1) = (p_2 \downarrow 1) \wedge \\
\qquad\qquad\qquad\qquad\qquad\qquad (p_1 \downarrow 2) = (p_2 \downarrow 2)) \to \mathit{true}, \\
\qquad\qquad\qquad\qquad\qquad\qquad \mathit{false}) \\
\qquad\qquad\qquad\qquad\quad (\epsilon_1 \mid \mathtt{E_p}) \\
\qquad\qquad\qquad\qquad\quad (\epsilon_2 \mid \mathtt{E_p})) \\
\qquad\qquad\qquad\qquad \kappa, \\
\qquad\qquad\qquad (\epsilon_1 \in \mathtt{E_v} \wedge \epsilon_2 \in \mathtt{E_v}) \to \ldots, \\
\qquad\qquad\qquad (\epsilon_1 \in \mathtt{E_s} \wedge \epsilon_2 \in \mathtt{E_s}) \to \ldots, \\
\qquad\qquad\qquad (\epsilon_1 \in \mathtt{F} \wedge \epsilon_2 \in \mathtt{F}) \to \\
\qquad\qquad\qquad\quad \mathit{send}\,((\epsilon_1 \mid \mathtt{F} \downarrow 1) = (\epsilon_2 \mid \mathtt{F} \downarrow 1) \to \mathit{true}, \mathit{false}) \\
\qquad\qquad\qquad\qquad \kappa, \\
\qquad\qquad\qquad \mathit{send}\ \mathit{false}\ \kappa)
\end{array}
$$

$$
\begin{array}{l}
\mathit{apply} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{apply} = \\
\quad \mathit{twoarg}\,(\lambda\epsilon_1\epsilon_2\kappa \,.\, \epsilon_1 \in \mathtt{F} \to \mathit{valueslist}\,\langle\epsilon_2\rangle(\lambda\epsilon^* \,.\, \mathit{applicate}\,\epsilon_1\epsilon^*\kappa), \\
\qquad\qquad\qquad\qquad \mathit{wrong}\ \text{“bad procedure argument to }\texttt{apply}\text{”})
\end{array}
$$

$$
\begin{array}{l}
\mathit{valueslist} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{valueslist} = \\
\quad \mathit{onearg}\,(\lambda\epsilon\kappa \,.\, \epsilon \in \mathtt{E_p} \to \\
\qquad\qquad\qquad\qquad \mathit{cdr}\langle\epsilon\rangle \\
\qquad\qquad\qquad\qquad\quad (\lambda\epsilon^* \,.\, \mathit{valueslist} \\
\qquad\qquad\qquad\qquad\qquad\qquad \epsilon^* \\
\qquad\qquad\qquad\qquad\qquad\qquad (\lambda\epsilon^* \,.\, \mathit{car}\langle\epsilon\rangle(\mathit{single}(\lambda\epsilon \,.\, \kappa(\langle\epsilon\rangle \S \epsilon^*))))), \\
\qquad\qquad\qquad\qquad \epsilon = \mathit{null} \to \kappa\langle\,\rangle, \\
\qquad\qquad\qquad\qquad \mathit{wrong}\ \text{“non-list argument to }\texttt{values-list}\text{”})
\end{array}
$$

$\mathit{cwcc} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C}$ [`call-with-current-continuation`]

$$
\begin{array}{l}
\mathit{cwcc} = \\
\quad \mathit{onearg}\,(\lambda\epsilon\kappa \,.\, \epsilon \in \mathtt{F} \to \\
\qquad\qquad\qquad\qquad (\lambda\sigma \,.\, \mathit{new}\,\sigma \in \mathtt{L} \to \\
\qquad\qquad\qquad\qquad\qquad\qquad \mathit{applicate}\,\epsilon \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad \langle\langle \mathit{new}\,\sigma \mid \mathtt{L}, \lambda\epsilon^*\kappa' \,.\, \kappa\epsilon^*\rangle \text{ in } \mathtt{E}\rangle \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad \kappa \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad (\mathit{update}\,(\mathit{new}\,\sigma \mid \mathtt{L}) \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad\quad \mathit{unspecified} \\
\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad\qquad\quad \sigma), \\
\qquad\qquad\qquad\qquad\qquad\qquad \mathit{wrong}\ \text{“out of memory”}\ \sigma), \\
\qquad\qquad\qquad\qquad \mathit{wrong}\ \text{“bad procedure argument”})
\end{array}
$$

$$
\begin{array}{l}
\mathit{values} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C} \\
\mathit{values} = \lambda\epsilon^*\kappa \,.\, \kappa\epsilon^*
\end{array}
$$

$\mathit{cwv} : \mathtt{E}^* \to \mathtt{K} \to \mathtt{C}$ [`call-with-values`]

$$
\begin{array}{l}
\mathit{cwv} = \\
\quad \mathit{twoarg}\,(\lambda\epsilon_1\epsilon_2\kappa \,.\, \mathit{applicate}\,\epsilon_1\langle\,\rangle(\lambda\epsilon^* \,.\, \mathit{applicate}\,\epsilon_2\,\epsilon^*))
\end{array}
$$

7.3. Derived expression types

This section gives macro definitions for the derived expression types in terms of the primitive expression types (literal, variable, call, `lambda`, `if`, `set!`). Definitions of `lazy` and `delay` depend on implementation details, and are not given here.

```
(define-syntax cond
  (syntax-rules (else =>)
    ((cond (else result1 result2 ...))
     (begin result1 result2 ...))
    ((cond (test => result))
     (let ((temp test))
       (if temp (result temp))))
    ((cond (test => result) clause1 clause2 ...)
     (let ((temp test))
       (if temp
           (result temp)
           (cond clause1 clause2 ...))))
    ((cond (test)) test)
    ((cond (test) clause1 clause2 ...)
     (let ((temp test))
       (if temp
           temp
           (cond clause1 clause2 ...))))
    ((cond (test result1 result2 ...))
     (if test (begin result1 result2 ...)))
    ((cond (test result1 result2 ...)
           clause1 clause2 ...)
     (if test
         (begin result1 result2 ...)
         (cond clause1 clause2 ...)))))
```

```
(define-syntax case
  (syntax-rules (else =>)
    ((case (key ...)
       clauses ...)
     (let ((atom-key (key ...)))
       (case atom-key clauses ...)))
    ((case key
       (else => result))
     (result key))
    ((case key
       (else result1 result2 ...))
     (begin result1 result2 ...))
    ((case key
       ((atoms ...) result1 result2 ...))
     (if (memv key '(atoms ...))
         (begin result1 result2 ...)))
    ((case key
       ((atoms ...) => result))
     (if (memv key '(atoms ...))
         (result key)))
    ((case key
       ((atoms ...) => result)
       clause clauses ...)
     (if (memv key '(atoms ...))
         (result key)
         (case key clause clauses ...)))
    ((case key
       ((atoms ...) result1 result2 ...)
       clause clauses ...)
     (if (memv key '(atoms ...))
         (begin result1 result2 ...)
         (case key clause clauses ...)))))
```

```
(define-syntax and
  (syntax-rules ()
    ((and) #t)
    ((and test) test)
    ((and test1 test2 ...)
     (if test1 (and test2 ...) #f))))
```

```
(define-syntax or
  (syntax-rules ()
    ((or) #f)
    ((or test) test)
    ((or test1 test2 ...)
     (let ((x test1))
       (if x x (or test2 ...))))))
```

```
(define-syntax let
  (syntax-rules ()
    ((let ((name val) ...) body1 body2 ...)
     ((lambda (name ...) body1 body2 ...)
      val ...))
    ((let tag ((name val) ...) body1 body2 ...)
     ((letrec ((tag (lambda (name ...)
                      body1 body2 ...)))
        tag)
      val ...))))
```

```
(define-syntax let*
  (syntax-rules ()
    ((let* () body1 body2 ...)
     (let () body1 body2 ...))
    ((let* ((name1 val1) (name2 val2) ...)
       body1 body2 ...)
     (let ((name1 val1))
       (let* ((name2 val2) ...)
         body1 body2 ...)))))
```

The following `letrec` macro uses the symbol `<undefined>` in place of an expression which returns something that when stored in a location makes it an error to try to obtain the value stored in the location (no such expression is defined in Scheme). A trick is used to generate the temporary names needed to avoid specifying the order in which the values are evaluated. This could also be accomplished by using an auxiliary macro.

```
(define-syntax letrec
  (syntax-rules ()
    ((letrec ((var1 init1) ...) body ...)
     (letrec "generate_temp_names"
       (var1 ...)
       ()
       ((var1 init1) ...)
       body ...))
    ((letrec "generate_temp_names"
       ()
       (temp1 ...)
       ((var1 init1) ...)
```

```
      body ...)
    (let ((var1 <undefined>) ...)
      (let ((temp1 init1) ...)
        (set! var1 temp1)
        ...
        body ...)))
   ((letrec "generate_temp_names"
      (x y ...)
      (temp ...)
      ((var1 init1) ...)
      body ...)
    (letrec "generate_temp_names"
      (y ...)
      (newtemp temp ...)
      ((var1 init1) ...)
      body ...))))
```

```
(define-syntax letrec*
  (syntax-rules ()
    ((letrec* ((var1 init1) ...) body1 body2 ...)
     (let ((var1 <undefined>) ...)
       (set! var1 init1)
       ...
       (let () body1 body2 ...)))))
```

```
(define-syntax let-values
  (syntax-rules ()
    ((let-values (binding ...) body0 body1 ...)
     (let-values "bind"
         (binding ...) () (begin body0 body1 ...)))

    ((let-values "bind" () tmps body)
     (let tmps body))

    ((let-values "bind" ((b0 e0)
         binding ...) tmps body)
     (let-values "mktmp" b0 e0 ()
         (binding ...) tmps body))

    ((let-values "mktmp" () e0 args
         bindings tmps body)
     (call-with-values
       (lambda () e0)
       (lambda args
         (let-values "bind"
             bindings tmps body))))

    ((let-values "mktmp" (a . b) e0 (arg ...)
         bindings (tmp ...) body)
     (let-values "mktmp" b e0 (arg ... x)
         bindings (tmp ... (a x)) body))

    ((let-values "mktmp" a e0 (arg ...)
        bindings (tmp ...) body)
     (call-with-values
       (lambda () e0)
       (lambda (arg ... . x)
         (let-values "bind"
             bindings (tmp ... (a x)) body))))))
```

```
(define-syntax let*-values
  (syntax-rules ()
    ((let*-values () body0 body1 ...)
     (begin body0 body1 ...))

    ((let*-values (binding0 binding1 ...)
         body0 body1 ...)
     (let-values (binding0)
       (let*-values (binding1 ...)
         body0 body1 ...)))))


(define-syntax begin
  (syntax-rules ()
    ((begin exp ...)
     ((lambda () exp ...)))))
```

The following alternative expansion for `begin` does not make use of the ability to write more than one expression in the body of a lambda expression. In any case, note that these rules apply only if the body of the `begin` contains no definitions.

```
(define-syntax begin
  (syntax-rules ()
    ((begin exp)
     exp)
    ((begin exp1 exp2 ...)
     (call-with-values
         (lambda () exp1)
       (lambda args
         (begin exp2 ...))))))
```

The following definition of `do` uses a trick to expand the variable clauses. As with `letrec` above, an auxiliary macro would also work. The expression `(if #f #f)` is used to obtain an unspecific value.

```
(define-syntax do
  (syntax-rules ()
    ((do ((var init step ...) ...)
         (test expr ...)
         command ...)
     (letrec
       ((loop
         (lambda (var ...)
           (if test
               (begin
                 (if #f #f)
                 expr ...)
               (begin
                 command
                 ...
                 (loop (do "step" var step ...)
                       ...))))))
       (loop init ...)))
    ((do "step" x)
     x)
    ((do "step" x y)
```

```
    y)))


(define-syntax case-lambda
  (syntax-rules ()
    ((case-lambda (params body0 body1 ...) ...)
     (lambda args
       (let ((len (length args)))
         (let-syntax
             ((cl (syntax-rules ::: ()
                    ((cl)
                     (error "no matching clause"))
                    ((cl ((p :::) . body) . rest)
                     (if (= len (length '(p :::)))
                         (apply (lambda (p :::)
                                  . body)
                                args)
                         (cl . rest)))
                    ((cl ((p ::: . tail) . body)
                         . rest)
                     (if (>= len (length '(p :::)))
                         (apply
                          (lambda (p ::: . tail)
                            . body)
                          args)
                         (cl . rest))))))
           (cl (params body0 body1 ...) ...)))))))
```

Appendix A. Standard Modules

This section lists the exports provided by the standard modules. The modules are factored so as to separate features which may not be supported by all implementations, or which may be expensive to load.

The `scheme` module prefix is used for all standard modules, and is reserved for use by future standards.

Base Module

The `(scheme base)` module exports many of the procedures and syntax bindings that are traditionally associated with Scheme.

```
_                   ...                 *
+                   -                   /
<=                  <                   =>
=                   >=                  >
abs                 and                 append
apply               assoc               assq
assv                begin               boolean?
bytevector-copy     bytevector-copy!
bytevector-length
bytevector-u8-ref
bytevector-u8-set!                      bytevector?
caar                cadr
call-with-current-continuation
call-with-values                        call/cc
car                 case-lambda         case
cdddar              cddddr              cdr
ceiling             char->integer       char<=?
char<?              char=?              char>=?
char>?              char?               complex?
cond                cond-expand         cons
define-syntax       define
define-record-type                      denominator
do                  dynamic-wind        else
eq?                 equal?              eqv?
error               error-object?
error-object-message
error-object-irritants                  even?
exact->inexact      exact-integer-sqrt
exact-integer?      exact?              expt
floor               for-each            gcd
guard               if                  import
inexact->exact      inexact?            integer->char
integer?            lambda              lcm
length              let*                let-syntax
letrec*             letrec-syntax       let-values
let*-values         letrec              let
list-copy           list->string        list->vector
list-ref            list-set!           list-tail
list?               list                make-bytevector
make-list           make-parameter      make-string
make-vector         map                 max
member              memq                memv
min                 modulo              negative?
not                 null?               number->string
number?             numerator           odd?
or                  pair?               parameterize
```

```
partial-bytevector
partial-bytevector-copy!          positive?
procedure?       quasiquote       quote
quotient         raise-continuable
raise            rational?        rationalize
real?            remainder        reverse
round            set!             set-car!
set-cdr!         string->list     string->number
string->symbol   string->vector   string-append
string-copy      string-fill!     string-for-each
string-length    string-map       string-ref
string-set!      string<=?        string<?
string=?         string>=?        string>?
string?          string           substring
symbol->string   symbol?          syntax-error
syntax-rules     truncate         values
unquote          unquote-splicing
vector-copy      vector->list     vector->string
vector-fill!     vector-for-each  vector-length
vector-map       vector-ref       vector-set!
vector?          vector           zero?
when             with-exception-handler
unless
```

Inexact Module

The `(scheme inexact)` module exports procedures which are typically only useful with inexact values.

```
exp              log              sqrt
sin              cos              tan
asin             acos             atan
finite?          nan?
```

Complex Module

The `(scheme complex)` module exports procedures which are typically only useful with complex values.

```
angle            magnitude        imag-part
real-part        make-polar
make-rectangular
```

Division Module

The `(scheme division)` module exports procedures for integer division.

```
floor/           floor-quotient   floor-remainder
ceiling/         ceiling-quotient
ceiling-remainder                 truncate/
truncate-quotient
truncate-remainder                round/
round-quotient   round-remainder  euclidean/
euclidean-quotient
euclidean-remainder
```

Lazy Module

The `(scheme lazy)` module exports forms for lazy evaluation.

```
delay            force            lazy
```

Eval Module

The `(scheme eval)` module exports procedures for evaluating Scheme data as programs.

```
eval             environment
null-environment
scheme-report-environment
```

Repl Module

The `(scheme repl)` module exports the `interaction-environment` procedure.

```
interaction-environment
```

Process Context Module

The `(scheme process-context)` module exports procedures for accessing with the program's calling context.

```
environment-variable
environment-variables             command-line
exit
```

Load Module

The `(scheme load)` module exports forms for loading and including Scheme expressions from files.

```
load             include          include-ci
```

I/O Module

The `(scheme io)` module exports procedures for general input and output on ports.

```
binary-port?     char-ready?      character-port?
close-port       close-input-port
close-output-port
current-error-port
current-input-port
current-output-port               eof-object?
flush-output-port
get-output-string
get-output-bytevector             input-port?
newline          open-input-string
open-output-string
open-input-bytevector
open-output-bytevector            output-port?
peek-char        peek-u8?         port?
port-open?       read-char        read-line
read-u8          u8-ready?        write-char
write-u8
```

File Module

The `(scheme file)` module provides procedures for accessing files.

```
call-with-input-file
call-with-output-file             delete-file
file-exists?     open-input-file
open-output-file
open-binary-input-file
open-binary-output-file
with-input-from-file
with-output-to-file
```

Read Module

The `(scheme read)` module provides procedures for reading Scheme objects.

```
read
```

Write Module

The `(scheme write)` module provides procedures for writing Scheme objects.

```
write            display
```

Char Module

The `(scheme char)` module provides procedures for dealing with Unicode character operations.

```
char-alphabetic?                  char-ci=?
char-ci<?        char-ci>?        char-ci<=?
char-ci>=?       char-upcase      char-downcase
char-foldcase    char-lower-case?
char-numeric?    char-upper-case?
char-whitespace?                  string-ci=?
string-ci<?      string-ci>?      string-ci<=?
string-ci>=?     string-upcase    string-downcase
string-foldcase
```

Char Normalization Module

The `(scheme char normalization)` module provides procedures for dealing with Unicode normalization operations.

```
string-ni=?      string-ni<?      string-ni>?
string-ni<=?     string-ni>=?
```

Time

The `(scheme time)` module provides access to the system time.

```
current-second   current-jiffy
jiffies-per-second
```

Appendix B. Standard Feature Identifiers

An implementation may provide any or all of the feature identifiers listed in table B.1, as well as any others that it chooses, but must not provide a feature identifier if it does not provide the corresponding feature. These features are used by `cond-expand` to conditionally include module declarations in a module.

Feature identifier	Feature description
r7rs	All R7RS Scheme implementations have this feature.
exact-closed	All rational operations except / produce exact values given exact inputs.
ratios	/ with exact arguments produces an exact result when the divisor is nonzero.
ieee-float	Inexact numbers are IEEE 754 floating point values.
full-unicode	All Unicode characters are supported.
windows	This Scheme implementation is running on Windows.
posix	This Scheme implementation is running on a Posix system.
unix, darwin, linux, bsd, freebsd, solaris, ...	Operating system flags (more than one may apply).
i386, x86-64, ppc, sparc, jvm, clr, llvm, ...	CPU architecture flags.
ilp32, lp64, ilp64, ...	C memory model flags
big-endian, little-endian	Byte order flags.
⟨name⟩	The name of this implementation.
⟨name-version⟩	The name and version of this implementation.

Table B.1: Standard Feature Identifiers

NOTES

Language changes

This section enumerates the changes that have been made to Scheme since the "Revised^5 report" [2] was published.

While this report is in draft status the list should be considered incomplete and subject to change.

- Modules have been added as a new program structure to improve encapsulation and sharing of code. Some existing and new identifiers have been factored out into separate modules.
- Exceptions can now be signalled explicitly with `raise`, `raise-continuable` or `error`, and can be handled with `with-exception-handler` and the `guard` syntax.
- New disjoint types supporting access to multiple fields can be generated with SRFI 9's `define-record-type`.
- Parameter objects can be created with `make-parameter`, and dynamically rebound with `parameterize`.
- *Bytevectors*, homogeneous vectors of integers in the range [0..255], have been added as a new disjoint type.
- *Ports* can now be designated as *binary* or *character* ports, with new procedures for reading and writing binary data.
- *String ports* have been added as a way to read and write characters to and from strings, and *bytevector ports* to read and write bytes to and from bytevectors.
- `Current-input-port` and `current-output-port` are now parameter objects, along with the newly introduced `current-error-port`.
- `Syntax-rules` now recognizes _ as a wildcard, allows the ellipsis symbol to be specified explicitly instead of the default `...`, allows template escapes with an ellipsis-prefixed list, and allows tail patterns to follow an ellipsis pattern.
- `Syntax-error` has been added as a way to signal immediate and more informative errors when a form is expanded.
- Internal `define-syntax` forms are now allowed wherever internal `define`s are.
- `Letrec*` has been added, and internal define specified in terms of it.
- `Case` now supports a `=>` syntax analogous to `cond`.
- `Case-lambda` has been added to the base module as a way to dispatch on the number of arguments passed to a procedure.
- `When` and `unless` have been added as convenience conditionals.
- Positive and negative infinity and a NaN object have been added to the numeric tower as inexact values with the written representations `+inf.0`, `-inf.0` and `+nan.0`, respectively.
- `Map` and `for-each` are now required to terminate on the shortest list when inputs have different length.
- `Member` and `assoc` now take an optional third argument for the equality predicate to use.
- `Exact-integer?` and `exact-integer-sqrt` have been added.
- `Make-list`, `list-copy`, `list-set!`, `string-map`, `string-for-each`, `string->vector`, `vector-copy`,

`vector-map`, `vector-for-each`, and `vector->string` have been added to round out the sequence operations.

- The set of characters used is required to be consistent with the Unicode Standard, but only in so far as the implementation supports Unicode.
- `string-ni=?` and related procedures have been added to compare strings as though they had gone through an implementation-defined normalization, without exposing the normalization.
- The case-folding behavior of the reader can now be explicitly controlled, with no folding as the default.
- The reader now recognizes the new comment syntax `#;` to skip the next datum, and allows nested block comments with `#| ... |#`.
- Data prefixed with reader labels `#<n>=` can be referenced with `#<n>#` allowing for reading and writing of data with shared structure.
- Strings and symbols now allow mnemonic and numeric escape sequences, and the list of named characters has been extended.
- `File-exists?` and `delete-file` are available in the `(scheme file)` module.
- An interface to the system environment and command line is available in the `(scheme process-context)` module.
- Procedures for accessing the current time are available in the `(scheme time)` module.
- A complete set of integer division operators is available in the `(scheme division)` module.
- `Transcript-on` and `transcript-off` have been removed.

ADDITIONAL MATERIAL

The Internet Scheme Repository at

`http://www.cs.indiana.edu/scheme-repository/`

contains an extensive Scheme bibliography, as well as papers, programs, implementations, and other material related to Scheme.

The Scheme community website at

`http://schemers.org/`

contains additional resources for learning and programming, job and event postings, and Scheme user group information.

A bibliography of Scheme-related research at

`http://library.readscheme.org/`

links to technical papers and theses related to the Scheme language, including both classic papers and recent research.

EXAMPLE

`Integrate-system` integrates the system

$$y_k' = f_k(y_1, y_2, \ldots, y_n),\ k = 1, \ldots, n$$

of differential equations with the method of Runge-Kutta.

The parameter `system-derivative` is a function that takes a system state (a vector of values for the state variables $y_1, \ldots, y_n$) and produces a system derivative (the values $y_1', \ldots, y_n'$). The parameter `initial-state` provides an initial system state, and `h` is an initial guess for the length of the integration step.

The value returned by `integrate-system` is an infinite stream of system states.

```
(define integrate-system
  (lambda (system-derivative initial-state h)
    (let ((next (runge-kutta-4 system-derivative h)))
      (letrec ((states
                (cons initial-state
                      (delay (map-streams next
                                          states)))))
        states))))
```

`Runge-Kutta-4` takes a function, `f`, that produces a system derivative from a system state. `Runge-Kutta-4` produces a function that takes a system state and produces a new system state.

```
(define runge-kutta-4
  (lambda (f h)
    (let ((*h (scale-vector h))
          (*2 (scale-vector 2))
          (*1/2 (scale-vector (/ 1 2)))
          (*1/6 (scale-vector (/ 1 6))))
      (lambda (y)
        ;; y is a system state
        (let* ((k0 (*h (f y)))
               (k1 (*h (f (add-vectors y (*1/2 k0)))))
               (k2 (*h (f (add-vectors y (*1/2 k1)))))
               (k3 (*h (f (add-vectors y k2)))))
          (add-vectors y
            (*1/6 (add-vectors k0
                               (*2 k1)
                               (*2 k2)
                               k3)))))))))

(define elementwise
  (lambda (f)
    (lambda vectors
      (generate-vector
        (vector-length (car vectors))
        (lambda (i)
          (apply f
                 (map (lambda (v) (vector-ref  v i))
                      vectors)))))))

(define generate-vector
  (lambda (size proc)
    (let ((ans (make-vector size)))
      (letrec ((loop
                (lambda (i)
                  (cond ((= i size) ans)
                        (else
                         (vector-set! ans i (proc i))
                         (loop (+ i 1)))))))
        (loop 0)))))

(define add-vectors (elementwise +))

(define scale-vector
  (lambda (s)
    (elementwise (lambda (x) (* x s)))))
```

`Map-streams` is analogous to `map`: it applies its first argument (a procedure) to all the elements of its second argument (a stream).

```
(define map-streams
  (lambda (f s)
    (cons (f (head s))
          (delay (map-streams f (tail s))))))
```

Infinite streams are implemented as pairs whose car holds the first element of the stream and whose cdr holds a promise to deliver the rest of the stream.

```
(define head car)
(define tail
  (lambda (stream) (force (cdr stream))))
```

The following illustrates the use of `integrate-system` in integrating the system

$$C\frac{dv_C}{dt} = -i_L - \frac{v_C}{R}$$

$$L\frac{di_L}{dt} = v_C$$

which models a damped oscillator.

```
(define damped-oscillator
  (lambda (R L C)
    (lambda (state)
      (let ((Vc (vector-ref state 0))
            (Il (vector-ref state 1)))
        (vector (- 0 (+ (/ Vc (* R C)) (/ Il C)))
                (/ Vc L))))))

(define the-states
  (integrate-system
     (damped-oscillator 10000 1000 .001)
     '#(1 0)
     .01))
```

REFERENCES

[1] Michael Sperber, R. Kent Dybvig, Mathew Flatt, and Anton van Straaten, editors. The revised[6] report on the algorithmic language Scheme.

[2] Richard Kelsey, William Clinger, and Jonathan Rees, editors. The revised[5] report on the algorithmic language Scheme.

[3] Harold Abelson and Gerald Jay Sussman with Julie Sussman. *Structure and Interpretation of Computer Programs, second edition.* MIT Press, Cambridge, 1996.

[4] Alan Bawden and Jonathan Rees. Syntactic closures. In *Proceedings of the 1988 ACM Symposium on Lisp and Functional Programming*, pages 86–95.

[5] Robert G. Burger and R. Kent Dybvig. Printing floating-point numbers quickly and accurately. In *Proceedings of the ACM SIGPLAN '96 Conference on Programming Language Design and Implementation*, pages 108–116.

[6] William Clinger, editor. The revised revised report on Scheme, or an uncommon Lisp. MIT Artificial Intelligence Memo 848, August 1985. Also published as Computer Science Department Technical Report 174, Indiana University, June 1985.

[7] William Clinger. How to read floating point numbers accurately. In *Proceedings of the ACM SIGPLAN '90 Conference on Programming Language Design and Implementation*, pages 92–101. Proceedings published as *SIGPLAN Notices* 25(6), June 1990.

[8] William Clinger and Jonathan Rees, editors. The revised[4] report on the algorithmic language Scheme. In *ACM Lisp Pointers* 4(3), pages 1–55, 1991.

[9] William Clinger and Jonathan Rees. Macros that work. In *Proceedings of the 1991 ACM Conference on Principles of Programming Languages*, pages 155–162.

[10] William Clinger. Proper Tail Recursion and Space Efficiency. To appear in *Proceedings of the 1998 ACM Conference on Programming Language Design and Implementation*, June 1998.

[11] R. Kent Dybvig, Robert Hieb, and Carl Bruggeman. Syntactic abstraction in Scheme. *Lisp and Symbolic Computation* 5(4):295–326, 1993.

[12] Carol Fessenden, William Clinger, Daniel P. Friedman, and Christopher Haynes. Scheme 311 version 4 reference manual. Indiana University Computer Science Technical Report 137, February 1983. Superseded by [13].

[13] D. Friedman, C. Haynes, E. Kohlbecker, and M. Wand. Scheme 84 interim reference manual. Indiana University Computer Science Technical Report 153, January 1985.

[14] *IEEE Standard 754-1985. IEEE Standard for Binary Floating-Point Arithmetic.* IEEE, New York, 1985.

[15] *IEEE Standard 1178-1990. IEEE Standard for the Scheme Programming Language.* IEEE, New York, 1991.

[16] Eugene E. Kohlbecker Jr. *Syntactic Extensions in the Programming Language Lisp.* PhD thesis, Indiana University, August 1986.

[17] Eugene E. Kohlbecker Jr., Daniel P. Friedman, Matthias Felleisen, and Bruce Duba. Hygienic macro expansion. In *Proceedings of the 1986 ACM Conference on Lisp and Functional Programming*, pages 151–161.

[18] Peter Landin. A correspondence between Algol 60 and Church's lambda notation: Part I. *Communications of the ACM* 8(2):89–101, February 1965.

[19] MIT Department of Electrical Engineering and Computer Science. Scheme manual, seventh edition. September 1984.

[20] Peter Naur et al. Revised report on the algorithmic language Algol 60. *Communications of the ACM* 6(1):1–17, January 1963.

[21] Paul Penfield, Jr. Principal values and branch cuts in complex APL. In *APL '81 Conference Proceedings,* pages 248–256. ACM SIGAPL, San Francisco, September 1981. Proceedings published as *APL Quote Quad* 12(1), ACM, September 1981.

[22] Kent M. Pitman. The revised MacLisp manual (Saturday evening edition). MIT Laboratory for Computer Science Technical Report 295, May 1983.

[23] Jonathan A. Rees and Norman I. Adams IV. T: A dialect of Lisp or, lambda: The ultimate software tool. In *Conference Record of the 1982 ACM Symposium on Lisp and Functional Programming*, pages 114–122.

[24] Jonathan A. Rees, Norman I. Adams IV, and James R. Meehan. The T manual, fourth edition. Yale University Computer Science Department, January 1984.

[25] Jonathan Rees and William Clinger, editors. The revised[3] report on the algorithmic language Scheme. In *ACM SIGPLAN Notices* 21(12), pages 37–79, December 1986.

[26] John Reynolds. Definitional interpreters for higher order programming languages. In *ACM Conference Proceedings*, pages 717–740. ACM, 1972.

[27] Guy Lewis Steele Jr. and Gerald Jay Sussman. The revised report on Scheme, a dialect of Lisp. MIT Artificial Intelligence Memo 452, January 1978.

[28] Guy Lewis Steele Jr. Rabbit: a compiler for Scheme. MIT Artificial Intelligence Laboratory Technical Report 474, May 1978.

[29] Guy Lewis Steele Jr. *Common Lisp: The Language, second edition.* Digital Press, Burlington MA, 1990.

[30] Gerald Jay Sussman and Guy Lewis Steele Jr. Scheme: an interpreter for extended lambda calculus. MIT Artificial Intelligence Memo 349, December 1975.

[31] Joseph E. Stoy. *Denotational Semantics: The Scott-Strachey Approach to Programming Language Theory.* MIT Press, Cambridge, 1977.

[32] Texas Instruments, Inc. TI Scheme Language Reference Manual. Preliminary version 1.0, November 1985.

ALPHABETIC INDEX OF DEFINITIONS OF CONCEPTS, KEYWORDS, AND PROCEDURES

The principal entry for each term, procedure, or keyword is listed first, separated from the other entries by a semicolon.